
第1章 走进AutoCAD ActiveX/VBA

1.1 ActiveX 自动化在AutoCAD中的应用

ActiveX是微软近年推出的一个基于COM（Component Object Model）的技术规范。它是在OLE（Object Linked and Embedded）基础上发展的新技术，其宗旨是在Windows系统的统一管理下协调不同的应用程序，构成复合文档。应用程序可以决定将自己哪些信息暴露出来，这个暴露自身信息的应用程序称为服务程序（Server）；而操纵服务程序的程序称为客户程序（Client）。所谓ActiveX部件（简称“部件”），是一段可重复使用的编程代码和数据，它由利用ActiveX Automation（自动化）技术创建的一个或多个对象所组成。由于编写客户程序比将其做成服务程序要简单得多，因此要利用已有的ActiveX部件作为被调用的服务程序。

Autodesk公司采纳了微软的这一技术规范，最早在AutoCAD R14.0版推出了基于ActiveX Automation技术的开发方式。随后就有AutoCAD R14.01版，该版支持“前期绑定”，并且内嵌了AutoCAD VBA 。到AutoCAD 2000/2002（属R15版）、AutoCAD 2004/2005/2006（属R16版）和AutoCAD 2007（属R17版），这种新型的开发方式得到了进一步的改进及完善。

AutoCAD ActiveX 使用户能够从 AutoCAD 的内部或外部以编程方式来操作 AutoCAD。它是通过将 AutoCAD 对象显示到“外部世界”来做到这一点的。一旦这些对象被“暴露”，许多不同的编程语言或其他应用程序（例如VB、C++、Java、Word VBA 或 Excel VBA等）就可以访问它们。VB功能强大、语法简单、应用广泛，是良好的外部编程环境。AutoCAD VBA是内嵌在AutoCAD内部的编程环境。

AutoCAD VBA 允许 VBA 环境与 AutoCAD 同时运行。VBA 通过 AutoCAD ActiveX Automation 接口将消息发送到 AutoCAD，并通过 ActiveX Automation 接口对 AutoCAD 进行编程控制。将AutoCAD 、ActiveX Automation 和 VBA/VB 相结合编程有三个基本要素。第一个要素是 AutoCAD 本身，它拥有丰富的封装了的 AutoCAD 图元、数据和命令的对象集。 AutoCAD 是一个具有多层接口的开放架构应用程序。第二个要素是 AutoCAD ActiveX Automation 接口，它建立与 AutoCAD 对象的消息传递（通信）。第三个要素是 VBA/VB 集成编程环境 （IDE），它具有自己的对象组、关键词和常量等，能提供程序流、控制、调试和执行等功能。

在AutoCAD 2000/2002中，几乎对外暴露了所有对象，包括图形对象（如直线、圆弧）、样式设置对象（如线型、文本样式）、组织结构对象（如图层）、图形显示对象（如视口）以及菜单和工具栏等。到AutoCAD 2004/2005/2006版，又得到逐步的改进，如真彩色控制和表等。

1.2 AutoCAD VBA基础

1.2.1概述

VBA （Visual Basic for Applications）与VB（Visual Basic）一样也是一种面向对象的程序设计语言，它继承了VB语法简单功能强大的特点。

AutoCAD 2000/2002 VBA相当于VB5，而AutoCAD 2004/2005/2006 VBA相当于VB6，VB所支持的对象属性和方法，VBA也支持。

AutoCAD VBA 允许其Visual Basic环境与 AutoCAD 同时运行，并通过 ActiveX Automation 接口提供对 AutoCAD 的编程控制。这样就把 AutoCAD、ActiveX Automation和 VBA 紧密连结在一起，提供一个非常强大的接口。它不仅能控制 AutoCAD 对象，也能向其他应用程序发送数据或从中提取数据。

在AutoCAD中与VBA应用程序开发有关的下拉菜单如图1-1所示。

图1-1与VBA应用程序开发有关的下拉菜单

注：VBA 不支持创建可执行文件，但它提供了口令保护，以工程的方式保护工程窗体、类和模块的可见性。用户可以在 VBA IDE 中选择菜单“工具（T）”|“ACAD Project属性（E）．．．”，在“ACAD Project- 工程属性”对话框的“保护”选项卡进行相应的操作。

VBA常用的工程术语有：

（1）内嵌工程(embedded project)：内嵌在AutoCAD图形中的VBA工程，打开图形时自动加载。

（2）全局（通用）工程（global project）：指独立存储在“.dvb”文件中的工程，需要先加载。

（3）一般（规则）文档：未包含VBA内嵌工程的AutoCAD图形。

（4）智能文档：包含一个或多个VBA内嵌工程的AutoCAD图形。

（5）当前工程：VBA IDE中当前选定的工程。

（6）宏：用户可直接执行的VBA通用子程序或函数。每一个VBA工程至少包含一个宏。

VBA工程可以包含的组件有对象、窗体、标准模块、类模块和引用。下面对这些元素分别介绍：

（1）ThisDrawing对象：代表当前（活动）的AutoCAD图形（文档）。对于全局工程，它指AutoCAD中的当前（活动）文档；对于内嵌工程，指包含此工程的文档。一般情况下，创建一个工程后，该工程只含有一个元素——ThisDrawing对象，其它部分要用户根据需要添加。

（2）窗体：窗体是放置其它对象的容器，是基本的构造块。用户可以通过它为应用程序创建自定义的对话框。它可以包括事件过程、通用过程和变量的窗体级声明。

（3）标准模块：标准模块也称作“代码模块”或简称为“模块”。 它包含用户的常用过程和函数。

（4）类模块：类模块创建具有属性和方法的用户自定义的对象。

（5）引用：指能被应用程序使用的外部ActiveX部件或其它工程文件。

1.2.2 VBA工程及其管理

当前AutoCAD进程中加载的所有VBA工程都可以用“VBA管理器”的对话框来管理。可使用它创建、加载、卸载、保存、内嵌或提取VBA工程，还可查看有哪些工程内嵌在打开的图形中（如果有的话）。可选择菜单“工具（T）”|“宏（A）”|“VBA管理器（V）...”或键入命令“VBAMAN”打开VBA管理器，如图1-2所示。

图1-2 VBA管理器

1.加载工程

当需要加载已有的工程时，可应用菜单“工具”|“宏(A)”|“VBA管理器...”打开“VBA管理器”对话框，点击“加载（L）...”，按钮或选择菜单“工具”|“宏(A)”|“加载工程（L）...”打开 ”打开VBA工程” 对话框。如图1-3所示。

图1-3 ”打开VBA工程” 对话框

当选取工程将其加载到AutoCAD图形中时，此工程引用的其它工程也会被自动加载。也可在命令提示行键入“VBALOAD”执行加载。

2．卸载工程

在”VBA管理器”对话框中选择要卸载的工程，单击“卸载（U）”即可。也可在命令提示行键入“VBAUNLOAD” 执行卸载。

3. 嵌入工程

将工程嵌入到AutoCAD图形中，是指把全局工程的副本放到图形的数据库中，此后该工程就将随着图形的打开而加载，以及随着图形的关闭而卸载。注意，一个图形只能嵌入一个内嵌工程。

欲将工程嵌入到AutoCAD图形中，应单击“VBA管理器”上的“嵌入（E）”按钮。

4．提取工程

当单击“VBA管理器”上的“提取（X）”按钮时，就使该工程从所在的图形数据库中删除。用户可将该工程另存到外部文件中，否则该工程的数据将会被删除。

5. 创建新工程

要创建新的VBA工程，应单击“VBA管理器”上的“新建（N）”按钮。一个新建VBA工程的默认名称为ACADProject，若要变更工程的名称，应在VBA IDE集成开发环境中的工程属性栏中修改。

6. 命名工程和保存工程

工程的名称是在VBA IDE的“属性”窗口中设定的，而保存全局工程的文件名（.dvb）应在“VBA管理器”中单击“另存为（S）．．．”按钮，在对话框的“文件名（N）”框内输入文件名（.dvb），文件名是存储工程文件的唯一标识。对于一个已嵌入到图形中的工程，在保存图形文件时将一起被保存。

1.2.3 VBA 集成开发环境

在AutoCAD中加载了工程后，就可以利用VBA集成开发环境(IDE)来编辑该工程的窗体、程序代码和引用等元素，也可以在VBA IDE中调试和运行工程。

1．打开VBA IDE

可用以下三种方法打开VBA IDE：

（1）在命令行中键入“VBAIDE”。

（2）应用菜单“工具”|“宏（A）”|“Visual Basic编辑器（B）”。

（3）直接按“Alt+F11”键。

打开的VBA IDE见图1-4。

图1-4打开的VBA IDE

注：可以设定在每次打开AutoCAD时让系统自动启动VBA IDE，只需在acad.arx文件中增添一行代码“acadvba.arx”即可。

2．工程信息

可在VBA IDE 中的“工程”窗口查看所有已经加载的VBA工程。并可查看这些工程中所包含的程序代码、类别、窗体及与工程相关的文件等。

3．工程的组成

每个工程都是由不同元素构成的。它可以包含AutoCAD图形、窗体、标准模块、类模块和引用。

4．添加新元素

用户可以在工程中添加新元素，其方法是从“插入(I)”菜单选择相应的选项进行添加。当新的元素添加到工程中以后，就会以默认的属性显示在“工程”窗口中。

5．导入已存在的元素

用户可以将已存在的元素导入到当前的工程中。可以导入的文件包括：窗体文件（.FRM）、标准模块文件（.BAS）、类模块文件(.CLS)。当导入已存在的元素时，新添加到当前工程中的是原文件的副本，对导入元素的修改不会影响到原始文件。

导入元素的方法是在菜单“文件(F)”中选“导入文件（I）...”项，打开“导入文件”对话框，选择需要的文件即可。导入后，即显示在“工程”窗口中。

6．编辑元素

1） 使用窗体编辑器

在窗体编辑器中，用户可以进行应用程序界面的设计。可通过“工程”窗口中选择要编辑的窗体，再选择菜单“视图（V）”|“对象窗口（B）”，将窗口编辑器切换出来（也可点击“工程” 窗口左上角的“查看对象”按钮）。接下来即可编辑窗体及其上面控件的属性。

2） 使用代码编辑器

在代码编辑器中，用户可以进行代码的编辑。可通过“工程”窗口中选择要编辑的模块，再选择菜单“视图（V）”|“代码窗口（C）”，将代码窗口切换出来（也可点击“工程” 窗口左上角的“查看代码”按钮）。接下来即可编辑事件过程的代码。

7．设置VBA IDE选项

选择“工具（T）”|“选项（O）...”调出“选项”对话框，如图1-5所示。应用该对话框可以设置VBA IDE的环境特征。它包括四个选项卡，其内容分别为：“编辑器”、“ 编辑器格式”、“通用”和“可连接的”。比如，利用“通用”选项卡，用户可以设置在设计状态下是否显示窗体上的网格，设置网格大小以及控件是否与网格对齐。又如利用“编辑器” 选项卡设定是否强制《要求变量声明》。


图1-5 VBA IDE的“选项”对话框

1.2.4宏及其管理

1．“宏“对话框

从 AutoCAD 的菜单“工具(T)”|“宏（A）” |“宏（M）...”或在 AutoCAD 命令提示中输入 “VBARUN” ，即可打开“宏”对话框。“宏”对话框显示了图形文件或工程文件中包含的所有宏。通过“宏”对话框，用户可以运行、编辑、删除、创建宏及设置 VBA 工程选项。用户可以通过“宏位置（A）”的下拉列表改变宏的选用范围。打开的“宏”对话框如图1-6所示。

如果在VBARUN命令前加“_”，则可直接在命令行提示输入要运行的宏及运行该宏。

图1-6“宏”对话框

2. 运行宏

运行宏就是在当前（活动）的 AutoCAD图形中执行宏的程序代码。当前（活动）的图形指的是宏开始执行时焦点所在的打开图形。对于全局工程中的宏，ThisDrawing 对象的所有 VBA 引用都将指向当前（活动）的图形。对于内嵌工程中的宏，ThisDrawing 对象总是指内嵌了宏的图形。

运行宏时，首先要打开“宏”对话框，选定要运行的宏，然后单击“运行（R）”按钮即可。

允许在AutoCAD命令行键入“VBASTMT”直接运行VBA语句。VBA语句中可以包括关键字、操作符、变量、常量、表达式等内容。语句一般只占一行，若需要多句，可用冒号（“：”）隔开。

3. 编辑和调试宏

若需修改和编辑已有的宏，需要在 VBA IDE开发环境中进行。

宏的调试是按单步运行的方式。其方法是：先在“宏”对话框中选择要调试的宏，然后选择按钮“逐语句（S）”。

4. 创建新宏

利用“宏”对话框可以创建新的宏，步骤：

（1）首先在“宏名称（M）”文本框内键入宏的名称，例如键入“acad_firstVBA”。

（2）在“宏位置（A）”下拉列表中选取要建立宏的工程，例如选取”acad_firstVBA.dwg”，自动生成的宏名称如图1-6所示。

若未选取“宏位置（A）”，当运行“创建”时，会自动弹出对话框提示操作者确认。

（3）单击“创建（C）”按钮，出现VBA IDE。在Sub acad_firstVBA( )和End Sub语句之间键入程序代码，如图1-7 的VBA IDE程序代码栏中所示。

图1-7创建新宏

5. 删除宏

利用“宏”对话框可以删除宏，步骤：

（1）选择要删除的宏的名称。

（2）单击“删除（D）”按钮。

6.设置选项

当单击“宏”对话框中的“选项（O）”按钮，即弹出设置VBA的“选项”对话框，如图1-8所示。

图1-8 宏的“选项”对话框

1.2.5 VBA中的窗体

1． 用户窗体

窗体（Form）是开发应用程序的基本内容之一，通过窗体可以为用户创建自定义对话框。新建的窗体是空的，需要用户调整其大小和位置，并在上面安放控件。VBA提供的控件都在工具箱中，用户根据设计的需要从中选取，并可调整控件的大小和位置。

AutoCAD VBA只支持自定义窗体。自定义窗体（本书也称为窗体）可以是模式窗体，也可以是无模式窗体。窗体的 ShowModal 属性确定该窗体是模式窗体还是无模式窗体。当在 AutoCAD VBA 中将一个对话框定义为模式对话框时，用户必须先响应该对话框，才允许继续操作应用程序的其他部分。

2． 设计模式和运行模式

当用户在创建窗体时，是属于设计模式状态，此时用户可以进行如下工作 ：

（1）向窗体中添加控件。

（2）更改窗体的属性。

（3）更改窗体上控件的属性。

（4）向窗体模块添加代码。

在设计模式中，用户、AutoCAD 的用户界面以及窗体之间没有数据交互发生。 

当运行应用程序时，窗体就变为运行模式。在运行模式中，用户不能直接调整窗体及其上的控件。窗体显示在 AutoCAD 的界面中，用户可以与窗体进行交互。

3．添加控件

向窗体中添加控件非常容易。只需从控件工具箱中选择控件，然后将该控件拖放到窗体上即可。释放鼠标时，控件的副本就被放到窗体上。当控件位于窗体上之后，用户就可以修改控件的位置和大小。用户可以根据自己的需要复制任意数量的控件。用户还可以再更改控件的大小和位置、使用格式化控制（使用VBA IDE的“格式（O）”下拉菜单中的选项）、 编辑控件的属性、以及添加控件的程序代码（双击控件进入代码窗口）。

4．显示和隐藏窗体 

显示窗体需要使用窗体本身的Show 方法。

例：本例显示一个名称为“UserForm1”的窗体： 

Public Sub MyApplication( )

UserForm1.Show

End Sub

默认情况下，创建的窗体是模式窗体，所以当显示该窗体时，用户不能与 AutoCAD 直接进行交互。例如，当窗体显示时，用户不能选择图形中的点和对象。如果要求用户访问 AutoCAD 图形，要使用窗体本身的Hide 方法先将该窗体隐藏起来。当使用 Hide 方法时，该窗体并未从内存中卸载，只是没有显示出来而已。 

5．加载和卸载窗体 

有时候用户可能希望在运行时将窗体加载到内存中，但不需要立即显示窗体。这种操作有助于更好地控制何时在应用程序中加载窗体，或者可以通过编程访问窗体同时又不会向用户显示窗体。 

要加载窗体同时又不显示该窗体，须使用窗体本身的Load 方法。 

如果调用 Show 方法时窗体尚未加载，将会自动加载窗体。 

用户也可能需要卸载窗体。卸载窗体会从内存中卸载该窗体并回收与窗体关联的所有内存。在使用 Load 或 Show 方法重新加载窗体之前，用户无法与窗体交互，而且不能以编程的方式来操作此窗体。当用户知道应用程序中不再使用该窗体并希望回收内存时，就可以选择卸载窗体。 

卸载窗体需要使用窗体本身的UnLoad方法。当应用程序结束时，窗体将会自动卸载。

1.2.6第一个VBA编程示例

利用AutoCAD ActiveX/VBA进行二次开发，在模型空间写“AutoCAD ActiveX/VBA”。

本例一般可采用以下两种方式：

《方式一》：添加用户窗体

步骤：

（1）在AutoCAD 打开一个新图形。

（2）选菜单“工具（T）”|“宏（A）”|“Visual Basic编辑器（B）”。

（3）在当前工程中添加一个用户窗体。

选菜单“插入（I）”|“用户窗体（U）”后，出现UserForm1。

在该窗体上添加按钮CommandButton1和CommandButton2 ，并依次将按钮的Caption属性设为“运行”和“退出”。

用户窗体如图1-9所示。

图1-9第一个VBA编程示例（用户窗体）

双击“运行” 按钮，出现该事件过程的程序代码窗口，输入下列代码：

	VB源码彩色显示 注：在[code]...[/code]内就以下面格式显示文字，双击源码拷贝到剪切板

	Private Sub CommandButton1_Click( )

Dim textPoint(0 To 2) As Double ′声明文字插入点

Dim textHeight As Double ′声明文字高度

Dim textStr As String ′声明字符串

Dim textObject As AcadText ′声明文字对象

textPoint(0) = 20 ′设置插入点的X坐标

textPoint(1) = 40 ′设置插入点的Y坐标

textPoint(2) = 0 ′设置插入点的Z坐标

textHeight = 10 ′设置文字高度为10

textStr = "AutoCAD ActiveX/VBA " ′设置字符串

′创建text对象

Set textObject = ThisDrawing.ModelSpace.AddText _

(textStr, textPoint, textHeight)

ThisDrawing.Application.ZoomExtents ′根据实际范围计算缩放

textObject.Update ′更新显示图形对象

End Sub

再双击“结束” 按钮，输入下列代码：

Private Sub CommandButton2_Click( )

Unload Me ′卸载窗体

End ′结束应用程序

End Sub



当点击菜单或工具栏上的运行钮时，图1-9中所示的用户窗体就变为显示在AutoCAD中的用户界面。单击其上的“运行” 按钮，运行结果如图1-10所示。

图1-10第一个VBA编程示例的运行结果

当单击用户界面上的“结束” 按钮，可使窗体卸载并结束应用程序。若直接点击VBA IDE工具栏上的重新设置（■）钮，也可结束应用程序。本例可按名为“acad_firstVBA.dvb”的全局工程保存该工程文件。

《方式二》：添加过程

步骤：

（1）在AutoCAD 打开一个新图形

（2）选菜单“工具（T）”|“宏（A）”|“Visual Basic编辑器（B）”

（3）在当前工程中添加一个标准模块

选菜单“插入（I）”|“模块（M）”后，系统会将代码窗口切换出来。

（4）添加一个通用子过程

选菜单“插入（I）”|“过程（P）...”后，弹出《添加过程》的对话框。如图1-11所示。 

图1-11“添加过程”的对话框

当键入名称“acad_firstVBA”并单击“确定”后，回到VBA IDE。在程序代码栏出现的Public Sub acad_firstVBA( )和End Sub语句间输入程序代码。

若省略添加“模块（M）”这一步，而在VBA IDE窗口的“视图”菜单中选择“代码窗口”项后，直接选菜单“插入”|“过程（P）...“，也可取得同样效果。结果如图1-12中所示。

图1-12第一个编程示例的VBA IDE（添加过程）

当点击下拉菜单或工具栏上的运行钮时，运行结果也如图1-10所示

第2章Visual Basic程序设计基础

本章简略介绍Visual Basic 6.0程序设计的基本概念和编程基础。

2.1设计VB的第一个应用程序

2.1.1创建VB应用程序引例

使用Visual Basic进行Windows程序设计的精髓是“可视化”和“事件驱动”。本引例通过生成一个简易加法计算器，引出上述两个特征。

1． 创建应用程序的窗体

当进入VB集成开发环境时，会出现一个默认的名叫“Form1”的窗体，该窗体就是创建应用程序的第一个用户界面。可用鼠标单击“工具箱”中的控件，将其拖到窗体需要的位置上。窗体和控件的属性由“属性”窗口中列出，在设计阶段，可采取修改其属性值的方式确定之。

对于本例，在窗体Form1上至少应创建以下控件:

（1）两个命令按钮（Command1和Command2），分别确定其Caption属性为“加法运算”和“结束”。

（2）两个文本框（TextBox1和TextBox2），供输入被加数和加数。

（3）一个标签（Label3），输出（显示）计算结果。

VB6.0集成开发环境及本例所创建的窗体如图2-1所示。

图2-1 VB6.0集成开发环境及创建的窗体

2．编写事件过程的代码

双击窗体或窗体上的控件（或单击“工程资源管理器”窗口上的“查看代码”按钮）进入代码编辑窗口。如图2-2所示，在代码编辑窗口内看到的字符就是相应事件驱动过程的程序代码。

当开始运行该程序，即可看到用户界面。首先在文本框输入相应数值，然后单击《加法运算》按钮，执行事件过程后得出运算结果。若《结束》按钮，程序结束。

图2-2 VB6.0集成开发环境及代码编辑窗口

本例窗体模块以文件名为”Ljfq.frm”存盘和工程文件以文件名为”VB引例1-加法器.vbp”存盘。显示出的文件名如图2-1和图2-2 中所示。

实例：若要开发出一个能进行加减乘除的简易计算器，可按下列步骤进行： 

（1）创建应用程序的窗体

窗体的用户界面如图2-3所示。


图2-3简易计算器的用户界面及运行结果

其中：文本框Text1,Text2——输入数据；

组合框Combo1——改换运算符：

标签Label1,Label3——显示“=”号及“计算结果”；

命令按钮Command1,Command2,Command3——运行《运算》、《清零》和《结束》。

（2）编写事件过程代码

	VB源码彩色显示 注：在[code]...[/code]内就以下面格式显示文字，双击源码拷贝到剪切板

	Private Sub Command1_Click( )

If Combo1.Text = "+" Then

Label3.Caption = Val(Text1.Text) + Val(Text2.Text)

ElseIf Combo1.Text = "-" Then

Label3.Caption = Val(Text1.Text) - Val(Text2.Text)

ElseIf Combo1.Text = "×" Then

Label3.Caption = Val(Text1.Text) * Val(Text2.Text)

ElseIf Combo1.Text = "÷" Then

If Val(Text2.Text) = 0 Then

MsgBox "不能被零除！", 0, "计算器"

Label3.Caption = "": Text2.Text = ""

Else

Label3.Caption = Val(Text1.Text) / Val(Text2.Text)

End If

End If

End Sub

Private Sub Command2_Click( )

Text1.Text = "": Text2.Text = "": Label3.Caption = ""

End Sub

Private Sub Command3_Click( )

Unload Me

End

End Sub



简易计算器应用程序的运行结果也如图2-3所示。本例的窗体模块可按文件名“Fjsq.frm”存盘；工程文件可按“Pjsq.vbp”存盘。

2.1.2 对象

1．对象和类

对象是对现实中对象的模型化，是代码和数据的组合。最常用的对象可以说是窗体和控件。所有控件都必须添加在窗体上。

类是创建对象的样板，而对象只是类的一个实例。在集成开发环境工具箱中的每一个控件都是一个类。

2． 对象的属性

对象的属性用数据表示，表示对象的状态。

格式：对象名.属性=新值

设置属性：

设计时由属性窗口确定，常用于属性的初设置。

运行时赋值，常用于修改属性。

3． 对象的方法

对象的方法表示对象的行为，由类似内部函数的特殊程序实现（包括过程和函数）。

格式：对象名.方法 

2.1.3事件和事件过程

事件就是能被对象所识别的动作，相同的事件发生在不同的对象上所得到的反应是不一样的，这是因为“事件过程”不同。将对象对事件的反应称作“事件过程”。

1．事件过程的语法：

Sub 对象名_事件（）

处理事件的代码

End Sub

2．事件驱动的程序与面向过程的DOS程序不同点：

（1）只有事件发生时，程序才进行。

（2）在程序中流动的不是数据，而是事件。

（3）事件决定了对象之间联系的手段。

在VB程序设计中，基本的设计机制就是改变对象的属性、使用对象的方法和为对象编写事件过程。需要特别指出的是，用户不必为所有的事件都编写事件过程，只有当用户要以某种特定的方式响应某个事件时，才需要编写针对这个事件的事件过程。

3．一个典型的事件序列：

（1）启动应用程序，装载并显示窗体（产生Form_Load和Form_Show事件）。

（2）窗体或窗体上的控件接收事件。事件可由用户引发（例如鼠标或键盘操作），可由系统引发（例如定时器事件），也可由代码间接引发（例如装载其它窗体时产生的Load事件）。

（3）如果在相应的事件过程中存在代码，就执行代码。

（4）应用程序等待下一次事件。

为了加深理解事件过程，先运行一个不添加任何控件和不编写任何代码的示例。当点击工具栏的运行钮时，虽然未看到什么变化，但已执行了一次默认的Load事件操作。事件代码为：

Sub Form_Load( )

End Sub

本事件过程的实质是将窗体装入内存。若想看的清楚些，可将上述Load事件过程改为：

Sub Form_Load( )

Form1.Caption=“事件过程示例！”

End Sub

当看到运行结果，就可说明发生了Load事件。它不但将窗体装入内存，而且还对窗体的Caption属性赋予了新值。

2.2 编写VB的程序代码

2.2.1程序书写规则

1．注释

用“′”开头。

2． 断行

用“ _”（空格接一个下划线）。

例如：strTemp=“床前明月光，” & _

“疑是地上霜，” 

3． 将多行语句写在同一行上

VB通常是一行一句，若要写多句,可用“:”分隔开。

4． 使用不同进制的数字

默认为10进制，若用16进制，要加“&H”。

2.2.2变量和赋值语句

1． 变量命名规则

字母开头不能加“．”或空格及嵌入“！、#、@、$、%、&”；不能和关键词同，如：If、Loop、Abs等。

2． 声明一个变量

声明一个变量要使用Dim语句，一般的格式为:

Dim 变量名[As类型]

（1）模块级变量

每个窗体都有一个相关的包含代码的窗体模块，其文件扩展名为.FRM。每个窗体模块都包含事件过程，即代码部分。

如果同一个窗体的所有过程希望分享同一个变量，就应把它声明为模块级变量。其方法是在窗体模块的声明段中定义，除使用Dim关键字外，还可使用Public和Private。当使用“Public”时，使该变量在整个应用程序中有效；当使用“Private”时，该变量只在本窗体模块中有效。

（2）静态变量

在过程中声明的变量称为局部变量，这种变量只能在过程执行过程中有效，在其它过程无效。在过程执行结束后，变量的值不能被保留下来。如果希望在离开该过程之后，还能保持过程中局部变量的值，就要用关键字Static声明为静态变量。

3． 变量的作用范围

表2-1表明了变量作用范围的三个层次及其声明方式。

表2-1变量的作用范围和声明方式

作用范围 局部变量 模块级变量 公用变量 声明方式 Dim,Static Dim,Private Public 变量的声明位置 过程之中 模块的声明段中 模块的声明段中 能否被本模块中其它过程访问 × √ √ 能否被其它模块访问 × × √ 4． 变量的数据类型

（1） 用于保存数字的几种数字类型

Integer（整型）、Long（长整型）、Single（单精度浮点型）、Double（双精度浮点型）、Currency（货币型）和Byte（字节型）

（2） 逻辑判断的Boolean类型

True（真）或False（假）

（3） 字符串String类型

（4） 日期Date类型

（5） 保存对象引用的Object类型

（6） 万能的Variant（变体）类型

当没有显式指定变量类型时（省略 “As类型”）,VB默认这个变量的类型为Variant类型。Variant是一种特殊的数据类型，除了定长的String数据及用户定义类型外，可以包含任何类型的数据。另外，Variant类型的变量还可以包含Empty、Error及Null等特殊值。

5． 显式和隐式声明

VB允许不先声明变量。但为了避免写错变量名引起的麻烦，可采取“强制显式声明变量”的方法。其方法是在窗体模块、标准模块或类模块的声明段中加入一条语句：Option Explicit

在添加Option Explicit 语句后，VB将自动检查程序中是否有未定义的变量。只要遇到一个未经明确声明的变量，就发出错误警告。

用户只要在“工具”菜单中选取“选项”命令，然后单击“选项”对话框中的“编辑器”选项卡，再选中“要求变量声明” 选项，这样VB就会在任何新模块中自动插入Option Explicit语句。注意，此项操作不能在已经建立起来的模块中自动插入“Option Explicit”，这种情况下只能用手工键入该语句。

6. 赋值语句

使用赋值语句可以在程序中改变对象的属性和变量的值。

语法: 对象属性或变量=表达式

该语句的含义就是将等式右边的表达式的值赋给等式左边的变量或对象属性。

例: Form1.Width=400

temp= Form1.Width

temp=temp+30

Form1.Caption=”设计齿轮”

上述四句都是赋值语句。

2.2.3常数

VB有许多内部常数，也可以建立自定义常数。

语法：[Public│Private] Const常数名[As 类型]=表达式

参数：

1）“常数名”——有效的符号名（其规则与建立变量名的规则同）

2）“表达式”——由数值常数或字符串函数以及运算符组成，但“表达式”中不能使用函数调用

3）Const——可以表示数量、字符串或时间/日期。

例：Const conPi=3.1415926

Const conBirthday =#6/22/2005#

说明：

1） 等号“=”右边的表达式常常是数字或字符串，但也可以是其结果为数字或字符串的表达式，甚至可用先前定义过的常数定义新常数。

例：Const conPi2= conPi*2

2） 一行中可放置多个常数声明，用逗号分隔即可。

3）和变量声明一样，Const语句也有作用范围。若要创建在整个应用程序中有效的常数，应在标准模块的声明段中用“Public Const” 声明。

4）在窗体模块或类模块中不能声明Public常数。

2.2.4运算

1．算术运算 ＋、－、*、／、\、∧、MOD

2．比较运算 ＞、＜、≥、≤、＝、＜＞

3．连接运算 &——用于字符串连接。

＋——若皆为字符串，则按字符串连接；若其中一个为数字，则按相加操作。

4．逻辑运算

5．运算的优先顺序

从优先级高到优先级低依次为:

算术运算 逻辑运算

^ Not

-（取负） And

* 、 / Or

\ Xor

Mod Eqv

+、 - Imp

&

注：字符串连接符“&”不是算术运算符，但就其优先顺序而言，它在所有算术运算符之后，在所有比较运算符之前。

2.2.5流程控制语句

1．条件判定

（1）If…Then结构

型式1：If 条件Then语句（适用于只选择执行一条语句时）

型式2：If 条件Then

[语句系列]

End If

（2）if…Then…Else结构

语法： if 条件1 Then

[语句系列1]

[Else if条件2 Then

[语句系列2] ]

…

…

[Else

[语句系列n] ]

End If

（3）Select Case结构

语法 Select Case測試条件

[Case表达式1

[语句系列1]]

[Case表达式2

[语句系列2]]

…

…

[Case Else

[语句系列n]]

End Select

2. 循环控制

（1）For…Next结构

当知道循环次数时，最好用For…Next循环。

语法：For计数器=初值To终止值 [Stop增量]

语句序列

Next [计数器]

注：多用于数组或对象集合，它对数组或对象集合中的每一个元素重复一组语句，而不是重复语句一定的次数。如果事先不知道一个集合或数组有多少元素，For Each…Next循环非常有用。

语法：For Each 元素 In 数组

语句序列

Next [元素]

还需注意以下两点：

1） 对于数组而言，循环中用的“元素”变量必须是Variant类型的变量；

2） For Each…Next循环不能与自定义类型的数组一起使用。

（2）Do…Loop结构

当不知道循环次数时，最好用Do…Loop 循环。该结构有四种形式，但每种形式都会计算数值条件以决定是否继续循环。循环条件必须是一个数值或其值为True（非零）或False（零）的表达式。

《形式1》：Do While 循环条件

语句序列

Loop

执行这个Do循环时首先测试循环条件，如果循环条件为False或零，则跳过后面的循环语句序列。如果循环条件为True或非零，则进入循环体执行语句，执行语句后退回到Do While语句再测试循环条件。

《形式2》：Do 

语句序列

Loop While 循环条件

该形式保证循环中的语句至少执行一次。

《形式3》： Do Until 循环条件

语句序列

Loop

该形式的语句循环零次或多次。

《形式4》： Do 

语句序列

Loop Until 循环条件

该形式的语句至少循环一次。

注：后两种形式需要循环条件为False或零时才执行循环。

（3） 从循环中退出

《形式1》：For 计数器=初值To 终止值 [Stop增量]

[语句序列]

Exit For

[语句序列]

Next [计数器]

《形式2》：Do [{While │Until}循环条件]

[语句序列]

Exit Do

[语句序列]

Loop

2.2.6数组

1． 声明固定大小的数组

声明固定大小的数组有三种方法：

（1） 建立公用数组，在模块的声明段用Public语句声明数组。

（2） 建立模块级数组，在模块的声明段用Private语句声明数组。

（3） 建立局部数组，在过程中用Dim语句声明数组。

语法 {Dim│Private│Public}数组名（[下界To]上界）As数据类型

VB默认下界为0, 数组的上界必须大于或等于下界。

例：Dim Counters (1 to 14) as Integer ′14个元素的局部数组

Dim Sums(20) As Double ′21个元素的局部数组

2． 声明多维数组

语法： {Dim│Private│Public}数组名（[下界to]上界，[[下界to]上界],……） As 数据类型

例: Dim ThreeD(3,1 to 10,1to 15) ＇声明的三维数组，大小为4×10×15。

若不写“下界to”，数组的下界就是0 (缺省值为0)。

例：下面三个例句等效。

Dim A(8,3)

Dim A(0 to 8,0 to 3)

Dim A(8, 0 to 3)

3. 声明动态数组

要使用动态数组，需要经过两个步骤：

（1） 声明动态数组

在模块的声明段或在过程中，用Public、Private、Static、Dim语句声明数组，给数组赋予一个空维数表，如：

Dim DynArray( )

（2） 用ReDim语句分配实际的元素个数

该语句的目的是重新定义数组。如：

ReDim DynArray(x+1) ′假设x是整型变量

注意：ReDim语句只能出现在过程中，ReDim语句是一个可执行语句，应用程序运行时执行一个操作。

2.2.7使用过程和函数

使用过程编程有两大好处：

（1） 过程可使程序划分成离散的逻辑单元，便于调试。

（2） 一个程序中的过程，往往不必修改或稍作改动，便可移植到另一程序中。

可使用下面两种过程：

（1） Sub过程

又称为子过程，该种过程不返回值。

（2） Function过程

又称为函数，该种过程可以返回值。

在VB中，为了与事件过程相区分，将自定义的Sub子过程称为通用过程。通用过程必须由应用程序调用才能运行；而事件过程除响应用户引发的事件外（或由系统引发），总是处于空闲状态，应用程序无法直接调用。

1． 定义和调用通用过程

定义通用过程语法：

[Private│Public│Static] Sub 过程名 （参数列表）

语句序列

End Sub

按默认规定，模块中的子过程都为Public（公用的），这意味着应用程序可随处调用它们。

过程的“参数列表”类似于变量声明，其语法结构为：

[ByVal] 参数名称 As 数据类型

在应用程序中调用Sub过程有两种型式，如下例：

（1） Call BeepSound(5)

（2） BeepSound 5

注：当使用Call 时，参数必须在括号内；若省略Call 关键字，则必须省略参数两边的括号。

2． 定义和调用函数

VB包含了许多内部的函数，如Abs、Cos、Sqr等。还可用Function语句编写自己的函数过程。

语法：[Private│Public│Static] Function函数名 (参数列表) [As 数据类型]

语句序列

End Fuction

与子过程不同，Function过程可返回一个值到调用的过程。Sub过程与Function过程有三大区别：

（1） 调用方法不同。一般地说，在赋值语句右边的表达式中包含函数过程名和参数，这就调用了函数。

（2） 函数过程中的数据类型，决定了返回值的数据类型（如果没有As子句，默认的数据类型为Variant）。

（3） 给函数名自身赋一个值，就可返回这个值。Function过程返回一个值时，该值可成为表达式的一部分。

调用Function过程与调用任何内部函数的方法是一样的，即在表达式中填写它的名字和参数列表。还可像调用Sub过程那样调用函数，但VB将放弃返回值。

2.3 集成开发环境及工程的管理

2.3.1 集成开发环境

VB 6.0的集成开发环境如图2-1和图2-2所示，其组成元素为：

（1）菜单条

（2）工具栏

（3）工具箱

（4）代码编辑器窗口

应用程序的每个窗体或代码模块都有一个单独的代码编辑器窗口。

（5）窗体编辑器

作为自定义窗口，它用来设计应用程序的界面。应用程序中的每一个窗体都有自己的窗体编辑器窗口。若要精确设置窗体的大小，可以在窗体属性窗口中设置窗体对象的Width和Height属性的值。这两个属性的单位是“缇”。“缇”是一种与屏幕分辨率无关的计量单位。1440缇=1英寸（567缇=1厘米）。

（6）工程资源管理器窗口

列出当前工程中的窗体和模块。

（7） 属性窗口

由“对象列表框”“排序选项卡”和“属性列表”三部分组成，列出对选定窗体控件的属性设定值。

（8）窗体布局窗口

2.3.2工程的管理

1． 工程文件

工程文件就是与该工程有关的全部文件和对象的清单。这些文件和对象也可供其它工程共享。一个工程包括：

（1）跟踪所有部件的工程文件（.vbp）。

（2）窗体模块文件（.frm）——是大多数VB应用程序的基础。窗体模块可以包含事件过程、通用过程以及变量、常数、类型和外部过程的窗体级声明。

（3）窗体的二进制数据文件（.frx）——它含有窗体上控件的属性数据。对含有二进制属性（例如图片）的任何.frx文件都是自动产生的（不可编辑的）。

（4）类模块的文件（.cls）——类模块用来创建对象，这些对象可被应用程序内的过程调用。类模块与窗体模块类似，只是没有用户界面。该文件是可选项。

（5）标准模块的文件（.bas）——存放的是那些与特定的窗体或控件无关的代码。该文件是可选项。

（6）一个或多个包含ActiveX控件的文件（.ocx）——该文件是可选项。

（7）单个资源文件（.res）——该文件是可选项。

2． 创建新工程

可有三个操作途径：

（1）选择菜单“文件”|“建新工程”，然后从“建新工程”对话框中选“标准EXE”选项，单击“确定”按钮。

（2）单击“标准”工具栏的建新工程（左起第一个）按钮。

（3）单击“标准”工具栏的左起第一个按钮右侧的下拉按钮，再选“标准EXE”选项。

注：可使用“建新工程”对话框中的“VB应用程序向导”来创建一个应用程序。

3． 保存和打开工程文件

选择菜单“文件”|“保存工程”或单击工具栏中的保存工程按钮，可以保存当前工程。集成开发环境会自动保存当前工程中的所有文件。

如果是第一次保存工程，或选择“工程另存为”命令，集成环境会提示用户输入一个文件名，并且提示用户保存所有修改过的窗体或模块。

要打开一个已有的工程，可选择菜单“文件”|“打开工程” 或单击工具栏中的打开工程按钮。在打开一个工程之前，集成开发环境会关闭当前工程，并提示保存所有变动。

4． 启动窗体

在默认情况下，应用程序的第一个窗体为启动窗体。应用程序开始运行时，此窗体就会显示出来。如果想显示另外的窗体，就必须改变启动窗体的设置。设置启动窗体步骤：

（1）从菜单“工程”中选取“工程属性...”命令。

（2）在显示的工程属性对话框中选取“通用”选项卡。

（3）选取要做为新启动窗体的窗体，如选取“Form1”。

（4）单击“确定”按钮。

以上操作如图2-4所示。


图2-4工程属性对话框

若由于某种原因系统自动弹出上述对话框时，也要求操作者确认“启动对象”列表框中的窗体名。运行正常后注意要重新存盘。

5． 从工程中删除文件和添加文件

当从工程中删除文件时，应先在“工程资源管理器”中选取某窗体或其它模块，再单击菜单“工程”中点击“删除（文件名）“选项。

当向工程中添加文件时，应单击菜单“工程”中相应要添加的选项，再依照“添加文件”对话框的提示操作即可。

6．生成可执行文件

从菜单“文件”中选取“生成（工程名）.exe．．．”项，出现“生成工程”对话框，在填入工程文件名。当有必要时输入“选项（O）”中的信息，最后单击“生成工程”对话框的“确定”按钮就可生成该可执行文件。

注：使用Shell函数运行可执行文件

Shell函数语法: Shell文件名[, 窗口样式]

例: Private Sub Command1_Click

Shell“e:\应用程序\Lchi-1.exe”,vbNormalFocus

End Sub

2.4 设计应用程序的界面

2.4.1设计窗体

窗体有自身的属性、方法和事件。设计窗体的第一步是设置它的属性。在窗体的设计模式状态，是设计其“属性”窗口中的值；而在运行模式状态，要由代码来实现。

窗体的属性很多，下面仅列出常用的属性并简述其用途和用法：

（1）BorderStyle：控制窗体如何调整大小。例如，若将窗体设置为不可调整大小，要将其设为1。

（2）Caption：确定窗体标题栏中显示的文本。

（3）Height 和Width：确定窗体初始的大小。

（4）Left 和Top：根据屏幕的左上角确定窗体的位置。

（5）MaxButton 和MinButton：确定窗体是否能最大化或最小化。

（6）Moveable：指定对象是否可移动。

（7）Name：设置窗体的名称。在工程中添加窗体时，窗体的默认名称为Form1、Form2、...

（8）WindowState：设定窗体在启动时为最大化、最小化或正常大小。

（9）Icon：设置窗体最小化时显示的图标。

2.4.2 在窗体上添加控件

基本的控件可分为内部控件、ActiveX控件及可插入的对象三类。

1．工具箱中常用的基本控件

（1）命令按钮（CommandButton）

（2）标签（Label）

（3）文本框（TextBox）

（4）复选框（CheckBox）和单选框（OptionButton）

（5）列表框（ListBox）和组合框（ComboBox）

（6）分组框（Frame）

（7）图片框（PictureBox）

（8）图象（Image）

2．使用ActiveX控件

ActiveX控件是VB工具箱的扩充部分。

加入ActiveX控件的方法： 选择“工程”菜单的“部件．．．”命令（或在工具箱中单击鼠标右键）出现“部件”对话框，选定所需控件名称左边的复选框，然后点击“确定”即可。

若要将其它ActiveX控件加入“部件”对话框，可以单击“部件”对话框中的“浏览”按钮，找到扩展名为.ocx的文件。

3． 直接使用控件名来引用属性

所有控件都有一个默认的属性，这是控件最重要也是最常用的属性，称这个属性为控件的值。最常用控件和控件的值如表2-2所示。


表2-2 常用控件及其默认的属性

控件 默认的属性（控件的值） CheckBox Value ComboBox Text Label Caption ListBox Text TextBox Text 当控件的属性为控件的值时，就不必在代码中指定其属性名。

例：由于TextBox控件的默认属性为“Text”，则以下两句等效。

txtPoem=“问君能有几多愁”

txtPoem.Text=“问君能有几多愁”

4．常用的显示方式：

（1）利用标签Label的Caption属性。

（2）利用文本框TextBox等的Text属性。

（3）使用窗体或图片框的Print属性。

语法：[Object.]Print[outputlist][{;│,}]

其中：Object——可选项，如省略则用于当前窗体；

outputlist——显示在窗体或图片框中之文本，该项可包括属性、常数和变量（字符串或数字）。

5. 窗体上的焦点

当对象具有焦点时，可接收用户的输入。在有几个控件的窗体中，只有具有焦点的控件才可以接受用户鼠标或键盘的输入。当对象得到或失去焦点时，会产生GotFocus或LostFocus事件。使用以下方法可以将焦点赋予对象：

（1）运行时选择对象。

（2）运行时用快捷键选择对象。

（3）在代码中用SetFocus方法。

只有对象的Enabled和Visible属性为True时，才能接受焦点。Enabled属性允许对象响应由用户产生的事件。Visible属性决定对象在屏幕上是否可见。

2.4.3消息框和输入框

1． MsgBox函数可以用来在对话框中显示消息。

语法：MsgBox(prompt[, buttons][, title][, helpfile, context] )

参数：

（1） prompt——必要。字符串表达式，显示消息。

（2） buttons——可选。消息框buttons参数的设置值如表2-3所示。数值表达式是值的总和，指定显示按钮的数目和形式、使用的图标样式、默认按钮是什么以及消息框的强制回应等。如果省略，则其默认值为零，只显示OK按钮。

表2-3消息框buttons参数的设置值

符号常量 值 含义 VbOKonly 0 只显示OK按钮 VbOKCancel 1 显示OK和Cancel按钮 VbAbortRetryIgnore 2 显示Abort、Retry和Ignore按钮 VbYesNoCancel 3 显示Yes 、No和Cancel按钮 VbYesNo 4 显示Yes和No按钮 VbRetryCancel 5 显示Retry和Cancel按钮 （3）title——可选。在对话框标题栏中显示字符串表达式。

（4）helpfile——可选。字符串表达式。

（5）context——可选。数值表达式。

除了显示消息，还可以选择点击哪一个按钮来控制MsgBox函数的返回值。MsgBox函数的返回值如表2-4所述。

表2-4 MsgBox函数的返回值

符号常量 值 含义 vbOk 1 选择Ok按钮 vbCancel 2 选择Cancel按钮 vbAbort 3 选择Abort按钮 vbRetry 4 选择Retry按钮 vbIgnore 5 选择Ignore按钮 vbYes 6 选择Yes按钮 vbNo 7 选择No按钮 2.输入框

可在本对话框中显示提示，等待用户输入正文或点击按钮，并返回包含文本框内容的String。

语法：

InputBox(prompt[, title][, default][, xpos][, ypos][, helpfile][, context])

参数：

（1）prompt——必要。字符串表达式，显示消息。

（2）title——可选。显示标题栏中的字符串表达式。

（3）default——可选。显示文本框中的字符串表达式。在没有其它输入时作为默认值。若省略default，则文本框为空。

（4）xpos——可选。数值表达式。指定对话框的上边与屏幕上边的距离。如省略则对话框居中。

（5）ypos——可选。数值表达式。与xpos成对出现。

（6）helpfile——可选。字符串表达式。

（7）context——可选。数值表达式。

如果用户单击“确定”按钮或按下“回车”键，则InputBox函数返回文本框中的内容；如果用户单击“取消”钮，则此函数返回一个长度为零的字符串（“”）。

2.4.4窗体的生命周期

1． 窗体的创建

Initialize事件是窗体的创建状态开始的标志。因此，放在Form_ Initialize事件过程中的代码，就是窗体创建时最先执行的代码。处于这种状态，还没有窗口，它的控件也不存在（窗体的可视部分尚未调入）。

2． 加载窗体

Load事件标志加载状态的开始。Form_Load事件过程中的代码开始执行。Form_Load事件过程开始后，窗体上所有的控件都被创建和加载，而且有了一个窗口。

窗体如果满足以下条件之一就会被自动加载：

（1）该窗体在“工程属性”对话框的“通用”选项卡中指定为启动对象。

（2）窗体中首先被调用的属性或方法是Show方法。例如： Form1.Show。

（3）首先被调用的窗体属性或方法是窗体内置的成员（因为每个控件定义了窗体的一个属性）。

（4）用Load语句加载窗体。

注: 对于后两种情况, 窗体将保持加载状态，但不显示。

3． 可见状态

使用窗体的Show方法，可以使窗体进入可见状态。

注：若要隐藏，应调用Hide方法，它的Visible属性将被设置为False，窗体返回到加载状态。

4． 窗体的卸载

窗体卸载前最后发生Unload事件。当卸载完毕，内存和资源完全收回。

5． 结束应用程序

当所有窗体都已关闭并且没有代码执行时，事件驱动的应用程序就停止运行。如果还有隐藏窗体存在，实际上应用程序将仍在继续进行。这是因为对已卸载窗体的属性或控件的任何访问，都将导致隐含地（不予显示地）加载那个窗体。为确保所有的窗体都已卸载，有以下方法：

（1）当只有一个窗体时，用Unload Me。

（2）有一个以上窗体时，可用Forms集合和Unload语句。

Private Sub Form_Unload( )

Dim I As Integer

For I=0 to Forms.Count-1 

′在Forms集合中循环并卸载每一个窗体

Unload Forms(I)

Next

End Sub

（3）End 语句。

End 语句使应用程序立即结束，End 语句之后代码不会执行，也不会再有事件发生。对象的各个引用将被释放。

（8） 开发环境中单击“结束”（工具栏的■）。

2.5 处理应用程序的错误

2.5.1应用程序中的错误

1． 编译错误

编译错误是用户在开发应用程序期间由于不正确构造代码而产生的，大部分是语法错误。如果已在“选项”对话框的“编辑器”选项卡选定“自动语法检测”选项，只要在“代码”窗口中出现一个语法错误，就会被捕捉。

2． 运行时错误

应用程序在运行期间，当一个语句试图执行一个不能执行的操作时，就会发生运行时错误。比如，某值被零除就属于这类的错误。次类错误比较难查找，用户必须预测可能会发生哪些问题，然后设置陷阱，编写代码加以处理。

3． 逻辑错误

应用程序可能没有编译错误和运行时错误，但仍未按预期方式执行时，这是发生了逻辑错误。这类错误的排除需要用户对程序代码甚至整个应用程序进行分析。

2.5.2捕获运行时错误

在 VB 中，使用 On Error 语句来捕获运行时错误。此语句为系统设置精确的捕获。当错误发生时，此语句会自动转到专门编写的错误处理程序，系统的默认错误处理则被略过。 

On Error 语句有三种形式： 

（1）On Error Resume Next

当希望忽略错误时，可以使用 On Error Resume Next 语句。此语句捕获错误但不会显示错误信息或者终止程序，而是移到下一行代码继续执行。

（2）On Error GoTo Label 

当希望编写明确的处理程序时，可以使用 On Error GoTo Label 语句。此语句捕获错误但不会显示错误信息或者终止程序，而是跳到代码中的特定位置。然后可以使用适当的方式来响应错误。

（3）On Error GoTo 0

该语句可停用当前的错误处理程序。

2.5.3响应捕获的错误

捕获到错误后，就应该对错误进行处理。 VB 提供了关于使用 Err 对象所捕获到的错误类型信息。此对象具有以下属性：Number、Description、Source、HelpFile、HelpContext 和 LastDLLError。其中最重要的是 Number 和 Description 属性。Number 属性包含与错误关联的唯一错误代码，而 Description 属性则包含通常显示的错误信息。在错误处理程序中，用户可以将错误的 Number 属性与期望值相比较，这样可帮助用户确定发生的错误的性质。借助Err对象与Select …Case 或If …Then… Else语句的组合，可对出现的错误采取针对性的操作。

注：关于响应AutoCAD用户的输入错误，可以使用输入函数的InitializeUserInput方法，该方法可以提供特别的用户输入控制，但最好仍需通过错误陷阱来确认，然后进行处理。

第3章 AutoCAD ActiveX对象

3.1 使用对象变量

3.1.1对象变量

变量除了可以存储值以外，还可以引用对象，引用就是将对象指定给变量。每个AutoCAD ActiveX 对象均精确代表一个 AutoCAD 组件。

1. 声明对象变量

语法: {Dim│ ReDim│Static│Private│Public} 变量名As [New] 类名

对象变量可分为一般的对象类型和特定的对象类型。AutoCAD R14.01版以后支持早期绑定，则可用其特定的类型声明，也可按一般的对象类型Object声明。

2. 用Set语句将对象赋予变量

语法: Set变量=对象

3. 释放对对象的引用

当不再使用某对象时，应及时释放这些资源。

语法：Set对象=Nothing

注意，当从属对象使用结束后，也要将其设为Nothing。

3.1.2对象的属性和方法

1. 属性的设置和查询

（1）属性的设置

语法: 对象表达式.属性=设置值

例：LineObj.Color=acRed ’设置LineObj为红色

（2）属性的查询

语法: 返回值=对象表达式.属性

例: Dim WhatColor As Integer

WhatColor= LineObj.Color

2．使用方法

方法的语法格式取决于该方法要求的参数及是否要求有返回值。

如果方法不要求参数，则用以下格式：

语法：对象表达式.方法

例：用Save方法保存当前文件。

ThisDrawing.Save

如果方法要求有参数，就在后面加空格再写入参数；若方法要求有多个参数，要用逗号将各参数隔开。

如果要用对象变量引用上述对象，就必须用括号把该方法需要的参数括起来。

例：用AddCircle方法在模型空间创建一个圆。

（其中的参数：圆心Center和半径Radium已经声明和赋值。）

ThisDrawing.ModelSpace.AddCircle Center,Radium

如果要用按特定类声明的对象变量引用上述对象，上例应写为：

Dim CircleObj As AcadCircle

Set CircleObj= ThisDrawing.ModelSpace.AddCircle (Center,Radium)

3.1.3使用事件

1. AutoCAD中的事件

对于窗体和控件等界面对象，大多有相同的事件，如单击鼠标（Click）、双击鼠标（Dbclick）等。用户可以针对不同的事件，编写相应的处理程序。AutoCAD的事件有三个级别：

（1）应用程序层事件

当AutoCAD应用程序或运行环境发生变化时，便触发应用程序层事件。这些事件包括：与文件操作有关的事件（新建、打开、保存、关闭和打印文件等）；与发出AutoCAD命令有关的事件；与加载和卸载ARX或LISP应用程序有关的事件；以及与改变系统变量、改变应用程序窗口有关的事件等。例如，打开图形时会触发 BeginOpen 事件，该事件中包含打开的 AutoCAD 图形的名称。而关闭图形时则会触发另一个事件。

（2）文档层事件

当改变特定的图形文件及其内容时，便触发文档层事件。这类事件包括：对象的添加、修改或删除；激活快捷菜单；改变选择集；改变图形窗口；图形的重新生成；以及从图形加载或卸载 ARX 和 LISP 应用程序等。

（3）对象层事件

当某一个图形对象被修改时，便触发对象层事件。AutoCAD只有一个对象层事件Modified。

2.事件处理程序的特点

响应事件的程序代码称作事件处理程序。当触发一个事件时，会自动执行这些程序，与事件有关的AutoCAD的状态或行为发生变化的信息作为参数由系统传递给事件过程。用户可以利用这些参数或不利用这些参数。注意，并不是所有的事件都传递参数。虽然事件处理程序可以响应事件，但触发事件处理程序时，AutoCAD 通常都是正处在处理命令的执行过程中。因此，如果要使事件处理程序在涉及 AutoCAD 和它的数据库时能安全运行，就要使事件处理程序符合一定的规范，主要的注意事项是：

（1）不要过分依赖事件发生的顺序。

（2）不要过分依赖操作的顺序。

（3）不要在事件处理程序中使用交互式函数。

（4）不要在事件处理程序中启动对话框。

（5）除了发出事件的对象外，可以向数据库中的任何对象写入数据。

（6）在事件处理程序中不能有可能会触发相同事件的操作。

（7）当 AutoCAD 显示模式对话框时，不会触发任何事件。

3.2 AutoCAD 的对象模型

对象按照分层结构来组织，这种层次结构的图称为“对象模型”。对象模型给出了上级对象与下级对象之间的访问关系。AutoCAD 2000/2002与AutoCAD 2004/2005/2006的类型库互不兼容。AutoCAD 2006的对象模型如图3-1所示。

图3-1 AutoCAD 2006的对象模型

3.2.1 Application 对象和Document对象

1．Application 对象

Application是对象模型的基础，是所有对象的“根”，称为“根对象”。 通过 Application 对象，用户可以访问任何其他的对象或任何对象指定的属性或方法。Application 对象还是 ActiveX 接口的全局对象，也就是AutoCAD应用程序自己，Application 对象的所有方法和属性都可以用于全局名称空间。

Application 对象通过Documents集合访问AutoCAD图形；通过MenuBar集合访问AutoCAD菜单；通过MenuGroups集合访问AutoCAD工具条；通过Preferences 对象可以访问存放在注册表中的AutoCAD“选项”对话框中的设置值。

2．Document对象

属于Documents集合的Document对象就是AutoCAD图形。它提供对所有图形对象和大多数非图形对象的访问。

Document对象之下有一个DatabasePreferences 对象，它包含存储在图形中的所有选项。这个对象指定当前AutoCAD 图形特有的设置值，而无须事先启动 AutoCAD 应用程序（ObjectDBXTM 应用程序）。

此外，Document 对象还提供对 Plot 和 Utility 对象的访问。

3.2.2 Preferences对象

在 Preferences 对象下有一组对象，分别对应“选项”对话框中的一个选项卡。这些对象提供对“选项”对话框中所有存储在注册表中的设置的访问。用户也可以利用SetVariable 和GetVariable系统变量设定和修改。

注：关于设置选项的详细信息，请参见 <4.7设置 AutoCAD 系统配置>一节。 

3.2.3图形对象和非图形对象

1． 图形对象

图形对象（也称“图元”） 是组成图形的可见对象。它构成AutoCAD图形的各种几何实体，如Line、Circle、Text及Table等。要创建图形对象需使用相应的Add方法，而编辑这些对象则应使用对象本身的方法和属性。例如应用Color属性改变该图形对象的颜色。

2．非图形对象

非图形对象包含了图形中不可见的数据信息。如Layer、LineType、DimStyle等。创建这些对象可使用各自集合的Add 方法，要修改或查询这些对象，要使用对象本身的方法或属性。每一个非图形对象都有设置和检索扩展数据以及删除自身的方法。

3.2.4 Collection对象

AutoCAD 将大部分对象进行编组，组成集合。虽然这些集合包含了不同类型的数据，但是可以使用相似的技术来处理它们。每一个集合都提供了向其中添加对象的方法，大多数集合使用的是 Add 方法。

集合还有一些其他的常用方法和属性。 Count 属性用于获取集合中的对象个数（从零开始）， Item 方法用于访问集合中的任何对象。

集合对象有:

（1）Documents：其成员为AutoCAD当前任务中打开的全部图形文件。

（2）ModelSpace: 包括模型空间中所有的图形对象。

（3）PaperSpace：包括图纸空间所有图形对象。

（4）Blocks：包括图形文件中定义的所有块、外部引用和布局对象。

（5）Block：包括构成这一个块的所有图形对象。

（6）LineTypes：包含图形中的所有线型。

（7）DimStyles：包含图形中的所有标注样式。

（8）Layers：包含图形中的所有图层。

（9）TextStyles：包含图形中的所有文字样式。

（10）Dictionaries ：包含图形中的所有词典。

（11）Groups ：包含图形中的所有编组。

（12）Hyperlinks ：包含给定图元的所有超链接。

（13）Layouts ：包含图形中的所有布局。

（14）MenuBar ：包含 AutoCAD 中当前显示的所有菜单。

（15）MenuGroups ：包含 AutoCAD 当前加载的所有菜单和工具栏。

（16）RegisteredApplications：包含图形中所有注册的应用程序。

（17）SelectionSets：包含图形中的所有选择集。

（18）UCSs：包含图形中的所有用户坐标系。

（19）Views 集合：包含图形中的所有视图。

（20）Viewports：包含图形中的所有视口。

3.2.5 Plot 对象和 Utility 对象 

Plot 对象提供对“打印”对话框中设置的访问，并为应用过程提供了打印图形的各种方法。 

Utility 对象提供用户输入和转换函数。用户输入函数是一些方法，用来提示用户在 AutoCAD 命令行中输入各种类型的数据，例如字符串、整数、实数、点等。转换函数是操作 AutoCAD 特有的数据类型（例如点和角度）以及处理字符串和数值的方法。 

3.3 根据对象模型访问对象

3.3.1访问AutoCAD对象时的切入点

1． 从外部访问

如利用VB编写的外部客户程序访问，其切入点必须是根对象。即必须用VB本身的语句或函数，先“引用”应用程序对象（详细内容参见<3.4创建对AutoCAD 部件对象的引用>一节），然后再按照对象模型从上级到下级逐级进行访问。除根对象外，都有一个永久性连接的上一级对象——父对象（只能有一个父对象）。要引用某对象，只能通过它的唯一父对象依次引用之。在对象模型结构中，所有的对象（最末梢的对象）都有一个名为Application的属性，通过它，可直接回到根对象。

例：访问打开的当前文档内模型空间的图形对象——圆。（假设该圆是第一个实体）

Dim MyCircle As AcadCircle

Set MyCircle=AutoCAD.Application.ActiveDocument. _

ModelSpace.Item(0)

2． 从AutoCAD内部用VBA开发的应用程序访问

VBA提供了一个特殊对象ThisDrawing。当用于全局工程时，它代表AutoCAD 中的当前（活动）图形；用于内嵌工程时，它代表包含该工程的图形。对于上例，可写为：

Dim MyCircle As AcadCircle

Set MyCircle=ThisDrawing. ModelSpace.Item(0) 

当需要访问在对象层次结构中位于 Document 对象之上的Application对象或不在Document对象之下的对象时，要使用Document对象的Application属性，先返回Application对象。例如若要访问Preferences对象，可按以下语句：

Dim MyPreferences As AcadPreferences

Set MyPreferences =ThisDrawing.Application.Preferences

3.3.2定义中间变量引用常用对象

设有语句:

Dim LineObj As AcadLine

Set LineObj=AutoCAD.Application.ActiveDocument. _

ModelSpace.AddLine(startP,endP)

若设置中间变量：

Dim moSpace As AcadModelSpace

Set moSpace= AutoCAD.Application.ActiveDocument.ModelSpace

则有: 

Set LineObj = moSpace. AddLine(startP,endP)

3.4 创建对AutoCAD 部件对象的引用

要利用VB编写的外部客户程序操纵AutoCAD，首先要创建对AutoCAD部件对象的引用。

要创建对AutoCAD 部件对象的引用，一般需要以下三个步骤：

（1）引用AutoCAD类型库。

（2）声明对象变量。

（3）将对象引用赋予变量。

3.4.1AutoCAD类型库

在使用应用程序提供的ActiveX 对象之前，必须先引用（References）其类型库。AutoCAD 2000/2002 的类型库文件名为ACAD.TLB，AutoCAD 2004/2005/2006 的类型库文件名为acax16enu.tlb。默认情况下该类型库位于 c:\program files\common files\autodesk shared 目录下（只要在操作系统中安装了AutoCAD，就可在VB的菜单“工程”中选择“引用...”项）。

现以AutoCAD 2004为例，说明“引用AutoCAD类型库”的步骤：

（1）从菜单“工程”中选择“引用...”命令。

（2）在“引用”对话框中，选择要添加的类型库“AutoCAD2004 Type Library”。然后单击“确定”钮，即添加了对“AutoCAD2004 Type Library”的引用。如图3-2所示。

图3-2 VB的“引用”对话框

在“引用”（References）对话框中添加了引用的ActiveX部件后，就可在菜单“查看”（View）中选择“对象浏览器”（Object Browser）查看它的类型库。可从“工程/库”下拉列表中查到“AutoCAD”，并在相应的“类”(Classes)列表中查找所需要的类名及其成员。如图3-3所示。

图3-3在对象浏览器中查看类名及其成员

3.4.2声明对象变量

添加部件类型库的引用后，即可声明对象变量。

语法: Dim 变量名 As [New] 类名 （Class 参数）

Class 参数由两部分组成，其格式为：Component.Class

其中：

“Component”——提供ActiveX对象的部件名，对应到“对象浏览器”中，就是“对象/库”列表中显示的内容，即“AutoCAD”；

“Class”——（对象的）“类名”，即“类”列表中显示的内容”AcadApplication”。

例：Dim AcadApp As AutoCAD.AcadApplication

注：也可写为 Dim AcadApp As AcadApplication

如使用关键字“New”声明一个对象变量，VB将自动创建一个对象，并在第一次使用该变量时赋值对象引用（省略Set）。

3.4.3将对象引用赋予变量

一般有三种方式赋值一个新的AutoCAD部件对象引用：

1．用New关键字赋值对象引用

如果ActiveX部件提供一个类型库，就可以在变量声明或Set语句中用New关键字创建一个新对象，并将对象引用赋予对象变量。

例: Set AcadApp=New AutoCAD.AcadApplication

使用New关键字仅适用于特定类型的新对象引用。

2． 用CreateObject 函数赋值对象引用

用该函数创建一个新对象，并将对象引用赋予一个对象变量（想要访问的对象必须是外部可创建的）。

语法：Set对象变量= CreateObject(“progID”)

通常，progID参数是正在创建对象的“完全限定类名”。

例：Set AcadApp = CreateObject (“AutoCAD.Application”)

3． 用GetObject函数赋值对象引用

一般用于赋值对已经存在的对象的引用。

语法：Set对象变量= GetObject([pathname][,progID])

参数pathname可以是现存文件的路径，也可完全省略或为空字符串。

例：Set AcadApp = GetObject(,“AutoCAD.Application”)

若运行该语句时AutoCAD尚未启动，上句程序将返回一个错误。

若pathname为空字符串，则功能与CreateObject函数一样。

例：Set AcadApp = GetObject(“”,“AutoCAD.Application”)

注：

1）当对AutoCAD引用后，要使用Application对象的Visible属性让应用程序窗口可见。

2）当对AutoCAD引用后，要将文档对象变量设置为 AutoCAD 应用程序中的 Document 对象。Document 对象由 Application 对象的 ActiveDocument 属性返回。 

例：Dim acadDoc as AcadDocument

Set acadDoc = acadApp.ActiveDocument

自此就可使用变量acadDoc引用当前的 AutoCAD 图形。 

注：

1）自AutoCAD 2004版起， 对 VBA/VB所使用的 ActiveX Automation接口进行了更改。AutoCAD 2004的COM应用程序访问的注册表项为"AutoCAD.Application.16"。若要将使用AutoCAD 2000/2002版开发的COM应用程序(.vbp)适用于 AutoCAD 2004版，应按以下的步骤进行修改:

（1）在 VB中打开工程文件，然后选菜单“工程”|“引用...”。 

（2）在“引用”对话框中，删除对名为 acad.tlb 的类型库（其默认位置是 c:\program files\AutoCAD2002）的引用；改添加对名为 acax16enu.tlb 的类型库（其默认位置是 c:\program files\common files\autodesk shared）的引用，然后单击“确定”。 

（3）如果程序中 CreateObject 或 GetObject 函数使用的是独立于版本的 ProgID，则将函数更改为使用依赖于版本的 ProgID。 例如，应将CreateObject("AutoCAD.Application") 替换为 :

CreateObject("AutoCAD.Application.16")。 

此外，如果 GetInterfaceObject 方法使用的是独立于版本的 ProgID，则必须将该方法更改为使用依赖于版本的 ProgID。 

（4）保存工程文件并重新编译。

2）自AutoCAD 2005版起，增加了接口对象的新功能。如果 ActiveX自动化工程使用了以前版本中没有的 AutoCAD 2005 新功能，则需要明确声明工程中正在使用 AutoCAD 2005 扩展接口。 以下接口包含 AutoCAD 2005 中新的属性或方法： 

●IAcadBlock2

●IAcadLayer2

●IAcadLayers2

●IAcadModelSpace2

●IAcadMText2

●IAcadPaperSpace2

●IAcadPreferencesFiles2

●IAcadPViewport2

●IAcadText2

●IAcadView2

3) AutoCAD 2007版对扩展接口进行了更改，以下 COM 扩展接口已被合并到下面指明的主接口中： 

旧接口 合并到 IAcadAttribute2 IAcadAttribute IAcadAttributeReference2 IAcadAttributeReference IAcadBlock2、IAcadBlock3 IAcadBlock IAcadBlockReference2 IAcadBlockReference IAcadExternalReference2 IAcadExternalReference IAcadHatch2 IAcadHatch IAcadLayer2 IAcadLayer IAcadLayers2 IAcadLayers IAcadMInsertBlock2 IAcadMInsertBlock IAcadMLine2 IAcadMLine IAcadModelSpace2、IAcadModelSpace3 IAcadModelSpace IAcadMText2 IAcadMText IAcadPaperSpace2、IAcadPaperSpace3 IAcadPaperSpace IAcadPreferencesFiles2 IAcadPreferencesFiles IAcadPreferencesOutput2 IAcadPreferencesOutput IAcadPViewport2 IAcadPViewport IAcadTable2 IAcadTable IAcadText2 IAcadText IAcadUtility2 IAcadUtility IAcadView2 IAcadView 3.5 使用变体传递数组

AutoCAD ActiveX Automation使用变体类型数据Variant来传递数组。这表示数组必须是变体，才能被 AutoCAD ActiveX Automation的方法和属性接受。同样，从 AutoCAD ActiveX Automation输出的数组数据也必须是以变体类型来处理。特别需要注意的是，VBA/VB的输入数组会自动被转换为变体。也就是说，应用 VBA/VB编程，遇到使用 ActiveX Automation的方法和属性时，不必一定要输入变体类数据的数组。但所有的输出数组都将会是变体的形式，所以需要进行相应的处理。

3.5.1创建变体类型的数组

在编程过程中，有时需要将数据转换成变体类型。AutoCAD ActiveX Automation提供了将数组转换为变体类型数据的实用方法。应用 CreateTypedArray 方法，可以创建包含整数数组、浮点数数组、双精度数组等的变体。只要方法或属性接受变体数组，用户就可以将产生的变体传递给任何 AutoCAD 的方法或属性。 

CreateTypedArray 方法需要输入数组中值的类型，将数组数据进行转换，以变体返回数组。 

例3-1：使用 CreateTypedArray 方法创建样条曲线 。 

以下代码使用 CreateTypedArray 转换的三个数组：样条曲线拟合点的坐标矩阵及其起始点和结束点相切点的矩阵。然后将变体传递给 AddSpline 方法，创建样条曲线。 

Sub Ch3_CreateSplineUsingTypedArray()

'本例使用 CreateTypedArray 方法

'在模型空间中创建样条曲线对象

Dim splineObj As AcadSpline

Dim startTan As Variant

Dim endTan As Variant

Dim fitPoints As Variant

Dim utilObj As Object '后期绑定 Utility 对象

Set utilObj = ThisDrawing.Utility

'定义 Spline 对象

utilObj.CreateTypedArray startTan, vbDouble, 0.5, 0.5, 0

utilObj.CreateTypedArray endTan, vbDouble, 0.5, 0.5, 0

utilObj.CreateTypedArray _

fitPoints, vbDouble, 0, 0, 0, 5, 5, 0, 10, 0, 0

Set splineObj = ThisDrawing.ModelSpace.AddSpline _

(fitPoints, startTan, endTan)

'放大新创建的样条曲线

ZoomAll

End Sub

3.5.2使用变体类型的数组

从 AutoCAD ActiveX Automation传回的数组信息是变体类型。如果知道数组的数据类型，就可以使用数组方式存取变体数据。如果不知道变体中包含的数据类型，可以使用 VB/VBA 的函数 VarType 或 Typename，这些函数的作用是返回变体中数据的类型。

例3-2：计算两点之间的距离。 

在本例中，因为所有的坐标值均为双精度数，即数据类型是已知的。三维坐标值是一个包含三个元素的双精度浮点数组，而二维坐标值是一个包含两个元素的双精度浮点数组。 

Sub Ch3_CalculateDistance( )

Dim point1 As Variant

Dim point2 As Variant

'获取用户输入的点

point1 = ThisDrawing.Utility.GetPoint (, vbCrLf & "First point: ")

point2 = ThisDrawing.Utility.GetPoint _

(point1, vbCrLf & "Second point: ")

'计算 point1 和 point2 之间的距离

Dim x As Double, y As Double, z As Double

Dim dist As Double

x = point1(0) - point2(0)

y = point1(1) - point2(1)

z = point1(2) - point2(2)

dist = Sqr((Sqr((x ^ 2) + (y ^ 2)) ^ 2) + (z ^ 2))

'显示计算出来的距离

MsgBox "两点间距离为: " & dist, , "计算两点间距离例"

End Sub

3.6 创建对AutoCAD部件引用的综合示例

3.6.1创建对AutoCAD部件引用综合示例的用户界面

综合示例的窗体如图3-4所示。

图3-4创建对AutoCAD部件引用综合示例的窗体

3.6.2创建对AutoCAD部件引用综合示例的源程序

该综合示例的源程序全部收录在本书附带的光盘中，下面仅介绍其中的一部分。

1．利用New创建对AutoCAD部件的引用

例3-3：利用New创建对AutoCAD部件的引用，设定AutoCAD窗口的状态及使其可见。

Sub Ch3_ConnectToAcad1( )

Dim AcadApp As AutoCAD.AcadApplication

Set AcadApp = New AutoCAD.AcadApplication

AcadApp.WindowState = acMax

AcadApp.Visible = True

End Sub

2． 利用GetObject和 CreateObject函数创建对AutoCAD部件的引用

例3-4: 利用GetObject和 CreateObject函数创建对AutoCAD部件的引用，设定AutoCAD窗口的位置和大小及使其可见。

Sub Ch3_ConnectToAcad2( )

Dim AcadApp As AutoCAD.AcadApplication

On Error Resume Next

Set AcadApp = GetObject(,"AutoCAD.Application.16")

If Err Then

Err.Clear

Set AcadApp = CreateObject("AutoCAD.Application.16")

If Err Then

MsgBox Err.Description

Exit Sub

End If

End If

MsgBox "现在运行: " + AcadApp.Name + " Version " + AcadApp.Version

AcadApp.WindowTop = 0

AcadApp.WindowLeft = 500

AcadApp.Width = 500

AcadApp.Height = 700

AcadApp.Visible = True

End Sub

注：上述源代码仅适用于AutoCAD2004/05/06版。

AutoCAD ActiveX VBA二次开发技术基础及应用实例-4

第4章 控制AutoCAD环境和组织图形元素

要控制AutoCAD环境，一般包括新建、打开、保存和关闭图形；设置 AutoCAD 系统配置；控制应用程序窗口；控制图形窗口；重置当前对象；设置和返回系统变量；精确绘图；提示用户输入；访问 AutoCAD 命令行；在没有文档打开时使用；输入其他文件格式和输出到其他文件格式等诸多内容，甚至也可将自定义菜单或工具栏属于这一范畴。本章只选择与开发参数化绘图关系密切的部分进行重点介绍。

4.1 控制应用程序窗口

要控制AutoCAD应用程序窗口，需要使用Application对象中的属性和方法，它可以改变窗口的大小、位置、可见性以及其最大化最小化等。

4.1.1改变应用程序窗口的位置和尺寸

使用Application对象的WindowTop和WindowLeft属性可以改变应用程序窗口的位置，使用Width 和 Height 属性可以改变应用程序窗口的大小。

例4-1：将 AutoCAD应用程序窗口放在屏幕的左上角，并将其大小调整为宽 400 像素、高 400 像素。 

Sub Ch4_PositionApplicationWindow( )

ThisDrawing.Application.WindowTop = 0

ThisDrawing.Application.WindowLeft = 0

ThisDrawing.Application.Width = 400

ThisDrawing.Application.Height = 400

End Sub

4.1.2设置和查询应用程序窗口状态

1.设置Application对象的WindowState属性 （acMax、asMin或acNorm）

例4-2：最大化应用程序窗口。

Sub Ch4_MaximizeApplicationWindow( )

ThisDrawing.Application.WindowState = acMax

End Sub

若最小化应用程序窗口,则按下列语句:

ThisDrawing.Application.WindowState = acMin

2.查询窗口当前状态

例4-3：查询应用程序窗口当前状态，并将该状态以消息框的形式显示给用户。

Sub Ch4_CurrentWindowState( )

Dim CurrWindowState As Integer

Dim Msgstr As String

CurrWindowState = ThisDrawing.Application.WindowState

Msgstr = Choose（CurrWindowState,“正常”,“最小化”,“最大化”）

MsgBox“AutoCAD窗口当前状态为：”+ Msgstr

End Sub

4.2 图形文件的建立、打开、关闭和保存

从AutoCAD 2000版开始，支持多文档界面（MDI）。在多文档界面条件下，可以同时打开多个图形。

4.2.1创建新图形

在MDI模式下，建立一个新图形，要用Documents集合的Add方法。

其可选参数为样板图文件名（.dwt文件），若未指定，系统将提供默认的样板图（例如：acad.dwt）。

通常存储在样板图文件中的设置包括 :

（1）单位类型和精度。

（2）标题栏、边框和徽标。 

（3）图层名。

（4）捕捉、栅格和正交设置。 

（5）图形（栅格）界限。

（6）标注样式。 

（7）文字样式。 

（8）线型。

AutoCAD2005版以后，可用IAcadPreferencesFiles2接口的QNewTemplateFile属性设置缺省样板图文件名，这样将会改变这个计算机的缺省样板图文件。

例4-4：用QNewTemplateFile属性设置缺省样板图文件acadiso.dwt。

Sub Ch4_ QnewTemplateFile( )

Dim MyTemplate As IAcadPreferencesFiles2

Set MyTemplate =AcadApplication.Preferences.Files

MyTemplate.QnewTemplateFile=“c:\AutoCAD2005\acadiso.dwt“

End Sub

例4-5：使用 Add 方法根据样板图创建新图形。

Sub Ch4_NewDrawing( )

Dim acadDoc As AcadDocument

Set acadDoc = ThisDrawing.Application.Documents.Add _

”c:\AutoCAD2004\Gb_a3 –Named Plot Style.dwt”

End Sub 

若未指定样板图，则没有后面的参数。如：

Set acadDoc = ThisDrawing.Application.Documents.Add 

注：若按单文档界面（SDI）模式，要使用Document对象的 New方法。例如：ThisDrawing.New 。

4.2.2打开现有图形

利用Open方法可将已有的图形（.dwg文件）打开并设为当前图形。

语法: Object.Open Name[,ReadOnly][,Password]

式中: Object——Documents，Document

Name——String

ReadOnly——变体；只可输入；可选项；只能用于Documents

缺省设置：FALSE

TRUE：以只读方式打开图形

FALSE：以可读可写方式打开图形

Password——变体；可选项（注：自AutoCAD 2004 版起增添了该选项）

口令用于打开一个加密的图形

例4-6：打开一个图形文件（包括路径的图形名），先用VB 中的 Dir函数检查它是否存在，若不存在，可显示警告，再改换为新建方式。

Sub Ch4_OpenDrawing( )

Dim FileNameStr As String

FileNameStr = App.Path & “\” & “bolt.dwg”

Dim FileNameStr1 As String

FileNameStr1 = App.Path & “\” & “acadiso.dwt”

If Dir(FileNameStr) <>“ ” Then

ThisDrawing.Application.Documents.Open FileNameStr 

Else

MsgBox“文件” & FileNameStr & “不存在！” _

&“改用样板图 acadiso.dwt”

ThisDrawing.Application.Documents.Add FileNameStr1

End If

End Sub

4.2.3保存图形

利用Save和SaveAs方法保存当前图形。SaveAs方法执行“另存为”操作，要求指定文件名。使用 Saved 属性可以确保当前图形中不存在任何未保存的更改。自AutoCAD 2004 版起增添了可保存加密图形的选项。

例4-7：本例首先检查是否存在未保存的更改，并验证用户是否选择“确定”以保存图形。如果用户选择“确定”，将使用 Save 方法用其当前名称保存当前的图形。然后再用新名保存该图形。

Sub Ch4_SaveActiveDrawing( )

'用当前名称保存当前的图形

If Not (ThisDrawing.Saved) Then

If MsgBox("您希望保存该图形吗?",vbYesNo) = vbYes Then

ThisDrawing.Save

End If

End If

'用新名称保存活动的图形

ThisDrawing.SaveAs "MyDrawing.dwg"

End Sub

4.2.4关闭图形

利用Close方法关闭指定的图形（当从Document对象调用该方法时）或关闭所有打开的图形（当从Documents对象调用该方法时）。

例4-8：本例循序处理图形文件集合，接着关闭所有打开的图形，如果没有打开的图形文件则退出。

Sub Ch4_CloseDrawing( )

Dim Doc As AcadDocument

' 如果没有打开的图形文件则退出

If Documents.count = 0 Then

MsgBox "没有打开的图形文件!"

Exit Sub

End If

' 关闭所有打开的图形文件

For Each Doc In Documents

If MsgBox("您希望关闭该图形文件吗: " & Doc.WindowTitle, vbYesNo & vbQuestion) = vbYes Then

If Doc.FullName <> "" Then

Doc.Close

Else

MsgBox Doc.name & " 尚未存储,因此不会被关闭！"

End If

End If

Next

End Sub

4.3 管理图形窗口

与 AutoCAD应用程序窗口类似，图形窗口也可以最小化、最大化、重新定位、调整大小以及查看其状态。可以使用 Document 对象的WindowState属性使图形窗口最小化或最大化。可以使用 Document 对象的Width和Height属性设定图形窗口的大小。除此之外，还可以使用视图、视口和缩放方法来更改图形在窗口中的显示方式；可以保存视图并在需要打印或参考特定细节时将其恢复以及将屏幕拆分为多个平铺视口同时显示多个视图。

4.3.1使用缩放

视区是一个图形按特定的比例、位置和方向的显示形式。改变视区最常用的方法就是使用缩放命令，以增大或缩小图形在绘图区中显示的尺寸。AutoCAD提供了窗口显示、按特定比例缩放、置中和显示整个图形等方法。

1． 定义“缩放”窗口

可以通过指定定义区域的对角两个点快速放大该区域。要通过指定边界来放大某个区域，需要使用Application 对象的ZoomWindow 或 ZoomPickWindow 方法。ZoomWindow 方法使用户可以通过编程的方式定义代表“缩放”窗口的两个点；而 ZoomPickWindow 则要求用户拾取两个点，这两个拾取点将形成“缩放”窗口。

例4-9：将当前图形缩放至两点定义的窗口。

Sub Ch4_ZoomWindow( )

Dim point1(0 To 2) As Double

Dim point2(0 To 2) As Double

point1(0) = 1.3: point1(1) = 7.8: point1(2) = 0

point2(0) = 13.7: point2(1) = -2.6: point2(2) = 0

ThisDrawing.Application.ZoomWindow point1, point2

End Sub

2． 缩放视图

如果需要按精确的比例缩放图像，需要使用Application对象的 ZoomScaled 方法。该方法需要输入两个参数：缩放比例和缩放类型。

可以使用以下三种方法指定缩放比例： 

（1）相对于图形界限。

（2）相对于当前视区。

（3）相对于图纸空间单位。

缩放类型取决于缩放比例是如何确定的。当相对于图形界限缩放时，使用常量 acZoomScaledAbsolute；相对于当前视区缩放时，使用常量 acZoomScaledRelative；相对于图纸空间单位缩放时，使用常量 acZoomScaledRelativePSpace。

例4-10：使用指定的缩放比例放大当前图形。

Sub Ch4_ZoomScaled( )

MsgBox "用缩放类型: acZoomScaledRelative" & _

"比例因子为2 " & "执行缩放", , "ZoomScaled方法例"

Dim scalefactor As Double

Dim scaletype As Integer

scalefactor = 2

scaletype = acZoomScaledRelative

ThisDrawing.Application.ZoomScaled scalefactor, scaletype

End Sub

3．居中对齐对象 

Application对象的ZoomCenter 方法可用于调整对象大小并将其置于视口的中心。

该方法需要输入两个参数：缩放中心点坐标和窗口的缩放程度。

例4-11：居中对齐对象。

Sub Ch4_ZoomCenter( )

MsgBox "使用中心点坐标（3, 3, 0）" & "缩放程度: 10" _

& "居中对齐对象", , "ZoomCenter方法例"

Dim Center(0 To 2) As Double

Dim magnification As Double

Center(0) = 3: Center(1) = 3: Center(2) = 0

magnification = 10

ThisDrawing.Application.ZoomCenter Center, magnification

End Sub

4．显示图形界限和范围

要根据图形边界或图形中对象的范围显示视图，要使用ZoomPrevious、 ZoomAll或ZoomExtents 方法。

（1） ZoomPrevious将当前视口缩放到前一个范围。

（2） ZoomAll用于显示整个图形范围。如果对象超出界限，ZoomAll 将显示对象的范围。如果对象绘制在界限内，ZoomAll 将显示界限。

（3） ZoomExtents根据当前视口的实际范围来计算缩放。若图形中没有对象，则显示图形范围。

注：ZoomAll和ZoomExtents属应用程序对象的方法，该两属性有时会导致图形重新生成（Regen）。

例4-12：将当前图形缩放至图形上的所有内容和图形范围。

Sub Ch4_ZoomAllZoomExtents ( )

'使用ZoomAll

MsgBox "执行 ZoomAll操作", , "ZoomAll方法例"

ThisDrawing.Application.ZoomAll

'使用ZoomExtents

MsgBox " 执行ZoomExtents操作", , "ZoomExtents方法例"

ThisDrawing.Application.ZoomExtents

End Sub

4.3.2 使用命名视图 

可以命名和保存要重复使用的视图。如果不再需要该视图，可以删除它。 所谓View对象，即从空间中特定位置（检视点）以图形表示的二维图形或三维模型。要创建新视图，要使用 Add 方法将新视图添加到 Views 集合中。保存图形时，会同时保存视图的查看位置和比例。 

可以在创建视图时命名该视图。视图名称最长可为 255 个字符，可以包含字母、数字和特殊字符（美元符 “$”、连字符 “-” 和下划线 “_”）。

要删除命名视图，只需使用View对象的Delete 方法。但Views集合不允许使用Delete方法。

例4-13：将命名视图添加到 Views 集合中。

Sub Ch4_AddView()

'将命名视图添加到 Views 集合中

Dim viewObj As AcadView

Set viewObj = ThisDrawing.Views.Add("View1")

End Sub

例4-14：删除 View 对象。 

Sub Ch4_DeleteView( )

Dim viewObj As AcadView

Set viewObj = ThisDrawing.Views("View1")

'删除视图

viewObj.Delete

End Sub

例4-15：从 Views 集合中删除命名视图。 

Sub Ch4_DeleteViewFromCollection()

ThisDrawing.Views("View1").Delete

End Sub

4.3.3 使用平铺视口

在模型空间中操作时，是在平铺视口（Viewport 对象）中绘制几何图形。用户一次可以显示一个或多个不同视口。当同时显示多个平铺视口时，一个视口中的编辑操作会影响所有其他视口。在每个平铺视口中，都可以进行以下操作。 

（1）在单个视口中缩放、设置捕捉、栅格和 UCS 图标模式，以及恢复命名视图。

（2）执行命令时，从一个视口绘制到另一个视口。

（3）命名视口的配置以便可以重复使用。

可以使用不同的配置来显示平铺视口。视口的显示方式取决于所需的视图数量和大小。 

1．拆分活动视口 

要拆分活动视口，应使用 Split 方法。该方法只需一个参数，即拆分视口的配置类型。该参数有以下设定值：

（1）acViewport2Horizontal

（2）acViewport2Vertical

（3）acViewport3Left

（4）acViewport3Right

（5）acViewport3Horizontal

（6）acViewport3Vertical

（7）acViewport3Above

（8）acViewport3Below 

（9）acViewport4

2.使另一个平铺视口成为当前视口 

要将某平铺视口置为当前视口，需使用 ActiveViewport 属性。 可以遍历现有视口以查找特定视口。首先使用 Name 属性识别所需视口所在的视口集合中的名称。如果视口已被拆分，则可以通过 LowerLeftCorner 和 UpperRightCorner 属性来标识配置中的每个单独视口的位置坐标。

例4-16：本例将视口拆分为四个平铺视口，然后遍历图形中的所有视口并显示视口名称以及每个视口的左下角点和右上角点。 

	VB源码彩色显示 注：在[code]...[/code]内就以下面格式显示文字，双击源码拷贝到剪切板

	Sub Ch4_IteratingViewportWindows( )

'创建新视口并将其置为活动视口

Dim vportObj As AcadViewport

Set vportObj = ThisDrawing.Viewports.Add("TEST_VIEWPORT")

ThisDrawing.ActiveViewport = vportObj

'将 vportObj 拆分为 4 个平铺视口

vportObj.Split acViewport4

'遍历视口，亮显每个视口并显示

'每个视口的左下角点和右上角点

Dim vport As AcadViewport

Dim LLCorner As Variant

Dim URCorner As Variant

For Each vport In ThisDrawing.Viewports

ThisDrawing.ActiveViewport = vport

LLCorner = vport.LowerLeftCorner

URCorner = vport.UpperRightCorner

MsgBox "视口: " & vport.Name & "现在激活" & _

vbCrLf & "左下角: " & _

LLCorner(0) & ", " & LLCorner(1) & vbCrLf & _

"右上角: " & URCorner(0) & ", " & URCorner(1), , _

“平铺视口”

Next vport

End Sub



先运行某程序生成一个图形，再运行上述通用过程生成四个平铺视口，然后调整各分拆视口的缩放比例和视点，结果如图4-1所示。

图4-1平铺视口

4.3.4图形对象的重新生成和更新

通过 AutoCAD ActiveX Automation执行的许多操作都会修改 AutoCAD 图形中显示的内容，但修改不是都能立即在绘图窗口中显示出来。可以使用Update 和Regen方法更新显示图形对象。其中Update只能更新单一对象的显示；而Regen适用于Document对象。Regen可以使整个图形重新生成，重新计算并刷新某一视图区或所有视图区。

例4-17：创建一个圆并将其设置为红色，然后使用 Update 方法更新该圆，使其在 AutoCAD 中可见。 

	VB源码彩色显示 注：在[code]...[/code]内就以下面格式显示文字，双击源码拷贝到剪切板

	Sub Ch4_UpdateDisplay( )

Dim circleObj As AcadCircle

Dim center(0 To 2) As Double

Dim radius As Double

center(0) = 1: center(1) = 1: center(2) = 0

radius = 1

'创建圆并将其设置为红色

Set circleObj = ThisDrawing.ModelSpace.AddCircle(center,radius)

circleObj.Color = acRed

'更新圆的显示

circleObj.Update

End Sub

例4-18：该程序修改了字体文件，然后调用Regen方法更新图形中的所有视口。

Sub Ch4_Example_Regen( )

ThisDrawing.ActiveTextStyle.FontFile = _

“c:\Windows\font\simfang.ttf”

ThisDrawing.Regen acAllViewports

End Sub



4.3.5重置当前对象

对于大部分当前对象（例如当前图层和当前线型），对其所做的修改会立即显示。但有些当前对象必须要重置然后再调用 Regen 方法才能显示其更改的结果。这些对象包括当前文字样式、当前 UCS 和当前视口。要重置这些对象，即设置这些已更新对象的 ActiveTextStyle、 ActiveUCS、或ActiveViewport 属性。

例4-19：该例返回当前的字型，接着设定新的型式，最后再将字型恢复为原来的设定值。

	VB源码彩色显示 注：在[code]...[/code]内就以下面格式显示文字，双击源码拷贝到剪切板

	Sub Ch4_ActiveTextStyle( )

Dim newTextStyle As AcadTextStyle

Dim currTextStyle As AcadTextStyle

'返回当前的字型

Set currTextStyle = ThisDrawing.ActiveTextStyle

MsgBox " 当前的字型是：" & currTextStyle.name, vbInformation, _ "ActiveTextStyle 例"

'创建新字型并设为当前的字型

Set newTextStyle = ThisDrawing.TextStyles.Add("TestTextStyle")

ThisDrawing.ActiveTextStyle = newTextStyle

MsgBox "新字型是：" & newTextStyle.name, vbInformation, _ "ActiveTextStyle 例"

'将字型重设为原来的字型

ThisDrawing.ActiveTextStyle = currTextStyle

MsgBox "字型重设为：" & currTextStyle.name, vbInformation, _ "ActiveTextStyle 例"

End Sub



4.4 系统变量的存取

AutoCAD将操作环境和一些命令的设置（或值）存储在系统变量中。每个系统变量都有一定的类型：整型、浮点型、数组、布尔类型或字符串类型。

Document 对象提供了 SetVariable 和 GetVariable 方法，分别用于设置和查询 AutoCAD 的系统变量。

GetVariable方法的语法：RetVal=object.GetVariable(Name) 

其中：

object —— Document

Name—— 字符串；只可输入

要返回的变量名称，必须含有有效的系统变量名称（不分字符的大小写）

RetVal（返回值）—— Variant型变量

SetVariable方法的语法：object.SetVariable Name,Value

其中：

object—— Document

Name —— 字符串；只可输入

要设置的系统变量名

Value —— Variant型变量；只可输入

设置系统变量的新值

例4-20: 查询和设置不同数据类型的系统变量。

	VB源码彩色显示 注：在[code]...[/code]内就以下面格式显示文字，双击源码拷贝到剪切板

	Sub Ch4_SetVariable( )

'该例查询和设置不同数据类型的系统变量

Dim sysVarName As String

Dim sysVarData As Variant

Dim DataType As Integer

'将FILEDIA系统变量设置为1(整型数)。

'注意：需要声明一个适当数据类型的变量，将数据赋给该变量， 

'然后再将变量设为变体类型

Dim intData As Integer

sysVarName = "FILEDIA"

intData = 1

sysVarData = intData '整型数变量设为变体类型

ThisDrawing.SetVariable sysVarName, sysVarData

'应用GetVariable查询系统变量

sysVarData = ThisDrawing.GetVariable(sysVarName)

MsgBox sysVarName & " = " & sysVarData, , "SetVariable 例"

'设置DCTCUST系统变量(字符串类型)值为"我的自定词典"

Dim strData As String

sysVarName = "DCTCUST"

strData = "我的自定义词典"

sysVarData = strData '字符串型

ThisDrawing.SetVariable sysVarName, sysVarData

'应用GetVariable查询系统变量

sysVarData = ThisDrawing.GetVariable(sysVarName)

MsgBox sysVarName & " = " & sysVarData , , "SetVariable 例"

'设置CHAMFERA系统变量(双精度浮点数)值为1.5

Dim dataDouble As Double

sysVarName = "CHAMFERA"

dataDouble = 1.5

sysVarData = dataDouble '双精度浮点数

ThisDrawing.SetVariable sysVarName, sysVarData

'应用GetVariable查询系统变量

sysVarData = ThisDrawing.GetVariable(sysVarName)

MsgBox sysVarName & " = " & sysVarData, , "SetVariable 例"

'设置INSBASE系统变量(双精度浮点数)值为(1.0,1.0,0)

Dim arrayData3D(0 To 2) As Double

sysVarName = "INSBASE"

arrayData3D(0) = 1#: arrayData3D(1) = 1#: arrayData3D(2) = 0

sysVarData = arrayData3D '三维数组

ThisDrawing.SetVariable sysVarName, sysVarData

'应用GetVariable查询系统变量

sysVarData = ThisDrawing.GetVariable(sysVarName)

MsgBox sysVarName & " = " & sysVarData(0) & ", " & _ 

sysVarData(1) & ", " & sysVarData(2), , "SetVariable 例"

End Sub




4.5 调用AutoCAD命令

利用Document对象的SendCommand方法，执行向命令行中发送命令的操作，就像在AutoCAD命令行直接键入命令一样。该方法发送一个字符串到命令行, 其中，必须包括“命令”及其参数，且其顺序必须和“命令”提示的顺序一致。字符串中的空格或回车键的ASCII码等效于键入回车键。

例4-21：绘制一个圆心为(150,100,0),半径为50的圆，然后放大到最大（代码形式1）。

Sub Ch4_SendACommandToAutoCAD1( )

ThisDrawing.SendCommand "_Circle 150,100,0 50 "

ThisDrawing.SendCommand "_zoom e "

End Sub

例4-22：绘制一个圆心为(150,100,0)，半径为50的圆，然后放大到最大（代码形式2）。

	VB源码彩色显示 注：在[code]...[/code]内就以下面格式显示文字，双击源码拷贝到剪切板

	Sub Ch4_SendACommandToAutoCAD2( )

ThisDrawing.SendCommand _

(“_Circle” & vbCr & “150,100,0” & vbCr & “50” & vbCr )

ThisDrawing. SendCommand _

(“_zoom” & vbCr & “e” & vbCr)

End Sub



4.6 组织图形元素

4.6.1颜色

自AutoCAD 2004版开始，颜色属性Color已被TrueColor替代，但应用旧版开发出的应用程序目前仍可使用。用户可以使用 AcCmColor 对象将真彩色指定给图形中的各个对象。使用 AcCmColor 对象中的 RGB 值，可以在设置图形对象的颜色时从上百万的颜色中进行选择。AcCmColor 对象包含的方法和属性还可以指定颜色名称、配色系统、颜色索引、颜色值和颜色方法。

指定颜色时，可以输入颜色名称或是AutoCAD颜色的索引 AutoCAD Color Index(ACI) 编号。ACI 提供 255 个颜色编号。只有颜色编号 1 到 7 的颜色具有标准颜色名称。如表4-1所示。

表4-1常用标准颜色

编号 名称 常数 1 红 acRed 2 黄 acYellow 3 绿 acGreen 4 青 acCyan 5 蓝 acBlue 6 紫 acMagenta 7 白/黑 acWhite 

要更改对象的颜色，应使用该对象的 TrueColor 属性。

例4-23：创建一个圆，然后将其着色为蓝色。 

	VB源码彩色显示 注：在[code]...[/code]内就以下面格式显示文字，双击源码拷贝到剪切板

	Sub Ch4_TrueColor( )

Dim color As AcadAcCmColor

Set color = _

AcadApplication.GetInterfaceObject("AutoCAD.AcCmColor.16")

Call color.SetRGB(80, 100, 244)

Dim circleObj As AcadCircle

Dim centerPoint(0 To 2) As Double

Dim radius As Double

centerPoint(0) = 0#: centerPoint(1) = 0#: centerPoint(2) = 0#

radius = 5#

Set circleObj = _

ThisDrawing.ModelSpace.AddCircle(centerPoint, radius)

circleObj.TrueColor = color

ZoomAll

End Sub



注：若用于AutoCAD2007版，应更改相应的类型库代码。

4.6.2线型

1．将需要的线型加载入图形中

所有的线型都定义在AutoCAD的线型文件acad.lin中。如果要使用某种线型，需要使用线型集合的Load方法。该方法需要两个参数，第一个参数为要加载的线型有效名称，第二个参数为线型所在的线型文件。

例4-24：寻找“CENTER” 线型，若未发现，则从acad.lin文件中加载。

	VB源码彩色显示 注：在[code]...[/code]内就以下面格式显示文字，双击源码拷贝到剪切板

	Sub Ch4_ LoadLinetype( )

Dim entry As AcadLineType

Dim found As Boolean

Found=False

For Each entry In ThisDrawing.LineTypes

If StrComp(entry.Name,”CENTER”,1) = 0 Then

Found=True

Exit For

End If

Next

On Error Resume Next

If Not(found) Then

ThisDrawing.LineTypes.Load “CENTER”,”acad.lin”

End If

End Sub



2.设置当前线型

（1） 选择BYLAYER——新对象会使用当前图层的线型属性所设置的当前（活动）线型。

（2） 使用ActiveLineType属性——如果要在当前图层上使用非本图层的线型绘图，必须先使该线型成为当前线型。

3.设置线型比例

对于一个具体的图形对象，要变换其线型比例，应直接使用LineTypeScale属性。

例：

CircleObj.LineTypeScale=2

CircleObj.Update

注:系统变量LTSCALE 控制全局更改现有对象和新对象的线型比例。要修改系统变量的值，要使用 SetVariable 方法。

例：用SetVariable将用于控制线型比例的系统变量LTSCALE设置为2，其代码为：

ThisDrawing.SetVariable “LTSCALE”，2

4.改变图形对象的线型

默认情况下，对象将继承创建时所在图层的线型。可以使用对象的 Linetype 属性更改该对象的线型。

例4-25：本例创建一个圆，然后从 acad.lin 文件加载“CENTER”线型。如果该线型已存在或者文件不存在，则显示相关消息。最后，将圆的线型设置为“CENTER”。 

	VB源码彩色显示 注：在[code]...[/code]内就以下面格式显示文字，双击源码拷贝到剪切板

	Sub Ch4_ChangeCircleLinetype( )

On Error Resume Next

'创建圆

Dim circleObj As AcadCircle

Dim center(0 To 2) As Double

Dim radius As Double

center(0) = 2: center(1) = 2: center(2) = 0

radius = 1

Set circleObj = ThisDrawing.ModelSpace.AddCircle(center, radius)

Dim linetypeName As String

linetypeName = "CENTER"

'从 acad.lin 文件加载“CENTER”线型

ThisDrawing.Linetypes.Load linetypeName, "acad.lin"

If Err.Description <> "" Then MsgBox Err.Description

'指定圆的线型为“CENTER”

circleObj.Linetype = "CENTER"

circleObj.Update

End Sub



4.6.3线宽

线宽可以由位于Document对象之下的 DatabasePreferences 对象、层或图形对象的Lineweight属性设定。线宽的值可以设为随层（值为acLnWtByLayer）、随块（值为acLnWtByBlock）、缺省（值为acLnWtByLwDefault）或直接设为标准值（如设为acLnWt080，即线宽为0.8mm）。缺省的线宽值由系统变量LWDEFAULT设定，其初始值为0.25mm。所有新图层中的线宽都使用默认设置。DatabasePreferences对象的LineweightDisplay属性决定在模型空间是否要显示线宽。

4.6.4图层

1． 查找图层

用户可以遍历Layers集合，用For Each语句查找。

例4-26：遍历Layers集合，查找图层。

Sub Ch4_FindLayer( )

Dim AllLayerNames As string

Dim EntryObj As AcadLayer

AllLayerNames=“ ”

For Each EntryObj In ThisDrawing.Layers

AllLayerNames= AllLayerNames+ EntryObj.Name+ -

VbCrLf

Next

MsgBox “本图形的图层有:” +vbCrLf+AllLayerNames

End Sub

2．创建图层和设置为当前图层

图形中的每一个图层都是Layers集合的成员，创建新图层需要使用Layers集合的Add方法。该方法需要一个参数，这个参数为新图层的名称。

对于自动创建的0层，默认颜色号码为7（白或黑色），线型为连续线CONTINUOUS，该图层是不可删除的。

要切换图层，需要使用Document对象的 ActiveLayer 属性。此属性适用于设置当前图形。如果某图层已被冻结，则不能设置为当前图形。

例4-27：创建一个名为“MyLayer”的图层，指定该层的颜色、线宽和线型，并将其设定为当前图层。设此前已加载了线型”CENTER”。

Sub Ch4_NewLayer( )

Dim MyLy As AcadLayer

Dim MyLayer As String

Set MyLy=ThisDrawing.Layers.Add(“MyLayer”) '创建图层

MyLy.Color=acRed '指定“MyLayer”层的颜色

MyLy.Lineweight=acLnWt080 '指定“MyLayer”层的线宽为0.8mm

MyLy.LineType=”CENTER” '指定“MyLayer”层的线型

ThisDrawing.ActiveLayer=MyLy '激活“MyLayer”层为当前层

End Sub

3． 更改对象的图层

用户创建对象之后，还可以使用 Layer 属性更改对象的图层。使用Layer 属性需要输入图层的名称。。

例4-28：本例在活动图层上创建一个圆，然后创建新图层“ABC”，接着将圆移到新图层上。 

Sub Ch4_MoveObjectNewLayer( )

'创建圆

Dim circleObj As AcadCircle

Dim center(0 To 2) As Double

Dim radius As Double

center(0) = 2: center(1) = 2: center(2) = 0

radius = 1

Set circleObj = ThisDrawing.ModelSpace.AddCircle(center, radius)

'创建新图层“ABC”

Dim layerObj As AcadLayer

Set layerObj = ThisDrawing.Layers.Add("ABC")

'将圆指定到“ABC”图层

circleObj.Layer = "ABC"

circleObj.Update

End Sub

4． 删除图层

要删除图层，要使用Layer对象的Delete 方法。 随时可以删除图层，但不能删除当前图层、0图层、依赖于外部参照的图层和包含对象的图层。 

对于被块定义所参照的图层以及特殊的图层 DEFPOINTS，即使其中没有包含可见对象也不能被删除。

5． 打开和关闭图层

关闭的图层仍会和图形一起重生成，但不能显示或打印。通过关闭图层，可以避免每次解冻图层时都重生成图形。打开已关闭的图层时，AutoCAD 将重画该图层上的对象。 

要打开和关闭图层，要使用层对象的 LayerOn 属性。如果此属性为TRUE，则打开图层；如果为 FALSE，将关闭图层。 

6．冻结和解冻图层

用户可以冻结图层以加快显示改变的速度，提高对象选择性能，以及减少复杂图形的重生成时间。AutoCAD 不会显示、打印和重生成冻结图层上的对象。可以冻结长时间不需要显示的图层。解冻冻结的图层时，AutoCAD 将重生成并显示该图层上的对象。 

要冻结或解冻图层，要使用Layer对象的 Freeze 属性。如果此属性为 TRUE，将冻结图层；如果为 FALSE，将解冻图层。

例4-29：创建新的图层“ABC”，然后冻结该图层。 

Sub Ch4_LayerFreeze( )

'创建新图层“ABC”

Dim layerObj As AcadLayer

Set layerObj = ThisDrawing.Layers.Add("ABC")

'冻结图层“ABC”

layerObj.Freeze = True

End Sub

7． 锁定和解锁图层

用户不能编辑锁定图层上的对象，但如果图层是打开和解冻的，这些对象仍然可见。可以将锁定图层置为当前并在其上添加对象。可以冻结和关闭锁定图层，也可以更改其关联的颜色和线型。 

要锁定或解锁图层，要使用Layer对象的Lock 属性。如果此属性为 TRUE，将锁定图层；如果为 FALSE，将解锁图层。

例4-30：创建新的图层“ABC”，然后将其锁定。 

Sub Ch4_LayerLock( )

'创建新图层“ABC”

Dim layerObj As AcadLayer

Set layerObj = ThisDrawing.Layers.Add("ABC")

'锁定图层“ABC”

layerObj.Lock = True

End Sub 

4.7 设置AutoCAD系统配置

选择菜单“工具（T）”|“选项（N）…”可打开“选项”对话框。该对话框是用来设置AutoCAD系统环境的，它共有九个选项卡。与其相对应，Prefernences对象下面的九个对象均对应该对话框的每一个选项卡。这些对象提供存取“选项”对话框中保存和登录的设定值。

（1）PrefernencesDisplay 对应“显示” 选项卡。

（2）PrefernencesDrafting 对应“草图” 选项卡。

（3）PrefernencesFiles 对应“文件” 选项卡。

（4）PrefernencesOpenSave 对应“打开和保存” 选项卡。 

（5）PrefernencesOutput 对应“打印” 选项卡。

（6）PrefernencesProfile 对应“配置” 选项卡。

（7）PrefernencesSelecting 对应“选择” 选项卡。

（8）PrefernencesSystem 对应“系统” 选项卡。

（9）PrefernencesUser 对应“用户系统配置” 选项卡。

可以通过Preferences 对象访问这些对象。要访问 Preferences 对象，要使用 Application 对象的 Preferences 属性。例如： 

Dim acadPref as AcadPreferences

Set acadPref = ThisDrawing.Application.Preferences

然后使用 Display、Drafting、Files、OpenSave、Output、Profile、Selection、System 和 User 属性来访问任何特定的 Preferences 对象。

例4-31：设置草图参数。

Sub Ch4_PrefsSetDrafting ( )

Dim AcadPreference As AcadPreferences

Set AcadPreference = ThisDrawing.Application.Preferences

AcadPreference.Drafting.AutoSnapMarker = True

AcadPreference.Drafting.AutoSnapMarkerColor = acGreen

AcadPreference.Drafting.AutoSnapMarkerSize = 6

End Sub

注：

1）除上述九个 Preferences 对象外，还有一个位于 Document 对象之下的 DatabasePreferences 对象，它包含存储在图形中的所有选项。它使得应用程序可以使用存储在图形中的各个选项以访问 AutoCAD 图形而无须事先启动 AutoCAD 应用程序（ObjectDBXTM）。 

2）用户也可以利用SetVariable和GetVariable方法设定和修改选项（以及不是“选项”对话框的一部分系统变量）。

4.8 设置Document对象的当前属性和确定容器对象

4.8.1设置Document对象的当前属性

Document对象的当前属性，指当前空间、当前层、当前文本样式等。

对于以后创建的任何图形对象，如不另外指定，就使用这些当前的属性。常用的当前属性如表4-2所示。

表4-2 Document对象常用的当前属性

属性 默认值 含义 ActiveDimStyle Standard 当前尺寸标注样式 ActiveSpace acModelSpace 当前空间 ActiveTextStyle Standard 当前文本样式 ActiveLayer 0层 当前层 ActiveLineType Bylayer 当前线型 ActiveLayout Model布局 当前布局 

4.8.2确定容器对象

图形对象是在 ModelSpace 集合、PaperSpace 集合或 Block 对象中创建的。这三种对象也称为容器对象。

ModelSpace 集合由 ModelSpace 属性返回，而 PaperSpace 集合由 PaperSpace 属性返回。 

用户可以直接引用或通过用户定义的中间变量来引用这些对象。要直接引用对象，要将对象包含在调用层次结构中。例如，以下语句将在模型空间中添加一条直线： 

Set lineObj = ThisDrawing.ModelSpace.AddLine(startPoint,endPoint)

下例定义两个变量，分别将其设置为当前模型空间和图纸空间： 

Dim moSpace As AcadModelSpace

Dim paSpace As AcadPaperSpace

Set moSpace = ThisDrawing.ModelSpace

Set paSpace = ThisDrawing.PaperSpace

以下语句使用用户定义的中间变量在模型空间中添加一条直线： 

Set lineObj = moSpace.AddLine(startPoint,endPoint)

4.9 输入和输出其他图形文件格式

可以用指定格式打开并使用其他应用程序的图形或图像。AutoCAD 可以处理 DXF、SAT 和 WMF 文件的某些转换格式。在所有版本中，都可以使用 Import 方法来输入文件。该方法需要三个输入值：要输入的文件的名称、在图形中放置文件的插入点以及放置输入图形时使用的缩放比例。 

如果需要在其他应用程序中使用 AutoCAD 图形，可以使用 Export 方法将图形转换为特定格式。可以将 AutoCAD 图形输出为 WMF、SAT、EPS、DXF 或 BMP 格式。Export 方法需要三个输入值：要创建的新文件的名称、新文件的扩展名和要输出的对象选择集。 

输出为 WMF、SAT 或 BMP 格式时，必须提供非空的选择集，该选择集指定要从图形中输出的对象。如果没有指定选择集，则不会输出任何内容，并且会产生可捕获的无效参数错误。 

输出为 EPS 和 DXF 格式时，Export 会忽略选择集参数，但仍需要提供该参数。对于这两种格式，会自动输出整个图形。 

例:在当前图形中创建一个圆，然后将图形输出到名为 DXFExprt.DXF 的文件，最后打开一个新图形并输入该文件。注意，这里提供了一个空的选择集作为 Export 的参数。当输出 DXF 文件时，虽然 Export 方法会忽略选择集信息，但省略该参数将导致语法错误。

例4-32： 输出和输入其它格式的图形文件

	VB源码彩色显示 注：在[code]...[/code]内就以下面格式显示文字，双击源码拷贝到剪切板

	Sub Ch4_ImportingAndExporting( )

'创建圆用于直观显示

Dim circleObj As AcadCircle

Dim centerPt(0 To 2) As Double

Dim radius As Double

centerPt(0) = 2: centerPt(1) = 2: centerPt(2) = 0

radius = 1

Set circleObj = ThisDrawing.ModelSpace.AddCircle(centerPt, radius)

ThisDrawing.Application.ZoomAll

'创建空的选择集

Dim sset As AcadSelectionSet

Set sset = ThisDrawing.SelectionSets.Add("NEWSSET")

'将当前图形输出到AutoCAD 临时文件目录下的 DXF 文件

Dim tempPath As String

Dim exportFile As String

Const dxfname As String = "DXFExprt"

tempPath =ThisDrawing.Application.preferences.Files.TempFilePath

exportFile = tempPath & dxfname

ThisDrawing.Export exportFile, "DXF", sset 

'删除空的选择集

ThisDrawing.SelectionSets.Item("NEWSSET").Delete

'打开新图形

ThisDrawing.Application.Documents.Add "acad.dwt"

'定义输入

Dim importFile As String

Dim insertPoint(0 To 2) As Double

Dim scalefactor As Double

importFile = tempPath & dxfname & ".dxf"

insertPoint(0) = 0: insertPoint(1) = 0: insertPoint(2) = 0

scalefactor = 2#

'输入文件

ThisDrawing.Import importFile, insertPoint, scalefactor

ThisDrawing.Application.ZoomAll

End Sub



4.10 控制AutoCAD环境和组织图形元素的综合示例

控制AutoCAD环境和组织图形元素综合示例的窗体如图4-2所示。


图4-2 AutoCAD环境设置和组织图形元素综合示例的窗体

本综合示例控制AutoCAD绘图环境的功能主要有:

（1）查询应用程序窗口状态。

（2）查询图形窗口状态。

（3）设置图层及其颜色、线型、线宽。

（4）查找图层。

（5）查找线型。

（6）设当前层。

（7）设置真彩色。

（8）设置草图参数。

（9）设置平铺视口。

（10）求图形的交点。

（11）查看直线与X轴的夹角。

（12）输出及输入其它文件格式。

该例的源程序收录在本书附带的光盘中，下面仅介绍其中的一段。

例4-33：设置图层（及其颜色、线型、线宽）及设为当前层。

	VB源码彩色显示 注：在[code]...[/code]内就以下面格式显示文字，双击源码拷贝到剪切板

	Sub Ch4_CreateLayer( )

'设置图层及其颜色、线型、线宽

Dim Layer0 As AcadLayer

Dim Layer1 As AcadLayer

Dim Layer2 As AcadLayer

Dim Layer3 As AcadLayer

Dim Layer4 As AcadLayer

Set Layer0 = AcadDoc.Layers.Item(0)

Set Layer1 = AcadDoc.Layers.Add("轮廓线层")

Set Layer2 = AcadDoc.Layers.Add("中心线层")

Set Layer3 = AcadDoc.Layers.Add("剖面线层")

Set Layer4 = AcadDoc.Layers.Add("标注层")

Layer1.Lineweight = acLnWt080 '轮廓线的线宽

Layer1.Color = acWhite

Layer2.Color = acRed

Layer3.Color = acCyan

Layer4.Color = acBlue

'线型

On Error Resume Next

Dim entry As AcadLineType

Dim found As Boolean

found = False

For Each entry In AcadDoc.Linetypes

If StrComp(entry.Name, "centerx2", 1) = 0 Then

found = True

Exit For

End If

Next

If Not (found) Then AcadDoc.Linetypes.Load "centerx2", "acad.lin"

Layer2.Linetype = "centerx2"

End Sub

Sub Ch4_ActiveLayer_Click( )

'设为当前层

If Combo1.Text = "中心线层" Then

AcadDoc.ActiveLayer = Layer2

ElseIf Combo1.Text = "轮廓线层" Then

AcadDoc.ActiveLayer = Layer1

ElseIf Combo1.Text = "剖面线层" Then

AcadDoc.ActiveLayer = Layer3

ElseIf Combo1.Text = "标注层" Then

AcadDoc.ActiveLayer = Layer4

Else:AcadDoc.ActiveLayer = Layer0

End If

End Sub



调出AutoCAD后，先单击“设置图层及其颜色、线型、线宽”按钮，若再单击“查找图层”按钮，即出现《查找图层》的消息框，如图4-3所示；若再单击“查找线型”

按钮，即出现《查找线型》的消息框，如图4-4所示。

图4-3《查找图层》的消息框

图4-4《查找线型》的消息框

5.1 创建图形对象

创建图形对象的方法：

(1) Document对象的SendCommand方法。

(2) 添加图形对象集合或块对象的成员，应用Add 类型的方法。

5.1.1创建点

1．点样式和大小

点的样式和大小是可以设置的。与“点样式”对话框相对应，共有20种样式。如图5-1所示。

图5-1点样式

PDMODE 和 PDSIZE 系统变量控制点对象的外观。

PDMODE 的值为 0、2、3 和 4 时，指定通过点绘制的图形。值为 1 时表示不显示任何图形。

PDSIZE 控制点图形的大小（PDMODE 的值为 0 和 1 时除外）。如果设置为 0，则点图形的高度是图形区高度的 5%。PDSIZE 正值指定点图形的绝对大小，负值将解释为视口大小的百分比。重生成图形时将重新计算所有点的大小。 用户修改 PDMODE 和 PDSIZE 之后，现有点的外观会在下次重生成图形时改变。 

要设置 PDMODE 和 PDSIZE，应使用 SetVariable 方法。

2．创建点

创建点要利用AddPoint方法。

例5-1：在模型空间中创建坐标为 (5,5,0) 的点对象，然后更新 PDMODE 和 PDSIZE 系统变量。 

Sub Ch5_CreatePoint( )

Dim pointObj As AcadPoint

Dim location(0 To 2) As Double

' 定义点的位置

location(0) = 5#: location(1) = 5#: location(2) = 0#

' 创建点

Set pointObj = ThisDrawing.ModelSpace.AddPoint(location)

ThisDrawing.SetVariable "PDMODE", 34

ThisDrawing.SetVariable "PDSIZE", 1

ZoomAll

End Sub

5.1.2 创建直线

直线是最基本的图形对象之一。广义的直线系指单一直线、多段线和多线。本节特指创建单一直线。

利用AddLine方法通过两点创建直线，该方法需要两个参数：起始点和终了点。

例5-2：在模型空间创建一条两点间的直线。

Sub Ch5_AddLine( )

Dim stP(0 to 2) As Double

Dim enP(0 to 2) As Double

Dim Line1 As AcadLine

stP(0)=50:stP(1)=70:stP(2)=0

enP(0)=80:enP(1)=80:enP(2)=0

Set Line1= ThisDrawing. ModelSpace.AddLine(stP,enP)

End Sub

5.1.3创建圆、圆弧和椭圆

1．利用AddCircle方法创建圆 

该方法需要两个参数：圆心坐标阵列和半径。

例5-3：在模型空间创建一个圆。

Sub Ch5_AddCircle( )

Dim cob1 As AcadCircle

Dim cp1(0 To 2) As Double

Dim r1 As Double

cp1(0) = 50: cp1(1) = 90: cp1(2) = 0

r1 = 20

Set cob1 = ThisDrawing.ModelSpace.AddCircle(cp1, r1)

End Sub

2． 利用AddArc方法创建圆弧

该方法需要四个参数：

（1） 圆心坐标阵列。

（2） 圆弧半径。

（3） 圆弧起始角（弧度）。

（4） 圆弧终止角（弧度）。

例5-4：在模型空间创建一段圆弧。

Sub Ch5_AddArc( )

Dim cp(0 To 2) As Double

Dim rr As Double

Dim startAng As Double

Dim endAng As Double

Dim arcObj As AcadArc

cp(0) = 30: cp(1) = 80: cp(2) = 0

r r= 25

startAng = 0: endAng = 150 * 3.14159 / 180

Set arcObj = ThisDrawing.ModelSpace.AddArc _

(cp, rr, startAng, endAng)

End Sub

3． 利用AddEllipse方法创建椭圆

该方法需要三个参数：

（1）椭圆中心坐标矩阵。

（2）主轴上的一点。

（3）椭圆两轴的半径比例（小于1）。

例5-5：创建一个椭圆。

Sub Ch5_AddEllipse( )

Dim EllObj As AcadEllipse

Dim CenPoint(0 To 2) As Double

Dim MajAxisPoint(0 To 2) As Double

Dim Radt As Double

CenPoint(0) = 100: CenPoint(1) = 100: CenPoint(2) = 0

MajAxisPoint(0) = 130

MajAxisPoint(1) = 100

MajAxisPoint(2) = 0

Radt = 0.6

Set EllObj = ThisDrawing.ModelSpace.AddEllipse(CenPoint, _ MajAxisPoint, Radt)

ZoomExtents 

End Sub

5.1.4创建多线

多（重）线是一种由多条平行线段组成的组合图象对象。用AddMline方法创建，该方法只需要的一个参数：多线的三维顶点坐标矩阵。

例5-6：创建一条多线。

Sub Ch5_AddMLine( )

Dim MLineObj As AcadMLine

Dim point(0 To 11) As Double

point(0) = 102: point(1) = 9: point(2) = 0

point(3) = 122: point(4) = 15: point(5) = 0

point(6) = 130: point(7) = 60: point(8) = 0

point(9) = 115: point(10) = 75: point(11) = 0

Set MLineObj = ThisDrawing.ModelSpace.AddMLine(point)

End Sub

5.1.5创建及编辑多段线

多段线是由多条线段组成的对象，这些线段可以是直线段，也可是圆弧段。

1． 创建多段线

创建多段线可用AddLightWeightPolyline方法和AddPolyline方法。其中AddPolyline方法可以创建二维或三维多段线，其参数是一个多段线顶点的三维坐标矩阵。而AddLightWeightPolyline方法能创建二维优化多段线（旧版为“轻便多段线”），它的参数是优化多段线顶点的二维坐标矩阵（创建时不能指定某段的宽度和凸度）。

注：单一直线和多线都是在世界坐标系（WCS）中的XY平面上创建，而多段线则是在对象坐标系（OCS）上创建。

例5-7：创建优化多段线。

Sub Ch5_AddLightWeightPolyline( )

Dim MyPline As AcadLWPolyline

Dim Vpoints(0 to 7) As Double

Vpoints(0)=10: Vpoints(1)=65 

Vpoints(2)=10: Vpoints(3)=80

Vpoints(4)=30: Vpoints(5)=80

Vpoints(6)=45: Vpoints(7)=80

Set MyPline=ThisDrawing.ModelSpace. _

AddLightWeightPolyline(VPoints)

MyPline.SetWidth 2,5,0 ′编辑线宽——画箭头

MyPline.Update

ThisDrawing.Application.ZoomAll

End Sub

2． 编辑多段线

二维和三维多段线、矩形、多边形和三维多边形网格都是多段线的变化形式而且编辑方式也都相同。 

要编辑多段线，应使用 LightweightPolyline 或 Polyline 对象的特性和方法。 

编辑多段线的常用方法和属性如表5-1中所述。

表5-1编辑多段线的常用方法和属性

方法和属性 描述 

方法 

AddVertex 增加一个顶点（只适用于优化多段线） Explode 炸开多段线 GetBulge 返回某给定序号段的凸度 GetWidth 返回某一段的线宽 Offset 对多段线执行偏移操作 SetWidth 设置给定序号段的开始线宽和结束线宽 SetBulge 设置给定序号段的凸度 属性 Closed 封闭多段线 凸度是表示弧线几何特征的一个指标。SetBulge方法可在给定的索引位置设定多段线的凸度。

语法：object.SetBulge Index,Value

其中：

Object——LightweightPolyline, Polyline

Index——以0开始的正整数；要设定的顶点的索引位置序号

Value——双精度浮点数；给定索引位置上的顶点凸度值

说明：凸度是在多段线顶点显示中，选取顶点与下一个顶点形成的弧之间角度四分之一的正切值。0表示直线，1表示半圆。凸度有正、负之分，当弧线是按反时针方向绘制的，凸度的值为“＋”；反之为“－”。

图5-2多段线的凸度

图5-2为一条由p1-p2-p3-p4构成的多段线，其中第二段为弧线：弧线弦长为d、拱高为h、 顶点序号（从0开始计）为1（p2点）。则该段的凸度b= -h/(d/2)。本简图从P2到P3的弧线是按顺时针方向绘制的，因此凸度b的计算式应加上“－”号。

例5-8：如图5-2所示，先创建一条皆由直线段组成的优化多段线p1-p2-p3-p4，其各顶点坐标为p1（100，100）、p2（150，100）、p3（170，60）、p4（150，45），再按拱高h=18将p2-p3直线段改为圆弧。

Sub Ch5_EditPolyline( )

Dim Vert(0 To 7) As Double

Dim LWP As AcadLWPolyline

Dim h As Double

Dim d As Double

Dim b As Double

Vert(0) = 100: Vert(1) = 100: Vert(2) = 150: Vert(3) = 100

Vert(4) = 170：Vert(5) = 60: Vert(6) = 150: Vert(7) = 45

Set LWP = ThisDrawing.ModelSpace. _

AddLightWeightPolyline(Vert)

LWP.Update

'用SetBulge方法编辑优化多段线的第二段

MsgBox "编辑多段线的第二段", 0, "应用SetBulge方法例"

h = 18

d = Sqr((170 － 150) ^ 2 + (60 － 100) ^ 2)

b = －h / (d / 2)

Call LWP.SetBulge(1, b)

LWP.Update

End Sub

5.1.6创建样条曲线

创建二次或三次NURBS（不均匀有理B-Spline）曲线需要使用AddSpline方法。

语法：RetVal = object.AddSpline(PointsArray, StartTangent, EndTangent)

式中：

object——模型空间、图纸空间或块

PointsArray——顶点的坐标矩阵（3D WCS坐标值的表），至少需要二个点

StartTangent——样条曲线起点的切点坐标（只可输入3D向量）

EndTangent——样条曲线终点的切点坐标（只可输入3D向量）

RetVal——新建的Spline对象

预设创建的样条曲线是首尾开放的。若需要为闭合的，在创建样条曲线后，再应用closed属性使其闭合。

例5-9：使用三点在模型空间绘制样条曲线，其中样条曲线起始相切点为（2，8，0），终点的切点为（3，1，0）。

Sub Ch5_CreateSpline( )

Dim anObj As AcadSpline

Dim VetPoints(0 To 8) As Double

Dim Stan(0 To 2) As Double

Dim Etan(0 To 2) As Double

VetPoints(0) = 16: VetPoints(1) = 90:VetPoints(2)=0

VetPoints(3)=48:VetPoints(4)= 120:VetPoints(5) = 0 

VetPoints(6) = 100: VetPoints(7) = 70: VetPoints(8) = 0

Stan(0) = 2: Stan(1) = 8: Stan(2) = 0

Etan(0) = 3: Etan(1) = 1: Etan(2) = 0

Set anObj =ThisDrawing.ModelSpace.AddSpline _

(VetPoints, Stan, Etan)

End Sub

5.1.7创建实体填充

实体填充对象是一种可以有添充颜色的对象，它可以是三角形，也可以是四边形。可以先设系统变量FILEMODE为0，关闭填充模式，创建填充对象的边界轮廓。最后再设系统变量FILEMODE为1，打开填充模式。可用两种方法创建实体填充对象：

1． 使用AddSolid方法

该方法需要四个参数，这四个参数组成填充实体的四个点坐标矩阵。第一点和第二点定义区域的一个边，第三点定义为第二点的对角点，创建四边形实体填充对象。如果将第四点定义为第二点的对角点，则创建的是两个对顶的三角形实体填充对象。如果第三点与第四点重合，则会创建一个三角形实体填充对象。

例5-10：创建一个四边形实体填充对象。

Sub Ch5_CreateSolid( )

Dim SolidObj1 As AcadSolid

Dim SolidObj2 As AcadSolid

Dim point1(0 To 2) As Double

Dim point2(0 To 2) As Double

Dim point3(0 T o 2) As Double

Dim point4(0 To 2) As Double

point1(0) = 0: point1(1) = 0: point1(2) = 0

point2(0) = 25: point2(1) = 0: point2(2) = 0

point3(0) = 0: point3(1) = 18: point3(2) = 0

point4(0) = 25: point4(1) = 18: point4(2) = 0

′设置系统变量FILLMODE为1(实心填充)

ThisDrawing.SetVariable "FILLMODE", 1

Set SolidObj1 = ThisDrawing .ModelSpace.AddSolid _

(point1, point2, point4, point3)

SolidObj1.Color = 6 ' 紫色

End Sub

2． 使用AddTrace方法

该方法的作用与AddSolid方法完全一样，四个顶点确定四条边的原则也与AddSolid方法相同。其不同点是该方法只需要一个参数——构成四个顶点的三维坐标数组。

例5-11：使用AddTrace方法绘制一个三角形填充实体对象。

Sub Ch5_AddTrace( )

Dim traceObj As AcadTrace

Dim tracePts(0 To 11) As Double

tracePts(0) = 100: tracePts(1) = 0: tracePts(2) = 0

tracePts(3) = 100: tracePts(4) = 10: tracePts(5) = 0

tracePts(6) = 110: tracePts(7) = 0: tracePts(8) = 0

tracePts(9) = 110: tracePts(10) = 0: tracePts(11) = 0

'设置系统变量FILEMODE为1(实心填充)

ThisDrawing.SetVariable "FILLMODE", 1

Set traceObj = ThisDrawing.ModelSpace.AddTrace(tracePts)

traceObj.Color = acGreen '绿色

End Sub

5.1.8面域

面域是用户从闭合形状（称为环）创建的二维闭合区域。环可以是一条曲线或一系列相连的曲线，这些曲线用不自交的边界在平面上定义一个区域。环可以是直线、优化多段线、圆、圆弧、椭圆、椭圆弧、样条曲线、三维面、宽线和实体的组合。组成环的对象必须是闭合的，或者是通过与其他对象共享端点而形成闭合的区域。所有这些对象还必须共面（在同一个平面上）。组成面域的环必须定义为对象的数组。

1．创建面域

创建面域要用AddRegion方法，该方法只需要一个参数：形成面域回路的图形对象数组。在面域建立以后，用于建立面域的图形对象可保留或删除（依赖于系统变量DELOBJ的值，0为不保留；1为保留）。

例5-12: 创建面域。

Sub Ch5_CreateRegion( )

' 定义保存面域边界的数组

Dim curves(0 To 0) As AcadCircle

' 创建圆，用作面域的边界

Dim center(0 To 2) As Double

Dim radius As Double

center(0) = 2

center(1) = 2

center(2) = 0

radius = 5#

Set curves(0) = ThisDrawing.ModelSpace.AddCircle (center, radius) 

' 创建面域

Dim regionObj As Variant

regionObj = ThisDrawing.ModelSpace.AddRegion(curves)

ZoomAll

End Sub

2．组合面域

对已经存在的面域进行差、并或交的布尔运算可创建新的组合面域。

（1） 差运算——使用被减面域的Boolean方法，第一个参数为acSubtraction，第二个参数为减去的面域。

（2） 并运算——可以使用任何一个参与运算面域对象的Boolean方法，第一个参数为acUnion，第二个参数为参与并运算的其它面域。

（3） 交运算——可以使用任何一个参与交运算面域对象的Boolean方法，第一个参数为acIntersection方法，第二个参数为参与交运算的其它面域。

例5-13：生成两个简单面域，并进行差运算创建组合面域。

Sub Ch5_CreateCompositeRegions( )

Dim border As AcadCircle

Dim cp(0 To 2) As Double

Dim rr As Double

cp(0) = 30: cp(1) = 15: cp(2) = 0

rr = 7

Set border = ThisDrawing.ModelSpace.AddCircle(cp,rr)

Dim RObject As Variant

Dim PObject As Variant

Dim curves1(0 To 0) As Object

Set curves1(0) = border

RObject = ThisDrawing.ModelSpace.AddRegion(curves1)

Dim polyobject As AcadLWPolyline

Dim vetP(0 To 7) As Double

vetP(0) = 5: vetP(1) = 25: vetP(2) = 45: vetP(3) = 25

vetP(4) = 45: vetP(5) = 5:vetP(6) = 5: vetP(7) = 5

Set polyobject = ThisDrawing.ModelSpace .AddLightWeightPolyline(vetP)

polyobject.Closed = True

Dim curves2(0 To 0) As AcadEntity

Set curves2(0) = polyobject

PObject = ThisDrawing.Modelspace.ModelSpace. AddRegion(curves2)

PObject(0).Boolean acSubtraction, RObject(0)

PObject(0).Update

End Sub

5.1.9图案填充

图案填充可以使用某种图案来填充图形中的指定区域。创建图案填充区域的过程通常是先用AddHatch方法创建Hatch对象，然后确定该区域的边界，最后执行Evaluate方法，先计算后才能显示。

1． 创建Hatch对象

创建Hatch对象要使用AddHatch方法。该方法有四个参数：依次为图案类型、图案名称、关联性和图案填充对象类型。

语法：

RetVal=object.AddHatch(PatternType, PatternName, Associativity [, HatchObjectType])

参数：

（1） PatternType——图案类型

该参数分为AcPatternType和AcGradientType两种类型。如果HatchObjectType参数的值为AcHatchObject，则该参数为AcPatternType类型；如果HatchObjectType参数的值为AcGradientObject，则该参数为AcGradientPatternType类型。

图案类型的常数按以下三种选用：

●AcHatchPatternTypePredefined——从AutoCAD提供的acad.pat文件中选择图案名称。

●AcHatchPatternTypeUserDefined——用当前线型定义直线图案。

●AcHatchPatternTypeCustomDefined——从用户定义的PAT文件中选择图案名称（而不是从acad.pat文件中选择）。

（2） PatternName——图案名称

该名称在所选的图案类型中必须是有效的名称。例如：当选用AcHatchPatternTypePredefined图案类型时，可选用“ANSI31”。

（3） Associativity——关联性

要创建与其边界相关联的关联性图案填充，应将其第三个参数Associativity设为TRUE；若创建非关联性图案填充，该值为FALSE。

（4） HatchObjectType——图案对象类型；可选

自AutoCAD 2004版起增添了此项可选性的更改。该参数的缺省值为AcHatchObject。如果图案类型参数PatternType选用AcGradientPatternType类型，则该参数为AcGradientObject，并且图案名称PatternName将含有渐变的图案名称。

2． 定义区域边界

区域的边界可以是直线、弧、圆、二维多段线、椭圆、样条曲线和面域，以及它们的任意组合。

第一个追加的边界必须是最外侧的边界，需要使用Hatch对象的AppendOuterLoop方法，它定义了图案填充的最外面范围。在此之后可以继续追加内部边界，这需要使用AppendInnerLoop方法。

AppendOuterLoop方法和AppendInnerLoop方法均需要一个参数，该参数是构成边界回路的图形对象数据阵列。

内边界定义图案填充内的孤岛。Hatch 对象如何处理这些孤岛，由该对象的 HatchStyle 属性决定。正常情况下该属性设为acHatchStyleNormal。

注：可使用InsertLoopAt方法在多处生成图案填充。该方法需要三个参数：构成封闭边界的顶点索引号（从0开始）、定义边界类型的常数、构成边界回路的图形对象数据阵列。

3．编辑图案填充 

用户可以更改现有填充图案的角度或间距，或者将其替换为实体填充或 AutoCAD 提供的预定义图案。使用以下属性和方法来编辑填充图案： 

（1）PatternAngle 指定填充图案的角度。

（2）atternDouble 指定用户定义的图案填充是否双向填充。 

（3）PatternName 指定填充图案的名称（不改变图案类型）。 

（4）PatternScale 指定填充图案的比例。

（5）PatternSpace 指定用户定义的填充图案间距。

（6）SetPattern 设置图案填充的图案名称和图案类型。 

例 5-14：在模型空间创建一个颜色深浅渐变的图案填充图。

Sub Ch5_AddCreateHatch1( )

Dim hatchObj As AcadHatch

Dim patternName As String

Dim PatternType As Long

Dim bAssociativity As Boolean

' 定义图案填充

patternName = "CYLINDER"

PatternType = acPreDefinedGradient 

bAssociativity = True

' 创建关联的 Hatch 对象

Set hatchObj = ThisDrawing.ModelSpace.AddHatch(PatternType, _

patternName, bAssociativity, acGradientObject)

Dim col1 As AcadAcCmColor, col2 As AcadAcCmColor

Set col1 = AcadApplication.GetInterfaceObject("AutoCAD.AcCmColor.16")

Set col2 = AcadApplication.GetInterfaceObject("AutoCAD.AcCmColor.16")

Call col1.SetRGB(255, 0, 0)

Call col2.SetRGB(0, 255, 0)

hatchObj.GradientColor1 = col1

hatchObj.GradientColor2 = col2

' 创建图案填充的外边界

Dim outerLoop(0 To 0) As AcadEntity

Dim center(0 To 2) As Double

Dim radius As Double

center(0) = 3: center(1) = 3: center(2) = 0

radius = 1

Set outerLoop(0) = ThisDrawing.ModelSpace.AddCircle(center, radius)

'显示图案填充

hatchObj.AppendOuterLoop (outerLoop)

hatchObj.Evaluate

ThisDrawing.Regen True

End Sub

例5-15：创建图案填充综合例,包括改变填充图案比例和图案填充样式等。

Private Sub addhatch_Click()

Dim hatchObj As AcadHatch

Dim patternName As String

Dim PatternType As Long

Dim bAssociativity As Boolean

patternName = "ANSI31"

PatternType = 0

bAssociativity = True

'创建Hatch对象

Set hatchObj = AcadDoc.ModelSpace.addhatch _

(PatternType, patternName, bAssociativity)

'用一段圆弧和一条直线构成外边界

Dim outerLoop(0 To 1) As AcadEntity

Dim center(0 To 2) As Double

Dim radius As Double

Dim startAngle As Double

Dim endAngle As Double

center(0) = 50: center(1) = 30: center(2) = 0

radius = 30

startAngle = 0

endAngle = 3.141592

Set outerLoop(0) = AcadDoc.ModelSpace.AddArc _

(center, radius, startAngle, endAngle)

Set outerLoop(1) = AcadDoc.ModelSpace.AddLine _

(outerLoop(0).startPoint, outerLoop(0).endPoint)

'追加外边界

hatchObj.AppendOuterLoop (outerLoop)

'追加内边界

Dim innerLoop1(0) As AcadEntity

center(0) = 50: center(1) = 45: center(2) = 0

radius = 10

Set innerLoop1(0) = AcadDoc.ModelSpace.AddCircle(center, radius)

hatchObj.AppendInnerLoop (innerLoop1)

'计算和显示

hatchObj.Evaluate

'插入圆作为第二个外边界

Dim LoopObj(0) As AcadEntity

radius = 5

Set LoopObj(0) = AcadDoc.ModelSpace.AddCircle(center, radius)

hatchObj.InsertLoopAt hatchObj.NumberOfLoops,_ 

acHatchLoopTypeDefault, LoopObj

'计算和显示

hatchObj.Evaluate

hatchObj.Update

'查找图案比例

Dim PatternScale As Double

PatternScale = hatchObj.PatternScale

MsgBox "现在的图案比例是: " & hatchObj.PatternScale, , _

"PatternScale 例"

'改变图案比例

hatchObj.PatternScale = PatternScale * 2

hatchObj.Evaluate

hatchObj.Update

MsgBox "现在的图案比例是: " & hatchObj.PatternScale, , _

"PatternScale 例"

hatchObj.color = 6

hatchObj.Update

AcadDoc.Regen True

'设置 hatchstyle 属性为 ：标准样式“normal”

hatchObj.HatchStyle = acHatchStyleNormal

hatchObj.Evaluate

AcadDoc.Regen True

MsgBox "将 HatchStyle 属性设置为:" & Choose _

(hatchObj.HatchStyle + 1, "acHatchStyleNormal", _

"acHatchStyleOuter", "acHatchStyleIgnore"), , "HatchStyle 例"

'设置 hatchstyle 属性为 ：忽略内部结构样式“ignore”

hatchObj.HatchStyle = acHatchStyleIgnore

hatchObj.Evaluate

AcadDoc.Regen True

MsgBox "将 HatchStyle 属性设置为:" & Choose _

(hatchObj.HatchStyle + 1, "acHatchStyleNormal", _

"acHatchStyleOuter", "acHatchStyleIgnore"), , "HatchStyle 例"

End Sub

5.1.10表

自AutoCAD 2005版起，增加了表的功能，到AutoCAD2007版又有改进。要创建Table对象，需使用AddTable方法。该方法可按当前的表样式生成一个空表。

语法：RetVal=object.AddTable(InsertionPoint,NumRows,NumColumns,RowHeight,ColWidth)

式中：

Object——ModelSpace,PaperSpace,Block

InsertionPoint——表的插入点 

NumRows——长整型；表的行数 

NumColumns——长整型；表的列数 

RowHeight——双浮点型；每行的高度

ColWidth——双浮点型；每列的宽度 

RetVal——Table对象

创建Table对象后,可采用该对象的SetText属性在指定的单元格插入文本。

语法：object.SetText row, col, pStr

式中：

Object——Table对象

Row——长整型；单元格的行序号（从0开始）

Col——长整型；单元格的列序号（从0开始）

pStr——字符串；文字

缺省的表样式名为 ”Standard”，可通过改变当前的表标准样式为需要的新表样式。然后创建需要的表。具体的方法请参阅下面的实例。

例5-16：通过改变当前的表标准样式为需要的新表样式，然后创建一个空表，并在指定的单元格插入文本。

Sub Ch5_CreateTableStyle( )

‘该例创建一个新表样式并设置其名称及其它属性

'设置新表样式属性：从下到上排列、取消表标题、数据表格的颜色等

Dim dictionaries As AcadDictionaries

Set dictionaries = ThisDrawing.Database.dictionaries

Dim dictObj As AcadDictionary

Set dictObj = dictionaries.Item("acad_tablestyle")

' 在字典中创建用户的 TableStyle 对象

Dim keyName As String

Dim className As String

Dim customObj As AcadTableStyle

keyName = "Standard"

className = "AcDbTableStyle"

Set customObj = dictObj.AddObject(keyName, className)

customObj.Name = "NewStandard"

customObj.Description = "New Style for My Tables"

customObj.FlowDirection = acTableBottomToTop

customObj.HorzCellMargin = 0.22

customObj.BitFlags = 0

customObj.SetTextHeight AcRowType.acDataRow + AcRowType.acTitleRow + AcRowType.acHeaderRow, 4.5

Dim col As AcadAcCmColor

Set col = AcadApplication.GetInterfaceObject("AutoCAD.AcCmColor.17")

Call col.SetRGB(12, 23, 45)

customObj.SetBackgroundColor AcRowType.acDataRow + AcRowType.acHeaderRow + AcRowType.acTitleRow, col

customObj.SetBackgroundColorNone AcRowType.acDataRow + AcRowType.acHeaderRow, True

customObj.SetGridVisibility AcGridLineType.acHorzInside + AcGridLineType.acHorzTop _

, AcRowType.acDataRow + AcRowType.acHeaderRow, True

customObj.SetAlignment AcRowType.acDataRow + AcRowType.acHeaderRow, acBottomCenter

col.SetRGB 244, 0, 0

customObj.SetGridColor 3, 1, col

customObj.TitleSuppressed = True

MsgBox "表样式名称 = " & customObj.Name & vbCrLf & _

"表样式描述= " & customObj.Description & vbCrLf & _

"表格排列方向 = " & customObj.FlowDirection & vbCrLf & _

"单元格边距 = " & customObj.HorzCellMargin & vbCrLf & _

"特征位 = " & customObj.BitFlags & vbCrLf & _

"数据行文字高 = " & customObj.GetTextHeight(acDataRow) & vbCrLf & _

"数据边框可见性 = " & customObj.GetGridVisibility(acHorzBottom, acDataRow) & vbCrLf & _

"数据行对齐方式 = " & customObj.GetAlignment(acDataRow) & vbCrLf & _

"不包含标题 = " & customObj.TitleSuppressed

End Sub

Sub Ch5_AddTable( )

'在模型空间创建一个表

Dim MyModelSpace As IAcadModelSpace

Set MyModelSpace = ThisDrawing.ModelSpace

Dim pt(2) As Double

Dim MyTable As AcadTable

Set MyTable = MyModelSpace.AddTable(pt, 5, 5, 10, 20)

MyTable.SetText 0, 0, "序号"

MyTable.SetText 0, 1, "图号"

MyTable.SetText 0, 2, "名称"

MyTable.SetText 0, 3, "数量"

MyTable.SetText 0, 4, "材料"

MyTable.SetText 1, 0, "1"

MyTable.SetText 2, 0, "2"

MyTable.SetText 3, 0, "3"

MyTable.SetText 4, 0, "4"

ZoomExtents

End Sub

注：该例适用于AutoCAD 2007版。

5.2 图形对象的公共属性和通用的编辑方法

5.2.1图形对象的公共属性

所谓图形对象的“公共属性”即所有图形对象均具有的属性。常用的公共属性见表5-2。

属性 功能 Application 返回应用程序对象 TrueColor 设置或查询实体的颜色 Document 返回实体所在的图形文件对象 Handle 返回实体的句柄 HasExtensionDictionary 查询实体是否有扩展记录 Hyperlinks 返回实体的超链接集合 Layer 设置或查询实体的层 Linetype 设置或查询实体的线型 LinetypeScale 设置或查询实体的线型比例 Lineweight 设置或查询实体的线重 ObjectID 返回实体标识符 ObjectName 返回实体的AutoCAD类名 OwnerID 返回实体的上级对象标识符 PlotStyleName 设置或查询实体的出图样式名 Visible 设置或查询实体的可见性 表5-2常用的公共属性

只能查询却不能设置的属性称之为只读属性。

要编辑图形对象，一般可采用以下的两种方式：

（1）使用图形对象本身的方法和属性（必须先用变量“引用” 对象）。

（2）使用图形对象的SendCommand方法调用AutoCAD命令。

5.2.2复制

使用图形对象的Copy方法可以对一个图形对象进行原位置复制，并返回该新图形对象。

注：使用Document对象的CopyObjects方法，可以同时对多个图形对象进行原位置复制。要将多个对象复制到不同的图形，须使用 CopyObjects 方法并将 Owner 参数设置为该图形的模型空间。 

5.2.3移动

使用Move方法用户可以沿指定的矢量移动所有的图形对象和属性参考对象。该方法需要两个参数：基点和目标点的坐标值。这两个坐标值定义了位移矢量，指示给定对象移动的距离和方向。

例5-17：先创建一个圆，用Copy方法进行原位置复制。然后将复制出的圆用Move方法移动并改为红色。

Sub Ch5_CopyMoveObject( )

Dim circleObj As AcadCircle

Dim center(0 To 2) As Double

Dim radius As Double

center(0) = 2#: center(1) = 2#: center(2) = 0#

radius = 0.5

Set circleObj = ThisDrawing.ModelSpace.AddCircle(center, radius)

ZoomAll

MsgBox "复制圆", , " Copy和Move示例"

' 复制圆

Dim copyCircleObj As AcadCircle

Set copyCircleObj = circleObj.Copy( )

' 确定构成移动矢量的两点

Dim point1(0 To 2) As Double

Dim point2(0 To 2) As Double

point1(0) = 0: point1(1) = 0: point1(2) = 0

point2(0) = 2: point2(1) = 0: point2(2) = 0

copyCircleObj.Move point1, point2

ZoomAll

MsgBox "移动及改变颜色已经完成！", , "Copy和Move示例"

End Sub

5.2.4旋转

利用图形对象的Rotate方法可将图形对象绕指定的基点旋转，改变对象的方位，但不改变其大小。该方法需要输入一个基点和一个旋转角度。基点是一个具有三个单精度数的变体数据矩阵（定义基点的三维WCS坐标）；旋转角度的单位是弧度。

例5-18：创建一条闭合的优化多段线，然后将该多段线绕基点 (4,4.25,0) 旋转 45 度。 

Sub Ch5_RotateObject( )

' 创建多段线

Dim plineObj As AcadLWPolyline

Dim points(0 To 11) As Double

points(0) = 1: points(1) = 2

points(2) = 1: points(3) = 3

points(4) = 2: points(5) = 3

points(6) = 3: points(7) = 3

points(8) = 4: points(9) = 4

points(10) = 4: points(11) = 2

Set plineObj = ThisDrawing.ModelSpace. AddLightWeightPolyline(points)

plineObj.Closed = True

ZoomAll

' 定义绕点 (4,4.25,0) 旋转45 度

Dim basePoint(0 To 2) As Double

Dim rotationAngle As Double

basePoint(0) = 4: basePoint(1) = 4.25: basePoint(2) = 0

rotationAngle = 0.7853981 ' 45 度

' 旋转多段线

plineObj.Rotate basePoint, rotationAngle

plineObj.Update

End Sub

5.2.5偏移

利用图形对象的Offset方法可执行图形对象的偏移操作。这些实体包括：弧、圆、椭圆、直线、二维优化多段线、多段线、样条曲线、构造线。执行“偏移”操作后原图形对象依然保留。该方法唯一参数是偏移距离，负数表示向“变小”的方向偏移（例如对于圆，是向圆内侧“偏移”）。

例5-19：创建一条优化多段线，然后偏移该多段线。 

Sub Ch5_OffsetObject( )

Dim pp1(0 To 5) As Double

Dim LWPline As AcadLWPolyline

pp1(0) = 250: pp1(1) = 200

pp1(2) = 280: pp1(3) = 200

pp1(4) = 290: pp1(5) = 170

Set LWPline = ThisDrawing.ModelSpace . AddLightWeightPolyline(pp1)

LWPline.Closed = True

Dim Offsetlwpline As Variant

Offsetlwpline = LWPline.Offset(5)

End Sub

5.2.6阵列

1.环形阵列——用ArrayPolar方法

该方法需要三个参数：

（1） 环形阵列创建的对象数目（大于1的正整数）。

（2） 阵列角度（非零数，单位为弧度，正值为逆时针）。

（3） 阵列的中心点坐标（包含三个数的Variant数组）。

注：ArrayPolar方法与AutoCAD的ARRAY命令的区别在于它不支持阵列操作时旋转对象。

例5-20：先创建一个圆，然后进行环形阵列。阵列数为6，均匀分布在圆周上。

Sub Ch5_ArrayingA( )

Const Pi As Double = 3.14159

Dim Ccd As AcadCircle

Dim Ccx As AcadCircle

Dim Rpoint(0 To 2) As Double

Dim Rx(0 To 2) As Double

Dim rrd As Double

Dim rrx As Double

Dim AAngle As Double

Dim ANumber As Integer

Rpoint(0) = 120: Rpoint(1) = 300: Rpoint(2) = 0

Rx(0) = 120: Rx(1) = 340: Rx(2) = 0

rrd = 40

rrx = 6

Set Ccd = ThisDrawing.ModelSpace.AddCircle(Rpoint, rrd)

Set Ccx = ThisDrawing.ModelSpace.AddCircle(Rx, rrx)

ANumber = 6

AAngle = (ANumber-1) * 2 * Pi / ANumber

Dim AP As Variant

AP = Ccx.ArrayPolar(ANumber, AAngle, Rpoint)

ThisDrawing.Regen acAllViewports

End Sub

（2） 矩形阵列——用ArrayRectangular方法

该方法需要四个参数，依次为:行数、列数、行距和列距。

注：1）若创建三维阵列，需要六个参数，依次为：行数、列数、层数、行距、列距和层距。

2）矩形阵列的行和列总是与坐标轴成正交的，可以通过SnapRotationangle属性来设置矩形阵列的倾斜角。

例5-21：创建一个圆，然后对该圆执行矩形阵列操作（创建 5 行 5 列2层）。 

Sub Ch5_ArrayRectangular( )

' 创建圆

Dim circleObj As AcadCircle

Dim center(0 To 2) As Double

Dim radius As Double

center(0) = 2#: center(1) = 2#: center(2) = 0#

radius = 0.5

Set circleObj = ThisDrawing.ModelSpace. AddCircle(center, radius)

ZoomAll

' 定义矩形阵列

Dim numberOfRows As Long

Dim numberOfColumns As Long

Dim numberOfLevels As Long

Dim distanceBwtnRows As Double

Dim distanceBwtnColumns As Double

Dim distanceBwtnLevels As Double

numberOfRows = 5

numberOfColumns = 5

numberOfLevels = 2

distanceBwtnRows = 1

distanceBwtnColumns = 1

distanceBwtnLevels = 1

' 创建对象的阵列

Dim retObj As Variant

retObj = circleObj.ArrayRectangular (numberOfRows, _

numberOfColumns, numberOfLevels, distanceBwtnRows, _

distanceBwtnColumns,distanceBwtnLevels)

ZoomAll

End Sub

5.2.7删除

可以使用对象本身的Delete方法删除单独的对象。该方法适用的对象除了全部的图形实体对象外，还有: Block, Dictionary, DimStyle, Group, Hyperlink, Layer, LayerStateManager, Layout, Linetype, PlotConfiguration, PopupMenuItem, PViewport, RegisteredApplication, SelectionSet, TextStyle, Toolbar, ToolbarItem, UCS, View, Viewport, XRecord 。

注：1）由于这些对象在类型库中的定义方式，ActiveX Automation 中的集合对象具有 Delete 方法。但是，不能删除集合对象（例如 ModelSpace 集合、Layers 集合和 Dictionaries 集合）。

2）当删除集合中的对象时，集合中所有剩余选项将依据当前的计数重新指定新的索引。

例5-22：创建一条优化多段线，然后将其删除。 

Sub Ch5_Delete( )

' 创建多段线

Dim lwpolyObj As AcadLWPolyline

Dim vertices(0 To 5) As Double

vertices(0) = 2: vertices(1) = 4

vertices(2) = 4: vertices(3) = 2

vertices(4) = 6: vertices(5) = 4

Set lwpolyObj = ThisDrawing.ModelSpace. AddLightWeightPolyline(vertices)

ZoomAll

' 删除多段线

lwpolyObj.Delete

ThisDrawing.Regen acActiveViewport

End Sub

5.2.8延伸和修剪

对象的延伸和修剪是通过修改对象相应的属性来实现。

对于直线，通过修改直线对象的StartPoint和EndPoint属性；对于圆弧，通过修改圆弧对象的StartAngle和EndAngle属性。当改变了对象的属性值后，应注意使用Update方法更新图形。

例5-23：创建一条直线，然后修改其端点拉长这条直线。 

Sub Ch5_LengthenLine( )

' 定义和创建直线

Dim lineObj As AcadLine

Dim startPoint(0 To 2) As Double

Dim endPoint(0 To 2) As Double

startPoint(0) = 0

startPoint(1) = 0

startPoint(2) = 0

endPoint(0) = 1

endPoint(1) = 1

endPoint(2) = 1

Set lineObj = ThisDrawing.ModelSpace. AddLine(startPoint, endPoint)

lineObj.Update

' 将端点更改为 4,4,4

' 拉长直线

endPoint(0) = 4

endPoint(1) = 4

endPoint(2) = 4

lineObj.endPoint = endPoint

lineObj.Update

End Sub

5.2.9分解

分解（炸开）对象就是将单一的组合对象转换成各组件间相互独立的一组对象。例如将一个块对象炸开后，就可以对它的每一部分分别进行编辑。要分解对象，需要使用对象的Explode方法。

例5-24：本例创建一个优化多段线对象，然后将多段线分解成多个对象。接着遍历产生的对象，显示含有每个对象名称的消息框，并显示分解对象在列表中的索引。

Sub Ch5_ExplodePolyline( )

Dim plineObj As AcadLWPolyline

Dim points(0 To 11) As Double

' 定义二维多段线的点

points(0) = 1: points(1) = 1

points(2) = 1: points(3) = 2

points(4) = 2: points(5) = 2

points(6) = 3: points(7) = 2

points(8) = 4: points(9) = 4

points(10) = 4: points(11) = 1

' 创建优化多段线

Set plineObj = ThisDrawing.ModelSpace. AddLightWeightPolyline(points)

' 在某个线段上设置凸度以改变

' 多段线中的对象类型

plineObj.SetBulge 3, -0.5

plineObj.Update

' 分解多段线

Dim explodedObjects As Variant

explodedObjects = plineObj.Explode

' 遍历分解的对象并以消息框来显示每个对象的类型

Dim I As Integer

For I = 0 To UBound(explodedObjects)

explodedObjects(I).Color = acRed

explodedObjects(I).Update

MsgBox "Exploded Object " & I & ": " & explodedObjects(I).ObjectName

explodedObjects(I).Color = acByLayer

explodedObjects(I).Update

Next

End Sub

5.2.10比例缩放

要按比例缩放来调整图形对象的大小，需要使用该对象的ScaleEntity方法，在X、Y和Z方向均匀地缩放。该方法需要两个参数：基点和比例系数。其中缩放基点为三维的WCS坐标（Variant阵列）。比例（缩放）系数为大于0的数。当大于1时，放大对象；0到1之间时，缩小对象。

例5-25：对已有优化多段线进行比例缩放。

Sub Ch5_ScalePolyline( )

' 复制多段线并移动优化多段线副本

Dim plineObjCopy As AcadLWPolyline

Dim MFrP(0 To 2) As Double

Dim MToP(0 To 2) As Double

Set plineObjCopy = plineObj.Copy

MFrP(0) = 0: MFrP(1) = 0: MFrP(2) = 0

MToP(0) = 50: MToP(1) = 0: MToP(2) = 0

PlineObjCopy.Move MFrP, MToP

' 定义比例缩放基点和比例系数

Dim BasePoint(0 To 2) As Double

Dim ScaleFactor As Double

BasePoint(0) = 50: BasePoint(1) = 65: BasePoint(2) = 0

ScaleFactor = 2

plineObjCopy.ScaleEntity BasePoint, ScaleFactor

plineObjCopy.Update

End Sub

5.2.11镜像

镜像可以相对镜像线对称地创建图形对象的副本。它需要该对象的Mirror方法（与AutoCAD中的命令不同，该方法会保留原来的对象，若要删除原对象，要用Erase方法）。Mirror方法需要输入构成两个线端点的坐标值。

当需要对文字和其它图形对象同时进行镜像操作时，往往不希望文字被镜像后产生反向，这时需要设置系统变量MIRRTEXT的值。MIRRTEXT的默认值为1，它使文字与其它对象同时反向；若MIRRTEXT的值为0，则文字的读写方向不会反向。

例5-26：镜像一段文字。

Sub Ch5_MirrorText( )

Dim textp1(0 To 2) As Double

Dim textheight1 As Double

Dim textstr1 As String

Dim textObject1 As AcadText

textp1(0) = 200

textp1(1) = 280

textp1(2) = 0

textheight1 = 20

textstr1 = "AutoCAD VBA ActiveX 二次开发"

Set textObject1 =ThisDrawing.ModelSpace.AddText(textstr1, textp1, textheight1)

' 镜像操作--让文字镜像反像或不反像

ThisDrawing.SetVariable "MIRRTEXT", 1 '重置默认值

Dim MIRRT As Integer

MIRRT =Val(MsgBox("若让文字镜像后不反向,请点击 _ 

" & vbCrLf & "若让文字镜像后反向,请点击", 4))

Dim sysVariableName As String

Dim sysVariableData As Variant

If MIRRT = 6 Then

ThisDrawing.SetVariable "MIRRTEXT", 1

ElseIf MIRRT = 7 Then

ThisDrawing.SetVariable "MIRRTEXT", 0

End If

sysVariableData = ThisDrawing.GetVariable("MIRRTEXT")

MsgBox "MIRRTEXT" & "=" & sysVariableData, 0, "镜像用的系统变量"

Dim st1(0 To 2) As Double

Dim st2(0 To 2) As Double

st1(0) = 200: st1(1) = 260: st1(2) = 0

st2(0) = 300: st2(1) = 260: st2(2) = 0

textobject1.Mirror st1, st2

textobject1.Update

End Sub

5.2. 12使用命名对象

在大多数情况下，是在创建对象时命名对象，以后再重命名。名称存储在符号表中。在指定命名对象时，将引用符号表中该对象的名称和关联数据。

1．清理命名对象 

在编辑期间，用户可以随时清理图形中无用的、未被引用的命名对象。清理可以缩小图形的大小。不能清理被其他对象引用的对象。例如，字体文件可能被文字样式引用，图层可能被图层上的对象引用。 

要清理图形，要使用 PurgeAll 方法，如：ThisDrawing.PurgeAll。

2．重命名对象

图形变得比较复杂时，用户可以重命名对象，使名称有针对性或者避免与用户插入主图形的其他图形中的名称相冲突。 

可以重命名任何命名对象，但 AutoCAD 默认的命名对象（例如图层 0 或 CONTINUOUS 线型）除外。 

名称最长可以包含 255 个字符。除了字母和数字以外，名称还可以包含空格（虽然 AutoCAD 会删除名称前后的空格）以及任何未被 Windows 操作系统或 AutoCAD 用作其他用途的特殊字符。不能使用的特殊字符，包括小于号和大于号 (<、>)、正斜杠和反斜杠 (/、\)、引号 (")、冒号 (:)、分号 (;) 、问号 (?)、逗号 (,)、星号 (*)、竖线 (|)、等号 (=) 和单引号 (')。也不能使用以 Unicode 字体创建的特殊字符。 

要重命名对象，要使用该对象的 Name 属性。

例5-27：本例创建名为“NewLayer”的图层，然后将其重命名为“MyLayer”。 

Sub Ch5_RenamingLayer( )

' 创建图层

Dim layerObj As AcadLayer

Set layerObj = ThisDrawing.Layers.Add("NewLayer")

' 更改图层的名称

layerObj.Name = "MyLayer"

End Sub

5.3 文字

5.3.1文字样式

绘制文字时，需要规定字体、字高、宽度比例和倾角等，这都需要设置不同的TextStyle（文字样式）对象。在TextStyles集合中，除系统自动生成的STANDARD文字样式外，用户可以修改预先定义好的文字样式，也可以用Add方法创建新的文字样式。要创建文字样式只需一个参数——“样式名称”。 样式名可以包含字母、数字和特殊字符，例如美元符 ($)、下划线 (_) 和连字符 (-)。AutoCAD 将字符转换为大写字符。如果用户不输入样式名，AutoCAD 会自动将其命名为 Stylen，其中 n 是从 1 开始的数字。每个新样式会加 1 来显示。文字样式创建之后，用户不能通过 AutoCAD ActiveX Automation修改其名称。

如果用户修改现有样式的字体或方向，所有使用该样式的文字均被修改为使用新的字体或方向。更改文字的高度、宽度比例和倾斜角不会改变现有的文字，但会应用到以后创建的文字对象上。

可通过修改TextStyle对象的属性来修改现有的文字样式。文字样式对象的属性如表5-3所述。

表5-3文字样式对象的属性

属性 预设值 描述 字体文件(FontFile) Txt.shx 指定与字体（字符样式）关联的文件 大字体文件 (BigFontFile) 无 指定适用于非ASCII字符集的特殊形定义文件 高度(Height) 0 指定字符高度（需要设置） 宽度(Width) 1 指定字符的宽度比例 倾斜角度(ObliqueAngle) 0 指定字符的倾斜度 文字生成 (TextGenerationFlag) No,No 指定左右反向、上下颠倒文字或两者同时兼有 注：

1） 要用ActiveTextStyle属性重置当前（激活的）文字样式。

2） 当修改字体后，所创建的文字都会被修改。

3） 当修改文字样式后，需要用Regen或Update重新生成对象才能看到修改的效果。

5.3.2设置字体

自AutoCAD 2000简体中文版起，支持TrueType字体和三种形字体（即中西文正体gbenor.shx、斜体gbeitc.shx和大字体gbcbig.shx）。若设置机械制图用的中西文混用的“工程字”，应侭量用形字体，否则特殊字易乱码。所有的AutoCAD SHX形字体都是Unicode字符。Unicode字体包含了（比定义在系统中的）更多的字符。对于无法直接从键盘输入的字符，就按输入“/U+nnnn”序列来实现。

TrueType字体（.ttf文件）是一种以实体填充的形式出现在图形文字中的字体，由Windows提供。

字体定义了构成每个字符集的文字字符的形状。一种字体可以被多个样式使用。可通过TextStyle对象的FontFile和BigFontFile属性指定字体文件；也可通过TextStyle对象的GetFont和SetFont方法获取当前文字样式的当前字体值及设置新的字体。

例5-28：先利用GetFont方法获取当前文字样式的当前字体值，将字体名改为"仿宋_GB2312"，然后使用 SetFont 方法设置新的字体。 

Sub Ch5_UpdateTextFont( )

' 用AddText方法写"ActiveX 您好"

Dim textp(0 To 2) As Double

Dim textheight As Double

Dim textstr As String

Dim textobject As AcadText

textp(0) = 100

textp(1) = 200

textp(2) = 0

textheight = 15

textstr = "ActiveX 您好"

Set textobject = ThisDrawing.ModelSpace.AddText(textstr, textp, textheight)

textobject.Update

MsgBox "注意看文字的字体"

Dim typeFace As String

Dim SavetypeFace As String

Dim Bold As Boolean

Dim Italic As Boolean

Dim charSet As Long

Dim PitchandFamily As Long

' 获取当前设置，填充SetFont 方法的默认值

ThisDrawing.ActiveTextStyle.GetFont typeFace, _

Bold, Italic, charSet, PitchandFamily

' 改变字体

SavetypeFace = typeFace

typeFace = "仿宋_GB2312"

ThisDrawing.ActiveTextStyle.SetFont typeFace, _

Bold, Italic, charSet, PitchandFamily

ThisDrawing.Regen acActiveViewport

MsgBox "注意看文字的字体的变化"

' 恢复原始字体

ThisDrawing.ActiveTextStyle.SetFont SavetypeFace, _

Bold, Italic, charSet, PitchandFamily

ThisDrawing.Regen acActiveViewport

End Sub

例5-29：通过更改文字样式对象的FontFile和BigFontFile属性设置工程汉字。

Sub Ch5_FontFileBigFontFile( )

Dim currFontFile As String

Dim newFontFile As String

Dim newbigFontFile As String

' 返回 FontFile 当前值

currFontFile = ThisDrawing.ActiveTextStyle.fontFile

MsgBox "FontFile 当前值为" & currFontFile, vbInformation, "设置工程汉字例"

' 改变 FontFile 和 bigfontFile 值

newFontFile = "C:/program files/AutoCAD 2004/Fonts/gbenor.shx"

newbigFontFile="C:/program files/AutoCAD 2004/Fonts/gbcbig.shx"

ThisDrawing .ActiveTextStyle.fontFile = newFontFile

MsgBox "FontFile新值为" & ThisDrawing.ActiveTextStyle.fontFile,_

vbInformation, "设置工程汉字例"

acadDoc.ActiveTextStyle.BigFontFile = newbigFontFile

MsgBox"bigFontFile新值为" & ThisDrawing.ActiveTextStyle.BigFontFile, _

vbInformation, "设置工程汉字例"

Dim textp(0 To 2) As Double

Dim textheight As Double

Dim textstr As String

Dim textobject As AcadText

textp(0) = 60

textp(1) = 70

textp(2) = 0

textheight = 5

textstr = "ActiveX 您好"

Set textobject = ThisDrawing.ModelSpace.AddText (textstr, textp, textheight)

textobject.Update

End Sub

5.3.3 Unicode字符和控制码

用户可以在一个字符串中使用Unicode字符和控制码来绘制一些特殊符号。如表5-4和表5-5所述。

表5-4 Unicode字符

Unicode字符 描述 \U+00B0 角度符号“°” \U+00B1 正负符号“±” \U+2205 直径符号“Ф” 

表5-5控制码

控制码 描述 %%o 切换打开或关闭顶线模式 %%u 切换打开或关闭底线模式 %%d 绘制角度符号“°” %%p 绘制正负符号“±” %%c 绘制直径符号“Ф” %%% 百分比符号“%” 5.3.4使用单行文字

1．创建单行文字

使用AddText方法创建Text对象需要输入三个参数：文字字符串、插入点和文字高度。字符串是要显示的实际文字，可以是 Unicode、控制码和特殊字符。插入点是包含三个双精度数的变量数组，这三个数代表图形中放置文字处的三维 WCS 坐标。文字高度是代表大写文字高度的正数，高度是以当前单位测量的。

例5-30：在模型空间中的 (20,20,0) 坐标处创建单行文字。 

Sub Ch5_CreateText( )

Dim textObj As AcadText

Dim textString As String

Dim insertionPoint(0 To 2) As Double

Dim height As Double

' 创建 Text 对象

textString = "Hello, AutoCAD ActiveX !"

insertionPoint(0) = 20

insertionPoint(1) = 20

insertionPoint(2) = 0

height = 0.5

Set textObj = ThisDrawing.ModelSpace. AddText(textString, insertionPoint, height)

textObj.Update

End Sub

2．格式化单行文字

Text对象是通过当前的文字样式创建的。用户可以通过修改文字样式或编辑Text对象的属性来修改单行文字的格式，但不能对其中的个别文字或字符格式化。

要改变与单个 Text 对象相关联的文字样式，需要将它的StyleName属性设置为新的文字样式。修改后，要使用 Text 对象的 Update 方法才能看到图形中的更改。

除了图形对象的公用可编辑属性（颜色、图层、线型等）外，用户还可以修改Text 对象的其它属性，这些属性包括：

（1）Alignment 指定文字的水平和垂直对齐方式。 

（2）InsertionPoint 指定文字的插入点。 

（3）ObliqueAngle 指定一个文字对象的倾斜角度。 

（4）Rotation 指定文字的旋转角度（以弧度为单位）。 

（5）ScaleFactor 指定文字的缩放比例。 

（6）TextAlignmentPoint 指定文字的对齐点。 

（7）TextGenerationFlag 指定文字是否要反向、倒置显示或两者兼有。 

（8）TextString 指定显示的实际字符串。 

修改属性之后，要用 Update 方法才能看到图形中的更改。

3．对齐单行文字

可以按多种水平或垂直方式来对齐单行文字。其中左下对齐是默认方式。可先创建一个单行文字和一个参考点对象，然后使用Alignment属性重新对齐文字。

例5-31：创建一个 Text 对象和一个 Point 对象。Point 对象被设置为文字对齐点，并变为红色的十字光标，使其可见。文字对齐将被修改，并显示消息框以终止宏的执行，这样用户就可以查看修改文字对齐的影响。 

Sub Ch5_TextAlignment( )

Dim textObj As AcadText

Dim textString As String

Dim insertionPoint(0 To 2) As Double

Dim height As Double

' 定义新的 Text 对象

textString = "Hello, World."

insertionPoint(0) = 3

insertionPoint(1) = 3

insertionPoint(2) = 0

height = 0.5

' 在模型空间中创建 Text 对象

Set textObj = ThisDrawing.ModelSpace. AddText(textString, insertionPoint, height)

' 在文字对齐点上创建一个点，以便更清楚地看到对齐过程

Dim pointObj As AcadPoint

Dim alignmentPoint(0 To 2) As Double

alignmentPoint(0) = 3

alignmentPoint(1) = 3

alignmentPoint(2) = 0

Set pointObj = ThisDrawing.ModelSpace. AddPoint(alignmentPoint)

pointObj.Color = acRed

' 将点样式设置为十字光标

ThisDrawing.SetVariable "PDMODE", 2

' 左对齐文字

textObj.Alignment = acAlignmentLeft

ThisDrawing.Regen acActiveViewport

MsgBox "左对齐"

' 居中对齐文字

textObj.Alignment = acAlignmentCenter

' 将文字对齐点（所有文字， 左对齐的文字除外）

textObj.TextAlignmentPoint = alignmentPoint

ThisDrawing.Regen acActiveViewport

MsgBox "居中对齐"

' 右对齐文字

textObj.Alignment = acAlignmentRight

ThisDrawing.Regen acActiveViewport

MsgBox "右对齐"

End Sub

4．编辑单行文字

和其它的图形对象一样，单行文字对象也可以进行下列的编辑操作：

（1）ArrayPolar 创建环形阵列。

（2）ArrayRectangular 创建矩形阵列。

（3）Copy 复制。

（4）Erase 删除。

（5）Mirror 镜像。 

（6）Move 移动。

（7）Rotate 旋转。

注：进行镜像操作时，如果不希望文字被反向，应将系统变量MIRRTEXT设为0。

5.3.5使用多行文字

所谓多行文字，是填入到指定的宽度区域内的由任何文字行或段落数目所组成的同一对象。可以同时对这一段文字进行移动、旋转、镜像、删除、复制、拉伸或调整比例。还可以对其中的个别文字使用不同的文字样式以及比单行文字更多的编辑选项。

1．创建多行文字

创建多行文字需要使用AddMText方法，这个方法需要三个参数，依次为：插入点坐标、多行文字宽度和文字字符串。

字符串是要显示的实际文字，可以是 Unicode、控制码和特殊字符。插入点是包含三个双精度数的变量数组，这三个数代表图形中放置文字处的三维 WCS 坐标。文字宽度是代表文字边框宽度的正数，宽度是以当前单位测量的。

创建 MText 对象之后，用户可以将文字高度、对正、旋转角度和样式应用到 MText 对象，或将字符格式应用到选定字符。

2．修改多行文字中个别文字的格式

用户可以编辑多行文字中个别文字的格式或特殊字符。常用的文字格式码如表5-6所示。

表5-6常用文字格式码

格式码 作用 输入字符 显示效果 \\ 插入反斜线 Autodesk\\ Autodesk\ \{…\} 插入开始和结束大括弧 A\{2000\} A{2000} \文件名 修改为指定的字体文件 Auto\Ftimes;CAD2004 \HvalueX 修改文字高度为当前文字高的倍数 ACAD\H2X;2 ACAD2 \S…^… 堆叠后续文字 120\S+0.01^-0.02 \Hvalue 修改文字高度值 \L…\l 打开和关闭底线 \L AutoCAD\l AutoCAD 例5-32：在多行文字中使用格式码 ，包括:打开和关闭底线、按比例修改字高、堆叠后续文字等。

Sub Ch5_FormatMText( )

' 定义控制字符的ASCII字符

Dim BS As Long

Dim TS As Long

Dim SC As Long

Dim LS As Long

Dim LE As Long

Dim XX As Long

Dim HH As Long

' 格式码的ASCII对应值

BS = Asc("\")

TS = Asc("S")

SC = Asc("^")

LS = Asc("L")

LE = Asc("l")

XX = Asc("X")

HH = Asc("H")

' 给字符串指定下列控制字符和文字字符

TextString = "零件材料为" + Chr(BS) + Chr(LS) _

+ "13CrMoV" + Chr(BS) + Chr(LE) + ",轴承内径%%C40" _

+ Chr(BS) + Chr(HH) + "0.5" + Chr(XX) + Chr(BS) _

+ Chr(TS) + "+0.02" + Chr(SC) + "-0.01。"

Set MtextObj = ThisDrawing.ModelSpace.AddMText _

(InsP, TextWidth, TextString)

End Sub

该通用过程的运行结果如图5-3所示。

图5-3 多行文字的创建与编辑

5.4 块和外部引用

使用块可以将多个对象作为一个部件进行组织和操作。块属性可以将信息项与图形中的块相关联。 

使用 AutoCAD 外部引用，可以将整个图形附着或覆盖到当前的图形中。打开当前图形时，外部引用图形中所做的任何更改都会出现在当前图形中。

5.4.1块操作

1.创建块对象

步骤：

（1）块定义：用块集合的Add方法，其参数为插入点和块名。

（2）向块中添加各图形对象：用Add类型的方法。

2. 插入块

使用InsetBlock方法，需要六个参数：插入点坐标（3D）、块名、X方向比例系数、Y方向比例系数、Z方向比例系数和旋转角度。

注：

1） 也可将已经存在的图形文件或用WBlock命令生成的块文件插入到图形中；

2） 插入后，块对象就成为块引用对象（或称“块参照”）（BlockRef）。块中含有的属性也就成为属性引用对象（或称“属性参照”）（AttributeRef）。

3） 也可用AddMInsertBlock方法将块以矩形阵列的方式插入——成为多重插入块对象MInsertBlock。

3．分解块参照

使用 Explode 方法可以炸开块参照。通过分解块参照，可以修改块或者添加或删除块参照的成员对象。

4．重定义块对象

要重新定义块对象，需要使用该对象本身的方法和属性。重定义块时，图形中对该块的所有参照也随着更新，以便反映新的定义。 

块重定义会影响以前和以后的块插入。固定属性会丢失，并且被新的固定属性所替换。即使新块不包含属性，可变属性也保持不变。 

5.4.2块属性

块属性是一种可以与块对象相关联的文字对象。当使用块对象时，它提供给用户一种交互式的标识或标签。

属性定义是与创建该属性定义所在的块相关联的。创建属性定义对象要使用块对象的AddAttribute方法，它需要六个参数，依次为：

（1） 属性文本高度。

（2） 属性模式。

acAttributeModeNormal 指定维持每个属性的当前模式

acAttributeModeInvisible（1）插入块时不显示属性值

acAttributeModeConstant（2）块在插入时属性值为常数

acAttributeModeVerify（4） 属性值是验证方式（提示用户确认属性值是否正确）

acAttributeModePreset（8） 属性值是预设方式，设置属性为其默认值（无法编辑）

模式参数可以为无（None）、上述五种常数的一个或任意组合。如果要指定选项组合，需要将各常数相加。

（3） 属性提示字符串prompt。

提示字符串是在插入带属性的块时出现的，该字符串的默认值是标签字符串。当属性模式为acAttributeModeConstant时，该参数值无效（不提示）。

（4） 属性文本插入点坐标。

（5） 标签字符串Tag。

标签字符串是提示字符串的默认值。用户可以使用除空格或感叹号外的任何字符。

（6） 默认属性值Value。

当插入块时，用户可以指定不同的值给属性参照。若未输入任何文字，AutoCAD将使用该默认属性值。

注：可以使用块属性对象的属性和方法来编辑属性，可以使用 GetAttributes 和 GetConstantAttributes 方法从图形中提取属性信息。

例5-33：创建带属性定义的粗糙度符号，并插入到模型空间。

Sub Ch5_InsertingABlockwithAttribute ( )

Dim myblock As AcadBlock ' 生成图块

Dim inp(0 To 2) As Double

inp(0) = 0: inp(1) = 0: inp(2) = 0

Set myblock = ThisDrawing.Blocks.Add (inp, "MYBwithAttribute")

Dim spe(0 To 7) As Double

spe(0) = 20: spe(1) = 0: spe(2) = 0: spe(3) = 0

spe(4) = 10: spe(5) = -17.32: spe(6) = 30: spe(7) = 17.32

Call myblock.AddLightWeightPolyline(spe)

' 将属性加入到块中

Dim attObj As AcadAttribute

Dim height As Double

Dim tag As String

Dim mode As Long

Dim prompt As String

Dim value As String

Dim insP(0 To 2) As Double

Dim BInsertPA(0 To 2) As Double

mode = acAttributeModeVerify

height = 8

prompt = "新值"

tag = "粗糙度"

insP(0) = 0: insP(1) = 2: insP(2) = 0

BInsertPA(0) = 120: BInsertPA(1) = 70: BInsertPA(2) = 0

value = "12.6"

Set attObj = myblock.AddAttribute(height, mode, prompt, insP, tag, value)

' 插入带属性的块

Dim blockRefObj As AcadBlockReference

Set blockRefObj = ThisDrawing.ModelSpace.InsertBlock

(BInsertPA, "MYBwithAttribute", 1, 1, 1, 0)

blockRefObj.Update

ZoomAll

End Sub

该通用过程的运行结果如图5-4所示。

图5-4插入带属性的块

5.4.3外部引用

外部引用对象又称为“外部参照”。它是一种特殊的块，所不同的是外部引用对象只是引用外部图形文件，当外部图形文件改变的时候，外部引用对象会自动更新。用户不能分解外部引用对象。外部引用不会改变图形数据库的内容，只在数据库中保留外部文件的有关信息。

1． 附着外部引用对象

附着外部引用对象可以将某个图形（参照文件或外部参照）链接到当前图形。当图形参照外部参照时，AutoCAD 仅将外部参照定义附着到图形。

要附着外部引用对象，需使用AttackExternalReference方法。

语法：AttackExternalReference(PathName,Name,InsertionPoint, _

XScale,YScale,ZScale,Rotation,bOverlay) 

其中：

PathName——指明要引用的对象的路径和文件名

Name——指明插入到文件中的引用对象名

InsertionPoint——指定外部文件的插入点

bOverlay——指明引用的是附着或是覆盖方式。在附着方式下，系统将把包含在外部引用文件中的其它外部引用包含进来；而在覆盖方式下，系统将忽略包含在外部引用文件中的其它外部引用，这就避免了与其它文件的多次衔接，消除了循环引用的可能性。若以覆盖方式引用，应将参数bOverlay 设为TRUE。

例5-34：将"c:\Program Files\AutoCAD 2004\Sample\Wilhome.dwg" 文件以外部引用方式附着到当前图形文件中。

Sub Ch5_AttachingExternalReference( )

Dim xrefInserted As AcadExternalReference

Dim insertionPnt(0 To 2) As Double

Dim PathName As String

insertionPnt(0) = 10: insertionPnt(1) = 100: insertionPnt(2) = 0

PathName = "c:\Program Files\AutoCAD 2004\Sample\ _

Wilhome.dwg"

Set xrefInserted = ThisDrawing.ModelSpace.AttachExternal _

Reference(PathName, "WXREF", insertionPnt, 1, 1, 1, 0, False)

End Sub

2． 外部引用对象的编辑

（1）分离外部引用对象——使用块对象的Detack方法。

例：ThisDrawing.Blocks.Item(“WXREF”).Detack

（2）更新外部引用对象——使用块对象的Reload方法。

例：ThisDrawing.Blocks.Item(“WXREF”). Reload

（3）卸载外部引用对象——使用块对象的Unload方法，该法与分离类似，但卸载外部引用时保留与外部引用文件的一个连接，以后用Reload方法可随时加载该文件。

例：ThisDrawing.Blocks.Item(“WXREF”). Unload

（4）绑定外部引用对象

所谓绑定外部引用对象，就是将外部引用对象作为块永久地保留在图形文件中，即并入外部参照。绑定外部引用对象要使用Bind方法，该方法需要一个布尔型参数bPrefixName。如果该值为TRUE，则保留符号表信息（如层、线型等）。外部参照图形的符号名称为：当前图形名称+块名+$x$，其中x是自动增加的整数，以避免取代现有的块定义。如果该值为FALSE，则外部参照图形的符号名称会合并到当前图形，而且没有字首（不再冠以前缀）。如果用户不能确定图形是否包含了重复的符号名称，一般该参数应设置为TRUE。

例：ThisDrawing.Blocks.Item(“WXREF”). Bind TRUE

5.5 标注和公差

5.5.1标注基础知识

标注可以显示对象的形位测量值、对象之间的距离或角度，以及 X 坐标和 Y 坐标。AutoCAD提供三种基本的标注类型:：线性标注、直径（半径）标注和角度标注。其中线性标注包括对齐式、旋转式和坐标式标注。可以为直线、圆等图形对象创建相关的标注对象，也可以是独立的标注对象（只要指定必要的参数）。

1．尺寸标注的组成

尺寸标注对象一般的由标注文字、箭头、尺寸线（标注线）、尺寸界线（延伸线）和尺寸界线起点组成。标注文字可以是实际测量值字符串，也可以是用户输入的尺寸等标注性文字，它还可以包括字首、字尾和公差。单行标注文字使用由 ActiveTextStyle 属性指定的当前激活的文字样式。多行段落文字也使用当前的文字样式，但可对其中的文字进行编辑。中心标记是标识圆或弧的中心点的小十字。

2．标注系统变量

标注系统变量可以用来控制标注对象的外观，需要用SetVariable方法设置。这些标注系统变量主要包括：DIMAUNIT、DIMUPT、DIMTOFL、DIMFT、DIMTIH、DIMJUST和DIMTAD。

例：设置角度标注的单位为度分秒

ThisDrawing.SetVariable”DIMAUNIT”,1

注:该系统变量的默认值为0（10进制度数）。

3．标注的关联性

当与关联标注关联的几何对象被修改时，关联标注会自动调整其位置、方向和测量值。由系统变量DIMASSOC控制关联式标注。

（1）当DIMASSOC值为2——创建关联标注对象。

（2）当DIMASSOC值为1——创建非关联标注对象。

（3）当DIMASSOC值为0——标注对象中的各组成部分会被绘成独立的对象，可以分别编辑。

注：在 AutoCAD 2002 之前的版本中，关联标注和无关联标注的定义不同，并通过 DIMASO 系统变量控制。

5.5.2创建标注

1．线性标注

（1）对齐式标注——用AddDimAligned方法，该法需要输入三个坐标值：依次为两边尺寸界线的引出点和文字位置。

例5-35：创建对齐式标注。

Sub Ch5_AddDimAligned( )

Dim dimObj As AcadDimAligned

Dim point1(0 To 2) As Double

Dim point2(0 To 2) As Double

Dim location(0 To 2) As Double

point1(0) = 5#: point1(1) = 5#: point1(2) = 0#

point2(0) = 10#: point2(1) = 5#: point2(2) = 0#

location(0) = 5#: location(1) = 7#: location(2) = 0#

Set dimObj = ThisDrawing.ModelSpace.AddDimAligned(point1, point2, location)

ZoomAll

End Sub

（2）旋转式标注——用AddDimRotated方法，该法需要三个坐标值和一个标注线的角度值。三个坐标值指定两边界的引出点和文字位置。角度值代表标注线的旋转角度，须以弧度为单位。

例5-36：创建旋转式标注。

Sub Ch5_AddDimRotated( )

Dim dimObj As AcadDimRotated

Dim point1(0 To 2) As Double

Dim point2(0 To 2) As Double

Dim location(0 To 2) As Double

Dim rotAngle As Double

point1(0) = 0#: point1(1) = 5#: point1(2) = 0#

point2(0) = 5#: point2(1) = 5#: point2(2) = 0#

location(0) = 0#: location(1) = 0#: location(2) = 0#

rotAngle = 120

rotAngle = rotAngle * 3.141592 / 180# 

Set dimObj = ThisDrawing.ModelSpace.AddDimRotated _

(point1, point2, location, rotAngle)

ZoomAll

End Sub

2. 径向标注

（1）直径标注——用AddDimDiametric方法，该法需要输入三个参数：依次为直径标注的起点坐标、直径标注的终点坐标和引线的长度。

例5-37：在模型空间中创建直径标注。

Sub Ch5_CreateDiametricDimension ( )

Dim dimObj As AcadDimDiametric

Dim chordPoint(0 To 2) As Double

Dim farChordPoint(0 To 2) As Double

Dim leaderLength As Double

chordPoint(0) = 5#: chordPoint(1) = 3#: chordPoint(2) = 0#

farChordPoint(0) = 5#: farChordPoint(1) = 5#

farChordPoint(2) = 0#

leaderLength = 1#

Set dimObj = ThisDrawing.ModelSpace.AddDimDiametric _

(chordPoint, farChordPoint, leaderLength)

ZoomAll

End Sub

（2）半径标注——用AddDimRadial方法，该法需要输入三个参数：依次为圆弧的中心坐标（半径标注起点）、引线终点坐标和引线长度。

例5-38：在模型空间中创建半径标注。 

Sub Ch5_CreateRadialDimension( )

Dim dimObj As AcadDimRadial

Dim center(0 To 2) As Double

Dim chordPoint(0 To 2) As Double

Dim leaderLen As Integer

' 定义标注

center(0) = 0:center(1) = 0:center(2) = 0

chordPoint(0) = 5:chordPoint(1) = 5:chordPoint(2) = 0

leaderLen = 5

' 在模型空间中创建半径标注

Set dimObj = ThisDrawing.ModelSpace. _

AddDimRadial(center, chordPoint, leaderLen)

ZoomAll

End Sub

3．角度标注

角度标注标识两线段或三个点之间构成的夹角大小。创建角度型标注需要使用AddDimAngular方法，该法需要输入四个参数：顶点坐标、两条边界线对应的两个引出点坐标和标注文字的位置坐标。其中边界线的引出点是边界线穿过的点。

例5-39：在模型空间中创建角度标注。 

Sub Ch5_CreateAngularDimension( )

Dim dimObj As AcadDimAngular

Dim angVert(0 To 2) As Double

Dim FirstPoint(0 To 2) As Double

Dim SecondPoint(0 To 2) As Double

Dim TextPoint(0 To 2) As Double

' 定义标注

angVert(0) = 0:angVert(1) = 5:angVert(2) = 0

FirstPoint(0) = 1:FirstPoint(1) = 7:FirstPoint(2) = 0

SecondPoint(0) = 1:SecondPoint(1) = 3:SecondPoint(2) = 0

TextPoint(0) = 3:TextPoint(1) = 5:TextPoint(2) = 0

' 在模型空间中创建角度标注

Set dimObj = ThisDrawing.ModelSpace. _

AddDimAngular(angVert, FirstPoint, SecondPoint, TextPoint)

ZoomAll

End Sub

1． 坐标式标注

坐标（或基准）标注用于测量从原点（称为基准点）到被标注的特征（例如部件上的孔）之间的垂直距离。 

坐标标注由带引线的 X 或 Y 坐标组成。X 基准坐标标注沿着 X 轴测量特征与基准点之间的距离。Y 基准坐标标注沿着 Y 轴测量特征与基准点之间的距离。AutoCAD 使用当前 UCS 的原点来确定测量出的坐标值。坐标使用绝对值。 

文字将与坐标引线对齐，而不考虑当前标注样式定义的文字方向。用户可以接受默认文字，也可以提供自己的文字。 

要创建坐标标注，要使用 AddDimOrdinate 方法。该方法需要输入三个值：用于指定要标注的点的坐标值，用于指定引线端点的坐标值，以及用于指定标注是 X 基准坐标标注还是 Y 基准坐标标注的布尔标志。如果为布尔标志输入 TRUE，该方法将创建 X 基准坐标标注。如果输入 FALSE，则会创建 Y 基准坐标标注。 

例5-40：在模型空间中创建坐标标注。 

Sub Ch5_CreateOrdinateDimension( )

Dim dimObj As AcadDimOrdinate

Dim definingPoint(0 To 2) As Double

Dim leaderEndPoint(0 To 2) As Double

Dim useXAxis As Long

' 定义标注

definingPoint(0) = 5：definingPoint(1) = 5：definingPoint(2) = 0

leaderEndPoint(0) = 10

leaderEndPoint(1) = 5

leaderEndPoint(2) = 0

useXAxis = 5

' 在模型空间中创建坐标标注

Set dimObj = ThisDrawing.ModelSpace. _

AddDimOrdinate(definingPoint, leaderEndPoint, useXAxis)

ZoomAll

End Sub

5.5.3编辑标注

1. 编辑标注对象常用的方法

（1）ArrayPolar 创建环形阵列。

（2）ArrayRectangular 创建矩形阵列。

（3）Copy 复制标注对象。

（4）Erase 删除标注对象。

（5）Mirror 镜象标注对象。

（6）Move 移动标注对象。

（7）Rotate 旋转标注对象。

（8）ScaleEntity 调整标注对象比例。

2. 编辑标注对象常用的属性

（1）Rotation 指定尺寸线的旋转角度（以弧度为单位）。

（2）StyleName 指定标注样式的名称。

（3）TextOverride 为标注指定字符串。

（4）TextPosition 指定标注文字的位置。

（5）TextRotation 指定标注文字的旋转角度。

（6）Measurement 指定标注的实际测量值。

例5-41：本例利用Rotate方法旋转直径标注对象的标注方位；利用多行文字格式码、Unicode字符输入字符串；利用TextOverride属性替换标注文字的办法生成特殊要求的标注文字及尺寸公差。

Sub Ch5_ RotateAndOverrideDimensionText( )

Dim fontfilename As String

Dim textbold As Boolean

Dim textitalic As Boolean

Dim textcharset As Long

Dim textpfamily As Long

ThisDrawing.ActiveTextStyle.GetFont _

fontfilename, textbold,textitalic, textcharset, textpfamily

fontfilename = "txt"

ThisDrawing.ActiveTextStyle.SetFont _

fontfilename, textbold, textitalic, textcharset, textpfamily

' 创建直径标注

Dim dimObjr As AcadDimDiametric

Dim pta(0 To 2) As Double

Dim ptb(0 To 2) As Double

Dim leaderlen As Double

pta(0) = 50: pta(1) = 150: pta(2) = 0

ptb(0) = 150: ptb(1) = 150: ptb(2) = 0

leaderlen=5

Set dimObjr = ThisDrawing.ModelSpace.AddDimDiametric _

(pta, ptb, leaderlen)

Dim cob1 As AcadCircle

Dim cp1(0 To 2) As Double

Dim r1 As Double

cp1(0) = 100: cp1(1) = 150: cp1(2) = 0

r1 = 50

Set cob1 = ThisDrawing. ModelSpace.AddCircle(cp1, r1)

' 利用Rotate方法编辑直径标注对象

dimObjr.Rotate cp1, 0.45

Dim dimObj As AcadDimRadial

Dim center(0 To 2) As Double

Dim chordPoint(0 To 2) As Double

Dim leaderLen1 As Integer

center(0) = 30#: center(1) = 80#: center(2) = 0#

chordPoint(0) = 55#: chordPoint(1) = 80#: chordPoint(2) = 0#

leaderLen1 = 5

Set dimObj = ThisDrawing.ModelSpace.AddDimRadial _

(center, chordPoint, leaderLen1)

dimObj.Rotate center, 0.707

dimObj.Update

' 创建线性标注对象

Dim dimOb1 As AcadDimAligned

Dim dimpoint1(0 To 2) As Double

Dim dimpoint2(0 To 2) As Double

Dim dimlocation(0 To 2) As Double

dimpoint1(0) = 170

dimpoint1(1) = 170

dimpoint1(2) = 0

dimpoint2(0) = 270

dimpoint2(1) = 170

dimpoint2(2) = 0

dimlocation(0) = 210

dimlocation(1) = 130

dimlocation(2) = 0

Set dimOb1= ThisDrawing.ModelSpace.AddDimAligned _

(dimpoint1,dimpoint2, dimlocation)

' 利用多行文字格式码、Unicode字符输入字符串

Dim TS As String

TS = "\A1;\H5;\U+2205100\H0.75x;\S+0.01^-0.15;\P"

' 利用TextOverride属性替换标注文字

dimOb1.TextOverride = TS

dimOb1.Update

End Sub

该通用过程的运行结果如图5-5所示。

图5-5尺寸标注综合示例的运行结果

5.5.4标注样式

标注样式是一组用于确定标注外观的设置。使用标注样式，可以建立和实施图形的绘图标准。 

所有标注都使用当前标注样式来创建。如果在创建标注之前没有定义或应用样式，AutoCAD 会应用默认的样式，即 STANDARD。要将某标注样式设置为当前标注样式，应使用 ActiveDimStyle 属性。

每个标注样式中都包含了两种设置：

（1）父样式：较为普遍的对大多数尺寸变量的设置。

（2）子样式：仅适用于某一种尺寸标注的设置。

父样式作用于所有的子样式设置，而每个子样式设置又优先于父样式设置。

要设置父标注样式，应从命名和保存样式开始。新样式以当前样式为基础，并包括对标注组成部分的布局、文字位置以及注释外观的所有后续更改。这里的注释指主单位、换算单位、公差和文字。 

1． 创建、修改和复制标注样式

要创建新的标注样式，须使用 Add 方法。这个方法需要输入新标注样式的名称。 

AutoCAD ActiveX Automation 允许添加用户自己创建的新标注样式或者更改当前标注样式。也可以通过 StyleName 属性来改变某一尺寸标注对象的样式。 

用户也可以使用 CopyFrom 方法，将一种标注样式从源对象复制到新标注样式中。源对象可以是另一个 DimStyle 对象、标注对象、公差对象 或 Leader 对象，甚至可以是 Document 对象。如果从另一个标注样式复制，则样式被精确复制。如果从标注对象、公差对象 或 Leader 对象复制样式的设置，则当前设置（包括所有对象替代）会被复制到新样式中（即新样式中将保留源对象的尺寸标注设置和当前设置中不与源对象设置发生冲突的设置）。如果复制 Document 对象的样式，则当前标注样式加上所有的图形替代，会被复制到新样式中（即新样式中将包括所有子样式的设置和当前设置中不与子样式发生冲突的设置）。 

2．改变标注样式中的设置值

每个标注对象在使用当前标注样式的设置时，都可以忽略该样式中的某些设置，而使用自己本身的属性来替代标注样式中的设置值。以下属性可用于大多数标注对象： 

（1）AltRoundDistance 指定换算单位的舍入。

（2）AngleFormat 指定角度标注的单位格式。

（3）Arrowhead1Block指定用作尺寸线第一端自定义箭头的块。

（4）Arrowhead2Block指定用作尺寸线第二端自定义箭头的块。

（5）Arrowhead1Type指定尺寸线的第一端箭头类型。

（6）Arrowhead2Type指定尺寸线的第二端箭头类型。

（7）ArrowheadSize 指定尺寸线箭头、引线箭头和钩线的大小。

（8）CenterMarkSize 指定半径标注和直径标注的中心标记大小。 

（9）CenterType 指定半径标注和直径标注的中心标记类型。

（10）DecimalSeparator 指定在十进制标注和公差值中用作小数分隔符的字符。 

（11）DimensionLineColor 指定标注、引线或公差对象的尺寸线颜色。

（12）DimensionLineWeight 指定尺寸线的线宽。

（13）DimLine1Suppress指定是否隐藏第一段尺寸线。

（14）DimLine2Suppress指定是否隐藏第二段尺寸线。

（15）DimLineInside 指定是否只在尺寸界线之内显示尺寸线。 

（16）ExtensionLineColor 指定标注的尺寸界线的颜色。 

（17）ExtensionLineExtend 指定尺寸界线超出尺寸线的距离。 

（18）ExtensionLineOffset 指定尺寸界线从原点偏移的距离。 

（19）ExtensionLineWeight 指定尺寸界线的线宽。

（20）ExtLine1EndPoint指定第一条尺寸界线的端点。

（21）ExtLine2EndPoint指定第二条尺寸界线的端点。

（22）ExtLine1StartPoint指定第一条尺寸界线的起点。 

（23）ExtLine2StartPoint指定第二条尺寸界线的起点。

（24）ExtLine1Suppress指定是否隐藏第一条尺寸界线。

（25）ExtLine2Suppress指定是否隐藏第二条尺寸界线。

（26）Fit 指定文字和箭头在尺寸界线的内部或外部。

（27）ForceLineInside 指定当文字放在尺寸界线之外时是否仍要在尺寸界线之间画一条尺寸线。

（28）FractionFormat 指定标注和公差中的分数值的格式。 

（29）HorizontalTextPosition 指定标注文字的水平对正方式。 

（30）LinearScaleFactor 指定线性标注测量值的全局缩放比例。

（31）PrimaryUnitsPrecision 指定标注或公差的主单位的小数位数。

（32）SuppressLeadingZeros指定是否隐藏标注值中的前导零。

（33）SuppressTrailingZeros指定是否隐藏标注值中的后续零。

（34）SuppressZeroFeet指定是否隐藏标注值中的零英尺测量值。

（35）SuppressZeroInches指定是否隐藏标注值中的零英寸测量值。

（36）TextColor 指定标注和公差对象的文字颜色。

（37）TextGap 指定当用户截断尺寸线以放置标注文字时，标注文字与尺寸线之间的距离。

（38）TextHeight 指定标注或公差文字的高度。

（39）TextInside 指定是否在尺寸界线之内绘制标注文字。 

（40）TextInsideAlign 指定所有标注类型（坐标标注除外）的标注文字是否位于尺寸界线之内。

（41）TextMovement 指定当文字移动时标注文字的绘制方式。

（42）TextOutsideAlign 指定所有标注类型（坐标标注除外）的标注文字在尺寸界线之外的位置。

（43）TextPosition 指定标注文字的位置。 

（44）TextPrecision 指定角度标注文字的精度。 

（45）TextPrefix 指定标注值的前缀。 

（46）TextRotation 指定标注文字的旋转角度。 

（47）TextSuffix 指定标注值的后缀。

（48）ToleranceDisplay 指定公差是否要与标注文字一起显示。

（49）ToleranceHeightScale 指定公差值文字高度相对于标注文字高度的比例。

（50）ToleranceJustification 指定公差值相对于标注文字的垂直对正方式。

（51）ToleranceLowerLimit 指定标注文字的下偏差。

（52）TolerancePrecision 指定主标注中公差值的精度。

（53）ToleranceSuppressLeadingZeros 指定是否隐藏公差值中的前导零。 

（54）ToleranceSuppressTrailingZeros 指定是否隐藏标注值中的后续零。 

（55）ToleranceUpperLimit 指定标注文字的上偏差。

（56）UnitsFormat 指定所有标注（角度标注除外）的单位格式。

（57）VerticalTextPosition 指定文字相对于尺寸线的垂直位置。

例5-42：利用TextPrefix 属性指定标注值的前缀“φ”和利用TextSuffix属性指定标注值的后缀“H7”。

Sub Ch5_ TextPrefixAndTextSuffix ( )

Dim dimObj As AcadDimAligned

Dim point1(0 To 2) As Double, point2(0 To 2) As Double

Dim location(0 To 2) As Double

Dim Prefix As String

Dim suffix As String

' 定义标注对象

point1(0) = 0: point1(1) = 5: point1(2) = 0

point2(0) = 5: point2(1) = 5: point2(2) = 0

location(0) = 5: location(1) = 7: location(2) = 0

' 创建对齐式标注

Set dimObj = ThisDrawing.ModelSpace.AddDimAligned(point1, point2, location)

ThisDrawing.Application.ZoomAll

' 改变标注前缀

Prefix = InputBox("键入前缀的文本", "设置Prefix", "%%c")

dimObj.TextPrefix = Prefix 

ThisDrawing.Regen acAllViewports

Prefix = dimObj.TextPrefix

If Prefix <> "" Then

MsgBox "尺寸的前缀已被改变为: " & Prefix

Else

MsgBox "尺寸的前缀已被移去"

End If

' 改变标注后缀

Suffix = InputBox("键入后缀的文本", "设置Suffix", "H7")

dimObj.TextSuffix = suffix 

ThisDrawing.Regen acAllViewports

suffix = dimObj.TextSuffix

If suffix <> "" Then

MsgBox "尺寸的后缀已被改变为: " & suffix

Else

MsgBox "尺寸的后缀已被移去"

End If

End Sub

5.5.5创建及编辑引线标注

引线标注由引线和注释两部分组成。在注释和引线之间有一小段自动产生的线条，称为标线。

1． 创建引线

要创建引线，需要使用AddLeader方法，该法需要输入三个参数，依次为：创建引线起点的坐标值阵列、注释对象（若没有注释则为NULL。例：Set annotationObject = Nothing）和要创建的引线类型。

常用引线类型：

（1）acLineNoArrow 引线为直线，无箭头。

（2）acLineWithArrow 引线为直线，有箭头。

（3）acSPLineNoArrow 引线为平滑样条曲线，无箭头。

（4）acSPLineWithArrow 引线为平滑样条曲线，有箭头。

2．引线标注的注释

引线标注的注释可以是Tolorance、Mtext或BlockRef对象，用户可以创建新的注释，也可添加现有注释的副本。要先创建注释对象，然后作为创建引线标注对象时AddLeader方法的一个参数。

引线与其注释是关联的，因此当注释移动时，引线的端点也会随着移动。在移动文字注释和特征控制框注释时，引线的最后一段会根据注释与引线的倒数第二个点之间的关系，决定是附加到注释的左侧还是右侧。如果注释的中点在引线的倒数第二个点的右侧，则引线将附着到右侧，否则将附着到左侧。除了引线和注释之间的关联性关系以外，引线与其注释在图形中是完全独立的两种对象。编辑引线不会影响注释，编辑注释也不会影响引线。

例5-43：使用 MText 对象作为注释来创建引线标注。

Sub Ch5_ CreateLeaderAddAnnotation( )

Dim leaderObj As AcadLeader

Dim mtextObj As AcadMText

Dim points(0 To 8) As Double

Dim insertionPoint(0 To 2) As Double

Dim width As Double

Dim leaderType As Integer

Dim annotationObject As Object

Dim textString As String, msg As String

' 在模型空间中创建 MText 对象

textString = "Hello, World！"

insertionPoint(0) = 45

insertionPoint(1) = 45

insertionPoint(2) = 0

width = 40

Set mtextObj = ThisDrawing.ModelSpace. _

AddMText(insertionPoint, width, textString)

' 引线的数据

points(0) = 10: points(1) = 10: points(2) = 0

points(3) = 40: points(4) = 40: points(5) = 0

points(6) = 45: points(7) = 40: points(8) = 0

leaderType = acLineWithArrow

' 在模型空间中创建 Leader 对象，使MText 对象和引线相关联

Set annotationObject = mtextObj

Set leaderObj = ThisDrawing.ModelSpace. _

AddLeader(points, annotationObject, leaderType)

ZoomAll

End Sub

3．编辑引线

任何会改变引线注释位置的修改都会影响到相关联的引线的端点位置。另外，旋转注释也会使引线的标线（如果有）随之旋转。 

要调整引线的大小，可以将其缩放。缩放只更新选定对象的比例。例如，如果缩放引线，注释会保持与引线端点的相对位置不变，但不会进行缩放。 除了缩放以外，还可以移动、镜像和旋转引线。使用 ScaleEntity、Move、Mirror 和 Rotate 方法来编辑引线。也可以使用 StyleName 属性来更改与注释关联的文字样式。 

5.5.6创建及编辑形位公差

1. 创建形位公差

形位公差由公差类型、公差值、基准和材料状况等组成，这些组成元素都放置在一个特征控制框中。创建形位公差要用Tolerence对象的AddTolerence方法，该方法需要三个参数，依次为：构成公差符号的字符串、图形中放置公差的位置和指定方向的方向向量。

要生成特征控制框中的公差符号字符串，还需要应用创建特殊字符的控制码。常用的有：

（1）{ \Fgdt；h}——圆跳动。

（2）{\ Fgdt；t}——全跳动。

（3）{\Fgdt；r}——同轴度。

（4）{ \Fgdt；i}——对称度。

（5）{ \Fgdt；c}——平面度。

（6）{ \Fgdt；b}——垂直度。

（7）{ \Fgdt；a}——倾斜度。

（8）{\Fgdt；d}——面轮廓度。

（9）{ \Fgdt；k}——线轮廓度。

（10）{ \Fgdt；e}——圆度。

（11）{ \Fgdt；g}——圆柱度。

（12）{ \Fgdt；u}——直线度。

（13）{ \Fgdt；f}——平行度。

（14）{ \Fgdt；j}——位置度。

（15）{ \Fgdt；m}——最大实体要求（MMC）。

（16）{ \Fgdt；l}——最小实体要求（LMC）。

（17）{ \Fgdt；s}——不考虑特征尺寸（RFS）。

（18）{ \Fgdt；p}——延伸公差带。

（19）{ \Fgdt；n}——φ

（20）%%v——特征控制框的分隔线。

（21） & vbCrLf & ——多行的特征控制框换行符。

2. 编辑形位公差

用户可以对形位公差符号进行复制、移动、删除、缩放和旋转等操作。

公差会受以下系统变量的控制：

（1）DIMCLRD 控制特性控制框的颜色。

（2）DIMCLRT 控制公差文字的颜色。

（3）DIMGAP 控制特性控制框和文字之间的间隙。

（4）DIMTXT 控制公差文字的大小。

（5）DIMTXTSTY 控制公差文字的样式。

要使用 SetVariable 方法来设置系统变量的值。

例5-44：创建带引线的形位公差。

Sub Ch5_CreateTolerance( )

Dim directionP(0 To 2) As Double

directionP(0) = 10: directionP(1) = 0: directionP(2) = 0

Dim insertP1(0 To 2) As Double

Dim zfc1 As String

insertP1(0) = 200

insertP1(1) = 150

insertP1(2) = 0

zfc1 = "//%%v0.05{\Fgdt;m}%%vA"

Call ThisDrawing.ModelSpace.AddTolerance (zfc1, insertP1, directionP)

' 生成引线

Dim leaderObj As AcadLeader

Dim Lpoints(0 To 8) As Double

Dim leaderType As Integer

Dim annotationObject As AcadObject

Lpoints(0) = 164: Lpoints(1) = 120: Lpoints(2) = 0

Lpoints(3) = 164: Lpoints(4) = 150:Lpoints(5) = 0

Lpoints(6) = 200: Lpoints(7) = 150: Lpoints(8) = 0

leaderType = acLineWithArrow

Set leaderObj = ThisDrawing.ModelSpace.AddLeader _

(Lpoints, annotationObject, leaderType)

' 生成多行的形位公差对象

Dim toleranceObj As AcadTolerance

Dim textString As String

Dim insertionPoint(0 To 2) As Double

Dim direction(0 To 2) As Double

textString = "{\Fgdt;r}%%v0.12{\Fgdt;l}%%vA{\Fgdt;m}" & vbCrLf & _

"{\Fgdt;t}%%v0.02%%vB{\Fgdt;s}"

insertionPoint(0) = 200#: insertionPoint(1) = 100#: insertionPoint(2) = 0#

direction(0) = 1#: direction(1) = 0#: direction(2) = 0#

Set toleranceObj = ThisDrawing.ModelSpace.AddTolerance _

(textString, insertionPoint, direction)

ZoomExtents

ThisDrawing.Regen acAllViewports

End Sub

该通用过程的运行结果如图5-6所示。

图5-6创建带引线的形位公差

5.6 图形对象创建与编辑的综合示例

5.6.1演示图形对象创建与编辑的综合示例

1. 综合示例的用户界面

综合示例的窗体如图5-7所示。该应用示例具有多项常用的创建或编辑图形对象的功能，如用户窗体界面上命令按钮的标签所示。单击各个按钮即运行不同的事件过程。

因该程序利用VB6.0开发，则使用时应首先单击《引用AutoCAD并建立图形文件》按钮，实现连接AutoCAD及建立图形文件，然后再根据需要单击某按钮演示其功能。


图5-7演示图形对象创建与编辑综合示例的窗体

2． 综合示例的源程序

这些事件过程的源程序代码均受录在本书附带的光盘中。此节仅介绍其中的一段。

例5-45：该例仅介绍《设新标注样式》的通用过程。这段程序的功能是通过改变系统变量的值，将角度单位由度改为度分秒，并将角度标注的文本精度改为3，最后从当前文挡对象复制出专用的标注样式。

Sub Ch5_CreateNewDimStyle( )

' 应用SetVariable方法设置系统变量 DIMAUNIT 为1

' 将角度单位由度改为度分秒

Dim dataIn As Integer

Dim sysVarName As String

Dim sysVarData As Double

sysVarName = "DIMAUNIT"

dataIn = 1

sysVarData = dataIn

acadDoc.SetVariable sysVarName, sysVarData

' 应用 GetVariable方法获取系统变量

sysVarData = acadDoc.GetVariable(sysVarName)

MsgBox sysVarName & " = " & sysVarData, , _

"角度单位由度改为度分秒"

' 设角度单位改为度分秒后的文本精度

Dim datadec As Integer

sysVarName = "DIMADEC"

datadec = 3

sysVarData = datadec

acadDoc.SetVariable "DIMADEC", datadec

sysVarData = acadDoc.GetVariable(sysVarName)

MsgBox "DIMADEC" & " = " & sysVarData, , _

"设角度单位改为度分秒后的文本精度"

' 从当前图形值复制出"TestDimStyle"

Dim newDimStyle As AcadDimStyle

Dim currDimStyle As AcadDimStyle

Set currDimStyle = acadDoc.ActiveDimStyle

Set newDimStyle = acadDoc.DimStyles.Add("TestDimStyle")

Call newDimStyle.CopyFrom(Doc)

' 激活"TestDimStyle"为当前尺寸标注样式

acadDoc.ActiveDimStyle = newDimStyle

End Sub

该通用过程的运行结果如图5-8所示。

图5-8 创建新的角度标注样式

5.6.2生成带键槽的轴剖面综合示例

1．设计用户界面 

生成带键槽的轴剖面综合示例的窗体，如图5-9所示。

本综合示例是通过文本框输入已知参数的。键入的已知条件（轴直径、键槽宽、键槽深及必要的尺寸公差），首先要由文本（例如：Text1.Text=”40”; Text2.Text=”12”; … ）转换为数值才能写入计算数值的程序（例如：d=Val(Text1.Text)）。运行时，应先单击命令按钮《引用AutoCAD》，待出现AutoCAD界面后，再单击命令按钮《运行》。运行结果如图5-10所示。

图5-9带键槽轴剖面综合示例的用户窗体界面

图5-10带键槽的轴剖面图综合示例的运行结果

2．带键槽轴剖面综合示例的《运行》事件过程

该综合示例的源程序全部收录在本书附带的光盘中，此节仅介绍其中的一段。

事件过程主要完成以下内容：

（1）新设置四个图层及其颜色、线型（包括加载点划线线型）和线宽。

（2）创建及编辑优化多段线——形成带键槽的轴剖面外轮廓。

（3）创建图案填充对象，并追加其外边界。

（4）画中心线。

（5）创建及编辑尺寸标注。

对于直径的标注，先按水平方向创建标注，然后利用标注对象的Rotate属性将其旋转到适当位置。

关于直径的尺寸公差，先采用多行文本格式码生成带尺寸公差的标注文本字符串，然后利用TextOverride属性覆盖原来的不带尺寸公差的标注文本。

例5-46：生成带键槽轴剖面的《运行》事件过程的部分代码。

Sub Ch5_L2( )

……

' 设置新图层

Dim layer1 As AcadLayer

Dim layer2 As AcadLayer

Dim layer3 As AcadLayer

Dim layer4 As AcadLayer

Set layer0 = acadDoc.Layers.Item(0)

Set layer1 = acadDoc.Layers.Add("粗实线层")

Set layer2 = acadDoc.Layers.Add("中心线层")

Set layer3 = acadDoc.Layers.Add("剖面线层")

Set layer4 = acadDoc.Layers.Add("标注层")

layer1.Lineweight = acLnWt080 ' 粗实线的线宽

layer1.Color = acWhite

layer2.Color = acRed

layer3.Color = acCyan

layer4.Color = acGreen

' 加载中心线线型

On Error Resume Next

Dim entry As AcadLineType

Dim found As Boolean

found = False

For Each entry In acadDoc.Linetypes

If StrComp(entry.Name, "centerx2", 1) = 0 Then

found = True

Exit For

End If

Next

If Not (found) Then acadDoc.Linetypes.Load "centerx2", "acad.lin"

layer2.Linetype = "centerx2"

' 输入原始参数

Dim Center(0 To 2) As Double

Dim Vert(0 To 7) As Double

Dim L04 As Double

Dim L01 As Double

Dim b As Double

Dim r As Double

b = Val(Text2.Text): r = Val(Text1.Text) / 2: L01 = r - Val(Text3.Text)

L04 = Sqr(r * r - (b / 2) ^ 2)

Center(0) = 100: Center(1) = 100: Center(2) = 0

' 画中心线

acadDoc.ActiveLayer = layer2

Dim line1 As AcadLine

Dim line2 As AcadLine

Dim pl1s(0 To 2) As Double

Dim pl1e(0 To 2) As Double

Dim pl2s(0 To 2) As Double

Dim pl2e(0 To 2) As Double

pl1s(0) = Center(0) - r - 1.5: pl1s(1) = Center(1): pl1s(2) = 0

pl1e(0) = Center(0) + r + 1.5: pl1e(1) = Center(1): pl1e(2) = 0

pl2s(0) = Center(0): pl2s(1) = Center(1) + r + 1.5: pl2s(2) = 0

pl2e(0) = Center(0): pl2e(1) = Center(1) - r - 1.5: pl2e(2) = 0

Set line1 = acadDoc.ModelSpace.AddLine(pl1s, pl1e)

Set line2 = acadDoc.ModelSpace.AddLine(pl2s, pl2e)

' 用AddHatch方法创建Hatch对象

acadDoc.ActiveLayer = layer3

Dim patternName As String

Dim PatternType As Long

Dim bAssociativity As Boolean

Dim anObj As AcadHatch

patternName = "ANSI31"

PatternType = acHatchPatternTypePreDefined

Associate = True

Set anObj = acadDoc.ModelSpace.AddHatch _

(PatternType, patternName, Associate)

' 确定剖面线外边框

' 创建辅助的优化多段线

acadDoc.ActiveLayer = layer1

Dim LWP As AcadLWPolyline

Vert(0) = Center(0) - b / 2: Vert(1) = Center(1) + L01

Vert(2) = Vert(0): Vert(3) = Center(1) + L04

Vert(4) = Vert(0) + b: Vert(5) = Vert(3)

Vert(6) = Vert(0) + b: Vert(7) = Vert(1)

Set LWP = acadDoc.ModelSpace.AddLightWeightPolyline(Vert)

LWP.Closed = True

' 例用SetBulge方法编辑编辑多段线,形成轴断面外轮廓

Dim h As Double

Dim bbb As Double

h = r + L04

bbb = h / (b / 2)

Call LWP.SetBulge(1, bbb)

LWP.Update

' 添加剖面线

Dim OuterLoop(0 To 0) As AcadEntity

Set OuterLoop(0) = LWP

anObj.AppendOuterLoop (OuterLoop)

' 让系统更新显示

anObj.Evaluate

' 尺寸标注

acadDoc.ActiveLayer = layer4

Dim dimL1Obj1 As AcadDimAligned

Dim dimL1Obj2 As AcadDimAligned

Dim dimDObj As AcadDimDiametric

Dim dimA(0 To 2) As Double

Dim dimB(0 To 2) As Double

Dim dimC(0 To 2) As Double

Dim dimD(0 To 2) As Double

Dim dimE(0 To 2) As Double

Dim dimF(0 To 2) As Double

Dim Leaderlen As Double

Dim Position1(0 To 2) As Double

Dim Position2(0 To 2) As Double

dimA(0) = Center(0) - Text2.Text / 2: dimA(1) = Center(1) + r - 2: dimA(2) = 0

dimB(0) = Center(0) + Text2.Text / 2: dimB(1) = Center(1) + r - 2: dimB(2) = 0

Position1(0) = Center(0): Position1(1) = Center(1) + r + 2

Position1(2) = 0

dimC(0) = Center(0): dimC(1) = Center(1) - r: dimC(2) = 0

dimD(0) = Center(0): dimD(1) = Center(1) + L01: dimD(2) = 0

Position2(0) = Center(0) + r + 5: Position2(1) = Center(1)

Position2(2) = 0

dimE(0) = Center(0) - r: dimE(1) = Center(1): dimE(2) = 0

dimF(0) = Center(0) + r: dimF(1) = Center(1): dimF(2) = 0

Leaderlen = 3

Set dimL1Obj1 = acadDoc.ModelSpace.AddDimAligned _

(dimA, dimB, Position1)

Set dimL1Obj2 = acadDoc.ModelSpace.AddDimAligned _

(dimC, dimD, Position2)

Set dimDObj = acadDoc.ModelSpace.AddDimDiametric _

(dimE, dimF, Leaderlen)

dimDObj.Rotate Center, 0.45

' 增添轴的尺寸公差

Dim ts As String

ts = "\A1;\H2.5;\U+2205" + Text1.Text + "\H0.75x;\S" _

+ Text4.Text + "^" + Text5.Text + ";\P"

dimDObj.TextOverride = ts

dimDObj.Update

acadApp.ZoomExtents

acadDoc.ActiveLayer = layer0

……

End Sub

6．1 概述

众所周知，二维图纸是三维实体的二维投影图。但习惯上二维图又都是由设计者根据投影关系独立创建的，不但费时费力而且有可能表达不清或产生绘图错误。有鉴于此，用户会希望创建真正的三维模型。AutoCAD支持创建真实、详尽的三维对象，并且能够以多种方式对其进行处理。AutoCAD支持创建线框、曲面和实（心）体三种类型的三维模型。同样，这些有关三维的操作，也可以用AutoCAD ActiveX/VBA 开发的应用程序来实现。

6.1.1三维模型

1．线框模型

线框模型是对象的一种骨架式描述。其信息由对象边缘的点、线或曲线组成。用 AutoCAD 可以在三维空间的任何位置放置二维（平面）对象来创建线框模型。AutoCAD 也提供一些三维线框对象，例如可以使用 Add3DPoly 方法创建三维多段线。由于构成线框模型的每个对象都必须单独绘制和定位，因此，这种建模方式可能最为耗时费力。

2． 曲面模型

曲面模型不仅定义了三维对象的边缘，还定义了其面特性。AutoCAD 曲面模型使用多边形网格和多面网格创建镶嵌面式的曲面。由于网格面是平面的，因此网格只能近似于曲面。

注：使用 Mechanical Desktop? 可以创建真正的曲面。为了区分这两种曲面，AutoCAD 称镶嵌面为网格。

3． 实体模型

三维实体对象（ 3DSolid 对象）代表对象的整个容积。实体是信息最完整和最确切的三维建模类型。复杂实体也比线框和网格更容易构造和编辑。 

有多种创建三维实体的方法：根据基本实体形（长方体、圆锥体、圆柱体、球体、圆环体和楔体）创建实体、沿路径拉伸二维对象及绕轴线旋转二维对象来生成实体等。

6.1.2三维坐标系

创建三维对象时，需要指定三维坐标。输入三维世界坐标系 WCS 坐标与输入二维 WCS 坐标类似，只是除了指定 X 和 Y 轴方向后，应依据右手定则确定 Z 轴方向和确定绕某轴进行正方向旋转的旋向。与二维坐标相同，三维坐标也是使用变体（Variant）类型的数据向 ActiveX 方法和属性传递坐标及查询坐标。

例6-1：创建两条多段线，每条都有三个坐标。第一条是二维多段线，第二条是三维多段线。

Sub Ch6_Polyline_2D_3D( )

Dim pline2DObj As AcadLWPolyline

Dim pline3DObj As AcadPolyline

Dim points2D(0 To 5) As Double

Dim points3D(0 To 8) As Double 

’定义三个二维多段线点

points2D(0) = 1: points2D(1) = 1

points2D(2) = 1: points2D(3) = 2

points2D(4) = 2: points2D(5) = 2

’定义三个三维多段线点

points3D(0) = 1: points3D(1) = 1: points3D(2) = 0

points3D(3) = 2: points3D(4) = 1: points3D(5) = 0

points3D(6) = 2: points3D(7) = 2: points3D(8) = 0

’创建二维优化多段线

Set pline2DObj = ThisDrawing.ModelSpace.AddLightWeightPolyline(points2D)

pline2DObj.Color = acRed

pline2DObj.Update

’创建三维多段线

Set pline3DObj = ThisDrawing.ModelSpace.AddPolyline(points3D)

pline3DObj.Color = acBlue

pline3DObj.Update

’查询多段线的坐标

Dim get2Dpts As Variant

Dim get3Dpts As Variant

get2Dpts = pline2DObj.Coordinates

get3Dpts = pline3DObj.Coordinates

’显示坐标

MsgBox ("2D polyline (red): " & vbCrLf & _

get2Dpts(0) & ", " & get2Dpts(1) & vbCrLf & _

get2Dpts(2) & ", " & get2Dpts(3) & vbCrLf & _

get2Dpts(4) & ", " & get2Dpts(5))

MsgBox ("3D polyline (blue): " & vbCrLf & _

get3Dpts(0) & ", " & get3Dpts(1) & ", " & _

get3Dpts(2) & vbCrLf & _

get3Dpts(3) & ", " & get3Dpts(4) & ", " & _

get3Dpts(5) & vbCrLf & _

get3Dpts(6) & ", " & get3Dpts(7) & ", " & _

get3Dpts(8))

End Sub

6.1.3定义用户坐标系

用户坐标系UCS是一个相对于世界坐标系WCS的坐标系。它的原点是世界坐标系中的一个点，而它的X、Y和Z轴方向可以由用户确定。定义用户坐标系即是重新设置坐标系的原点（0，0，0）的位置以及XY平面和Z轴的方位。用户工作时只能使用一个当前的用户坐标系。

用户可以使用用户坐标系集合（UserCoordinateSystems）的Add方法创建新的用户坐标系。该方法需要四个参数：

（1） 原点坐标（UCS原点在WCS中的坐标）。

（2） X轴上一点（UCS的X轴通过的点在WCS中的坐标）。

（3） Y轴上一点（UCS的Y轴通过的点在WCS中的坐标）。

（4） UCS的名称（字符串）。

2．将已创建的用户坐标系设置为当前坐标系

方法是将Document对象中的ActiveUCS属性设置为该用户坐标系。

注：

1）如果对当前的UCS进行修改，新的UCS对象必须重置为当前UCS，才能看到修改的效果。

2）使用当前视口对象的UCSIconatOrigin属性（设为True），可将UCS图标显示在UCS原点上。（如果已打开了UCS图标但不是显示在原点，则是显示在UCSORG系统变量所定义的WCS坐标中）。

例6-2：创建新的用户坐标系并设置为当前的UCS。

Sub Ch6_NewUCS( )

Dim NewUCSobject As AcadUCS

Dim UCSorigin(0 To 2) As Double

Dim UCSXAxisPoint(0 To 2) As Double

Dim UCSYAxisPoint(0 To 2) As Double

UCSorigin(0) = 40: UCSorigin(1) = 50: UCSorigin(2) = 30

UCSXAxisPoint(0) = 50

UCSXAxisPoint(1) = 50

UCSXAxisPoint(2) = 30

UCSYAxisPoint(0) = 40

UCSYAxisPoint(1) = 60

UCSYAxisPoint(2) = 30

’创建UCS

Set NewUCSobject = ThisDrawing.UserCoordinateSystems _

.Add(UCSorigin, UCSXAxisPoint, UCSYAxisPoint, "新建UCS")

’显示UCS图标

ThisDrawing.ActiveViewport.UCSIconAtOrigin = True

’设置当前UCS

ThisDrawing .ActiveUCS = NewUCSobject

MsgBox "当前UCS为：" & AcadDoc.ActiveUCS.Name

End Sub

6．2 创建三维实体对象

6.2.1根据基本实体形创建三维实体

1．AddBox方法

创建边缘与WCS轴平行的3D矩形体。

语法：RetVal=object.AddBox(Origin,Lenglh,Width,Height)

式中：

object——指模型空间、图纸空间或块

Origin——3D WCS坐标值，指定矩形体的原点，该坐标代表对象边缘界框的中心点（不是角点）

Lenglh——矩形体的长度

Width——矩形体的宽度

Height——矩形体的高度

RetVal——新建的3DSolid对象

2．AddCone方法 

创建一个底面在WCS XY平面上的圆锥体。

语法：RetVal=object.AddCone(Center,BaseRadius,Height)

式中：

object——指模型空间、图纸空间或块

Center——3D WCS坐标值，指定边界框的中心点

BaseRadius——圆锥体底面的半径

Height——圆锥体的高度

RetVal——新建的3DSolid对象

3．AddCylinder 方法 

创建一个底面在WCS XY平面上的圆柱体。

语法：RetVal=object.AddCylinder(Center,BaseRadius,Height)

式中：

object——指模型空间、图纸空间或块

Center——3D WCS坐标值，指定边界框的中心点

BaseRadius——圆柱体的半径

Height——圆柱体的高度

RetVal——新建的3DSolid对象

4． AddEllipticalCone 方法 

以给定的Center、MajorRadius、MinorRadius以及Height在WCS XY平面上创建椭圆锥体。

5． AddEllipticalCylinder 方法 

以给定的Center、MajorRadius、MinorRadius以及Height在WCS XY平面上创建椭圆柱体。

6． AddSphere 方法 

以给定的Center和Radius创建球体。

7． AddTorus 方法 

以给定的圆环中心点、圆环半径和圆管半径创建圆环体。

8． AddWedge 方法 

以给定的楔形面中心点、与X轴对应的楔形表面的长度、与Y轴对应的楔形表面的宽度和与Z轴对应的楔形表面的高度创建楔形体。

6.2.2使用AddExtrudeSolid 方法创建三维实体

该方法以给定的参数Profile、Height以及TaperAngle创建三维实体。

语法：RetVal= object.AddExtrudedSolid(Profile, Height, TaperAngle)

式中：

object——模型空间、图纸空间或块

Profile——只适用于面域Region对象

Height——沿Z轴的拉伸高度（可正或负）

TaperAngle——拉伸的锥角（弧度）。其默认值为零。当输入正值时，从基准开始逐渐变尖

RetVal——新建立的拉伸3D Solid实体对象

例6-3：先由圆弧和直线生成面域，再按高度拉伸为三维实体。

Sub Ch6_AddExtrudedSolid( )

Dim curves(0 To 1) As AcadEntity

Dim centerPoint(0 To 2) As Double

Dim radius As Double

Dim startAngle As Double

Dim endAngle As Double

centerPoint(0) = 85#: centerPoint(1) = 60#: centerPoint(2) = 0#

radius = 20#

startAngle = 0

endAngle = 3.141592

Set curves(0)=ThisDrawing.ModelSpace.AddArc(centerPoint, _ 

radius, startAngle, endAngle)

Set curves(1) = ThisDrawing.ModelSpace.AddLine _

(curves(0).StartPoint, curves(0).EndPoint)

’生成面域

Dim regionObj As Variant

regionObj = ThisDrawing.ModelSpace.AddRegion(curves)

’设定拉伸参数

Dim height As Double

Dim taperAngle As Double

height = 30

taperAngle = 0.15

’拉伸为三维实体

Dim solidObj As Acad3DSolid

Set solidObj = ThisDrawing.ModelSpace.AddExtrudedSolid _

(regionObj(0), height, taperAngle)

ZoomAll

ThisDrawing.Regen acAllViewports

End Sub

该通用过程的运行结果如图6-1所示。

图6-1按高度拉伸面域生成三维实体

6.2.3使用AddExtrudedSolidAlongPath 方法创建三维实体

该方法以给定的Profile及拉伸路径Path创建三维实体。

语法：RetVal = object.AddExtrudedSolidAlongPath(Profile, Path)

式中：

object——指模型空间、图纸空间或块

Profile——只适用于面域Region对象

Path—只限于Polyline、Circle、Ellipse、Spline或Arc对象

注：面域Region对象与拉伸路径不可以在同一平面内，也不可以有高曲率的面域。

例6-4：先生成一个面域，再定义一条样条曲线，最后沿该曲线将面域拉伸成三维实体。

Sub Ch6_AddExtrudedSolidAlongPath( )

’定义面域

Dim curves(0 To 1) As AcadEntity

Dim centerPoint(0 To 2) As Double

Dim radius As Double

Dim startAngle As Double

Dim endAngle As Double

centerPoint(0) = 5#: centerPoint(1) = 3#: centerPoint(2) = 0#

radius = 2#

startAngle = 0

endAngle = 3.141592

Set curves(0)=ThisDrawing.ModelSpace.AddArc(centerPoint,_ 

radius, startAngle, endAngle)

Set curves(1) = ThisDrawing.ModelSpace.AddLine _

(curves(0).startPoint, curves(0).endPoint)

Dim regionObj As Variant

regionObj = ThisDrawing.ModelSpace.AddRegion(curves)

’定义路径 (样条曲线)

Dim splineObj As AcadSpline

Dim startTan(0 To 2) As Double

Dim endTan(0 To 2) As Double

Dim fitPoints(0 To 8) As Double

startTan(0) = 10: startTan(1) = 10: startTan(2) = 10

endTan(0) = 10: endTan(1) = 10: endTan(2) = 10

fitPoints(0) = 0: fitPoints(1) = 10: fitPoints(2) = 10

fitPoints(3) = 10: fitPoints(4) = 10: fitPoints(5) = 10

fitPoints(6) = 15: fitPoints(7) = 10: fitPoints(8) = 10

Set splineObj=ThisDrawing.ModelSpace.AddSpline _

(fitPoints,startTan, endTan)

’生成三维实体

Dim solidObj As Acad3DSolid

Set solidObj=ThisDrawing.ModelSpace.AddExtrudedSolidAlongPath _

(regionObj(0), splineObj)

ZoomAll

End Sub

该通用过程的运行结果如图6-2所示。

图6-2沿曲线路径拉伸面域生成三维实体的消隐图

6.2.4利用AddRevolvedSolid方法创建三维实体

该方法以给定的Profile沿指定轴旋转创建三维实体。

语法：RetVal = object.AddRevolvedSolid(Profile, AxisPoint, AxisDir, Angle)

式中：

object——指模型空间、图纸空间或块

Profile——只适用于面域Region对象

AxisPoint——3D WCS坐标值，指旋转轴的起点

AxisDir——指定旋转轴方向的3D向量

Angle——旋转的角度（弧度），输入6.28表示旋转一周

RetVal——新建立的旋转3D Solid实体对象

例6-5：先由圆弧和直线生成面域，再绕指定的轴线旋转形成三维实体。

Sub Ch6_AddRevolvedSolid( )

Dim curves(0 To 1) As AcadEntity

’生成弧

Dim centerPoint(0 To 2) As Double

Dim radius As Double

Dim startAngle As Double

Dim endAngle As Double

centerPoint(0) = 50#: centerPoint(1) = 230#: centerPoint(2) = 0#

radius = 20#

startAngle = 0

endAngle = 3.1415926

Set curves(0)=ThisDrawing.ModelSpace.AddArc(centerPoint, _ 

radius, startAngle, endAngle)

’生成直线

Set curves(1) = ThisDrawing.ModelSpace.AddLine _

(curves(0).StartPoint, curves(0).EndPoint)

’创建面域

Dim regionObj As Variant

regionObj = ThisDrawing.ModelSpace.AddRegion(curves)

’确定旋转轴线及旋转角

Dim axisPt(0 To 2) As Double

Dim axisDir(0 To 2) As Double

Dim angle As Double

axisPt(0) = 20: axisPt(1) = 201: axisPt(2) = 0

axisDir(0) = 0: axisDir(1) = 1: axisDir(2) = 0

angle = 6.28

’旋转面域形成三维实体

Dim solidObj As Acad3DSolid

Set solidObj = ThisDrawing.ModelSpace.AddRevolvedSolid _

(regionObj(0), axisPt, axisDir, angle)

solidObj.Update

End Sub

该通用过程的运行结果如图6-3所示。

图6-3旋转面域生成三维实体的消隐图

6．3 在三维空间中编辑

6.3.1在三维空间中旋转对象

要旋转三维对象，既可以使用三维实体对象的 Rotate 方法，也可以使用 Rotate3D 方法。使用 Rotate 方法可以在二维空间中绕指定的基点旋转对象，旋转的方向由 WCS 确定。而使用 Rotate3D 方法可以在三维空间中绕指定轴旋转对象。

使用Rotate3D方法，需要输入三个参数：

（1）旋转轴的第一点3D WCS坐标。

（2）旋转轴的第二点3D WCS坐标。

（3）旋转角度（弧度）。 

例6-6：先创建一个三维长方体，然后定义旋转轴，将长方体绕该轴旋转 30 度。 

Sub Ch6_RotateSolid( )

Dim boxObj As Acad3DSolid

Dim length As Double

Dim width As Double

Dim height As Double

Dim center(0 To 2) As Double

’定义长方体

center(0) = 5: center(1) = 5: center(2) = 0

length = 5

width = 7

height = 10

’在模型空间中创建长方体对象

Set boxObj = ThisDrawing.ModelSpace. _

AddBox(center, length, width, height)

’用两点定义旋转轴

Dim rotatePt1(0 To 2) As Double

Dim rotatePt2(0 To 2) As Double

Dim rotateAngle As Double

rotatePt1(0) = -3: rotatePt1(1) = 4: rotatePt1(2) = 0

rotatePt2(0) = -3: rotatePt2(1) = -4: rotatePt2(2) = 0

rotateAngle = 30

rotateAngle = rotateAngle * 3.141592 / 180#

’旋转长方体

boxObj.Rotate3D rotatePt1, rotatePt2, rotateAngle

ZoomAll

End Sub

6.3.2在三维空间中创建阵列

使用ArrayRectangular方法可在三维空间创建三维矩形阵列，该方法需要六个参数，依次为：列数（X方向）、行数（Y方向）、层数（Z方向）、列距、行距和层距。

例6-7：创建一个球，然后用该球创建4 列、5 行、3 层的矩形阵列。

Sub Ch6_CreateRectangularArray( )

’创建三维实体——球

Dim SphereObj As Acad3DSolid

Dim pointC(0 To 2) As Double

Dim rr As Double

pointC(0) = 20: pointC(1) = 20: pointC(2) = 0

rr = 3

Set SphereObj = ThisDrawing.ModelSpace.AddSphere (pointC, rr) 

SphereObj.Update

’定义矩形阵列

Dim RowNum As Long

Dim ColNum As Long

Dim LevNum As Long

Dim DisBwtnRos As Double

Dim DisBwtnCol As Double

Dim DisBwtnLev As Double

RowNum = 4

ColNum = 5

LevNum = 3

DisBwtnRos = 10

DisBwtnCol = 10

DisBwtnLev = 30

’形成球的矩形阵列

Dim RetObject As Variant

RetObject =SphereObj.ArrayRectangular _

(RowNum,ColNum,LevNum, DisBwtnRos, DisBwtnCol, DisBwtnLev)

ZoomAll

End Sub

6.3.3在三维空间中镜像对象

使用 Mirror3D 方法可以在三维空间中沿三点指定的特定镜像平面来镜像对象。

例6-8：在模型空间中创建一个长方体，然后根据某个平面镜像该长方体，并将镜像得到的长方体着色为红色。 

Sub Ch6_MirrorABox3D( )

’创建长方体对象

Dim boxObj As Acad3DSolid

Dim length As Double

Dim width As Double

Dim height As Double

Dim center(0 To 2) As Double

center(0) = 5#: center(1) = 5#: center(2) = 0

length = 5#: width = 7: height = 10#

’在模型空间中创建长方体 3DSolid对象

Set boxObj = ThisDrawing.ModelSpace.AddBox(center, length,width,height)

’用三个点定义镜像平面

Dim mirrorPt1(0 To 2) As Double

Dim mirrorPt2(0 To 2) As Double

Dim mirrorPt3(0 To 2) As Double

mirrorPt1(0) = 60: mirrorPt1(1) = 0: mirrorPt1(2) = 0

mirrorPt2(0) = 60: mirrorPt2(1) = 2: mirrorPt2(2) = 0

mirrorPt3(0) = 60: mirrorPt3(1) = 2: mirrorPt3(2) = 2

’镜像长方体

Dim mirrorBoxObj As Acad3DSolid

Set mirrorBoxObj = boxObj.Mirror3D (mirrorPt1, mirrorPt2, mirrorPt3)

mirrorBoxObj.Color = acRed

ZoomAll

End Sub

6．4 编辑三维实体

6.4.1布尔运算和检查干涉

创建三维实体之后，可以通过组合操作来创建更复杂的形状。可以使用 Boolean 或 CheckInterference 方法完成这些组合操作。

1．利用Boolean方法进行布尔运算

该方法可在三维实体（或面域）之间进行并、交、差运算。

语法：object.Boolean(Operation, Object)

式中：

object——3DSolid、Region对象

（该方法适用的对象）

Operation——AcBooleanType：枚举，只可输入

AcUnion：并运算

AcIntersection：交运算

AcSubtraction：差运算

Object——执行操作的对象

操作的结果，第一个对象会被改变。

例6-9：先生成一个长方体和一个圆柱，然后进行布尔运算。

Sub Ch6_Boolean( )

Dim boxObj As Acad3DSolid

Dim boxLength As Double

Dim boxWidth As Double

Dim boxHeight As Double

Dim boxCenter(0 To 2) As Double

boxCenter(0) = 5#: boxCenter(1) = 5#: boxCenter(2) = 0

boxLength = 10#: boxWidth = 7: boxHeight = 10#

SetboxObj=ThisDrawing.ModelSpace.AddBox(boxCenter, _

boxLength, boxWidth, boxHeight)

Dim cylinderObj As Acad3DSolid

Dim cylinderCenter(0 To 2) As Double

Dim cylinderRadius As Double

Dim cylinderHeight As Double

cylinderCenter(0) = 0#

cylinderCenter(1) = 0#

cylinderCenter(2) = 0#

cylinderRadius = 5#

cylinderHeight = 20#

Set cylinderObj = ThisDrawing.ModelSpace.AddCylinder _

(cylinderCenter, cylinderRadius, cylinderHeight)

’改变视口方向

Dim NewDirection(0 To 2) As Double

NewDirection(0) = -1: NewDirection(1) = -1: NewDirection(2) = 1

ThisDrawing.ActiveViewport.direction = NewDirection

ThisDrawing.ActiveViewport = ThisDrawing.ActiveViewport

ZoomAll

’执行交运算

MsgBox "执行交运算",vbOKOnly , "布尔运算例"

boxObj.Boolean acIntersection, cylinderObj

ThisDrawing.Regen True

MsgBox "交运算完成! ", , "布尔运算例"

End Sub

2．利用CheckInterference方法进行检查干涉

该方法检查两个三维实体之间的干涉。如果指定的话，则从干涉处创建一个三维实体。

语法：RetVal = object.CheckInterference (Object, CreateInterferenceSolid)

式中：

object——3DSolid对象。只可输入

（要检查的对象）

CreateInterferenceSolid——布尔值

TRUE：创建干涉检查实体

FALSE：不建立干涉检查实体

RetVal——3DSolid对象

（CreateInterferenceSolid设成TRUE时的结果对象）

例6-10：先创建一个长方体和一个圆柱体，然后检查是否发生干涉。

Sub Ch6_CheckInterference( ) 

Dim color As AcadAcCmColor

Set color = AcadApplication.GetInterfaceObject _

("AutoCAD.AcCmColor.16")

’定义长方体

Dim boxObj As Acad3Dsolid

Dim boxLength As Double

Dim boxWidth As Double

Dim boxHeight As Double

Dim boxCenter(0 To 2) As Double

boxCenter(0) = 5#: boxCenter(1) = 5#: boxCenter(2) = 0

boxLength = 10#: boxWidth = 7: boxHeight = 10#

’在模型空间中创建长方体对象

Set boxObj = ThisDrawing.ModelSpace.AddBox _

(boxCenter, boxLength, boxWidth, boxHeight)

Call color.SetRGB(80, 100, 244)

boxObj.TrueColor = color

’定义圆柱体

Dim cylinderObj As Acad3DSolid

Dim cylinderCenter(0 To 2) As Double

Dim cylinderRadius As Double

Dim cylinderHeight As Double

cylinderCenter(0) = 0#

cylinderCenter(1) = 0#

cylinderCenter(2) = 0#

cylinderRadius = 5#

cylinderHeight = 20#

’创建圆柱体并着色为青色

Set cylinderObj = ThisDrawing.ModelSpace.AddCylinder _

(cylinderCenter, cylinderRadius, cylinderHeight)

Call color.SetRGB(244, 150, 50)

cylinderObj.TrueColor = color

’找出两个实体之间的干涉

Dim solidObj As Acad3DSolid

Set solidObj = boxObj.CheckInterference(cylinderObj, True)

Call color.SetRGB(200, 150, 244)

solidObj.TrueColor = color

’改变视口方向

Dim NewDirection(0 To 2) As Double

NewDirection(0) = -1: NewDirection(1) = -1: NewDirection(2) = 1

ThisDrawing.ActiveViewport.direction = NewDirection

ThisDrawing.ActiveViewport = ThisDrawing.ActiveViewport

ZoomAll

End Sub

6.4.2三维实体的剖切和生成相交截面

1． 利用Slicesolid方法剖切三维实体

该方法以三个点定义的平面剖切实体，返回剖切后的三维实体。

语法：RetVal = object..SliceSolid(Point1, Point2, Point3, Negative)

式中：

object——3DSolid对象

（该方法适用的对象）

Point1——第一点的3D WCS 坐标

Point2——第二点的3D WCS 坐标

Point3——第三点的3D WCS 坐标

Negative——布尔值，指定是否以平面的负边返回结果实体

RetVal——剖切后的3DSolid 对象

例6-11：在模型空间中创建一个长方体，然后基于由三个点定义的平面剖切该长方体。剖切后的结果作为三维实体返回。 

Sub Ch6_SliceABox( )

’创建长方体对象

Dim boxObj As Acad3DSolid

Dim length As Double

Dim width As Double

Dim height As Double

Dim center(0 To 2) As Double

center(0) = 5#: center(1) = 5#: center(2) = 0

length = 5#: width = 7: height = 10#

’在模型空间中创建长方体 (3DSolid) 对象

Set boxObj = ThisDrawing.ModelSpace. AddBox(center, length, width, height)

boxObj.Color = acWhite

’用三点定义剖切平面

Dim slicePt1(0 To 2) As Double

Dim slicePt2(0 To 2) As Double

Dim slicePt3(0 To 2) As Double

slicePt1(0) = 1.5: slicePt1(1) = 7.5: slicePt1(2) = 0

slicePt2(0) = 1.5: slicePt2(1) = 7.5: slicePt2(2) = 10

slicePt3(0) = 8.5: slicePt3(1) = 2.5: slicePt3(2) = 10

’剖切长方体并将新实体着色为红色

Dim sliceObj As Acad3DSolid

Set sliceObj = boxObj.SliceSolid (slicePt1, slicePt2, slicePt3, True)

sliceObj.Color = acRed

ZoomAll

End Sub

2．利用SectionSolid方法得到实体的相交截面

该方法利用三点定义的平面与三维实体相交，所形成的交集以面域返回。

语法：RetVal = object.SectionSolid (Point1, Point2, Point3)

式中：

object——3Dsolid对象

Point1——指定的第一点3D WCS坐标值

Point2——指定的第二点3D WCS坐标值

Point3——指定的第三点3D WCS坐标值

RetVal——作为剖面轮廓的Region对象

例6-12：在模型空间中创建一个长方体，然后基于由三个点定义的平面剖切该长方体，所形成的交集以面域返回。

Sub Ch6_SectionSolid( )

Dim solid1 As Acad3DSolid

Dim point1(0 To 2) As Double

Dim point2(0 To 2) As Double

Dim point3(0 To 2) As Double

point1(0) = -5:point1(1) = 10:point1(2) = -15

point2(0) = -5:point2(1) = 10:point2(2) = 15

point3(0) = 5:point3(1) = -10:point3(2) = 15

Dim cenPnt(0 To 2) As Double

cenPnt(0) = 0#:cenPnt(1) = 0#:cenPnt(2) = 0#

’生成六方体

Set solid1 = ThisDrawing.ModelSpace.AddBox _

(cenPnt, 10, 15, 20)

’用三点定位的斜平面剖切六方体形成剖切截面交线

Dim sectionobj As AcadRegion

Set sectionobj = solid1.SectionSolid(point1, point2, point3)

sectionobj.Color = acRed

sectionobj.Update

’改变视口观察方向

Dim NewDirection(0 To 2) As Double

NewDirection(0) = -1: NewDirection(1) = -1: NewDirection(2) = 1

ThisDrawing.ActiveViewport.direction = NewDirection

ThisDrawing.ActiveViewport = ThisDrawing.ActiveViewport

ZoomAll

End Sub

6．5 三维实体创建与编辑的综合示例

6.5.1三维实体综合示例的用户界面

三维实体综合示例的客户程序是利用VB开发的，其窗体如图6-4所示。操作时，应先单击《引用AutoCAD并建立图形文件》按钮，实现与AutoCAD的连接和建立新图形文件。然后按需要单击各项功能的命令按钮，进行三维实体的创建及编辑的演示。

图6-4三维实体综合示例的窗体

6.5.2三维实体综合示例的源程序

该综合示例的源程序收录在本书附带的光盘中，本节只介绍其中《布尔运算综合示例》事件过程的功能和编程方法。

该例的运行结果如图6-5所示。

图6-5经布尔运算生成的组合三维实体

该例左边的三维实体是先利用面域通过差运算，生成组合面域，然后再运用AddExtrudedSolid方法拉伸生成三维实体。右边的三维实体是先分别生成长方体和圆柱体，然后进行并运算最后生成组合三维实体。

该事件过程调用的通用过程源程序代码：

Sub Ch6_LBoolean( )

’利用面域进行布尔运算，再生成三维实体

’生成优化多段线

Dim points(0 To 9) As Double

Dim LWP As AcadLWPolyline

points(0) = 102: points(1) = 2

points(2) = 115: points(3) = 2

points(4) = 110.5: points(5) = 15

points(6) = 102: points(7) = 15

points(8) = 102: points(9) = 2

Set LWP = AcadDoc.ModelSpace.AddLightWeightPolyline(points)

Const n = 5

ReDim anObjM(n) As Object

Dim centerPoint(0 To 2) As Double

Dim radius As Double

Dim l As Double

l = 20

Dim point1(0 To 2) As Double

Dim point2(0 To 2) As Double

point1(0) = 200#

point1(1) = 100#

point1(2) = 0#

point2(0) = point1(0) + l

point2(1) = 100#

point2(2) = 0#

Dim i As Integer

Dim curves1(0 To 0) As AcadEntity

Dim curves2(0 To 0) As AcadEntity

’生成圆

centerPoint(0) = 160#

centerPoint(1) = -20#

centerPoint(2) = 0#

radius = 60#

Set curves1(0) = AcadDoc.ModelSpace.AddCircle(centerPoint, radius)

Set curves2(0) = LWP

Dim regions1 As Variant

Dim regions2 As Variant

’生成面域

regions1 = AcadDoc.ModelSpace.AddRegion(curves1)

regions2 = AcadDoc.ModelSpace.AddRegion(curves2)

curves1(0).Erase

curves2(0).Erase

’执行一个圆面域和由优化多段线形成的5个面域之间的差运算

MsgBox "执行一个圆面域和由优化多段线形成的5个面域之间的 _

差运算", vbOKOnly, "布尔运算例"

For i = 1 To n

Set anObjM(i) = regions2(0).Copy

Call regions2(0).Move(point1, point2)

Call regions1(0).Boolean(acSubtraction, anObjM(i))

point1(0) = point1(0) + l

point2(0) = point2(0) + l

Next i

regions2(0).Erase

regions1(0).Update

Dim h As Long

Dim t As Long

h = 40

t = 0

’沿直线拉伸组合面域生成三维实体

MsgBox "执行沿直线拉伸组合面域生成三维实体", vbOKOnly, "布尔运算例"

Dim anObj As Acad3DSolid

Set anObj = AcadDoc.ModelSpace.AddExtrudedSolid (regions1(0), h, t)

anObj.Update

’改变视口方向

Dim NewDirection(0 To 2) As Double

NewDirection(0) = -1: NewDirection(1) = 1: NewDirection(2) = 1

AcadDoc.ActiveViewport.Direction = NewDirection

AcadDoc.ActiveViewport = AcadDoc.ActiveViewport

AcadDoc.Regen True

ZoomAll

’在模型空间生成一个长方体和一个圆柱体

Dim boxObj As Acad3DSolid

Dim boxLength As Double

Dim boxWidth As Double

Dim boxHeight As Double

Dim boxCenter(0 To 2) As Double

boxCenter(0) = 30#: boxCenter(1) = 30#: boxCenter(2) = 200

boxLength = 20#: boxWidth = 17: boxHeight = 20#

Set boxObj = AcadDoc.ModelSpace.AddBox _

(boxCenter, boxLength, boxWidth, boxHeight)

Dim cylinderObj As Acad3DSolid

Dim cylinderCenter(0 To 2) As Double

Dim cylinderRadius As Double

Dim cylinderHeight As Double

cylinderCenter(0) = 20#: cylinderCenter(1) = 20#

cylinderCenter(2) = 200#

cylinderRadius = 10#

cylinderHeight = 40#

Set cylinderObj = AcadDoc.ModelSpace.AddCylinder _

(cylinderCenter, cylinderRadius, cylinderHeight)

’执行一个长方体和一个圆柱体的并运算

MsgBox "执行一个长方体和一个圆柱体的并运算", vbOKOnly, "布尔运算例"

boxObj.Boolean acUnion, cylinderObj

AcadDoc.Regen True

ZoomAll

End Sub

第7章 应用程序实例

—参数化自动绘制机械零件图和三维实体

7.1 概述

本章简介几例利用AutoCAD ActiveX VBA/VB 二次开发技术开发出的参数化绘制机械零件图和三维实体的应用程序。这几例应用程序是用VB6.0开发的，其通用性较强，便于编译为可执行文件。程序的组成结构，都只用了一个窗体模块。使用时，只需按照用户界面上的指示和要求输入必要的原始参数，即可自动绘制出需要的图形。若绘出的结果不满意，应改换参数后再重新绘制。也可在AutoCAD绘图环境下直接进行细微的修改，如个别移动某图形实体的位置，直到完全满意为止。最后可将包括已知条件和计算结果的窗体界面打印出来。绘制出的图形文件由AutoCAD打印和输出。

7.1.1选择或设计样板图

开发机械零件图形的参数化绘图应用程序，首先要考虑绘图标准、图形幅面、绘图单位、标题栏、通用的零件特性表以及文字样式和标注样式等。选择或设计一张（套）样板图是一种简便有效的方法。根据样板图生成的图形文件，将能适用于该类图形的绘图标准和绘图环境。

AutoCAD中文版带有多种式样的样板图，如：acad.dwt，acadiso.dwt和适用于我国国家标准的样板图Gb_a3 –Named Plot Styles.dwt等。若已有的样板图不能满足使用需要，可利用中文版的通用样板图更改设计为专用的样式，然后再按.dwt存盘即可。应该注意的是，如能满足需要，要侭量采用低版本的通用样板图更改设计为专用的样板图。这样设计出的专用样式样板图，能适用于从低到高各种版本的AutoCAD绘图软件。

7.1.2设计应用程序的界面和编写应用程序的代码

1.设计窗体的用户界面

如前所述，窗体是设计应用程序的一个基本平台，几乎所有的控件都是添加在窗体上的，而大多数应用程序也都是由窗体开始执行的，是与用户交互操作的窗口。窗体有自己的属性、方法和事件。要设置窗体的属性，可在设计阶段，修改“属性”窗口中的缺省值；也可在程序运行时由代码来实现。当向窗体上添加控件后，便可设置控件的属性，其实现的方法与窗体一样也是上述的两种方式。

2.编写事件过程的程序代码

如前所述，AutoCAD VBA工程文件一般包括：AutoCAD对象、窗体模块、标准模块、类模块和引用等。最常用的是窗体模块，它可以包括描述窗体及其控件的各种事件过程、通用过程和变量的窗体模块级声明。

通用的参数化自动绘图事件过程的程序框图如图7-1所示。

图7-1自动绘图事件过程的程序框图

注：

1） 该框图适用于在VB环境下编写客户程序。若利用AutoCAD VBA编程，应省去“引用AutoCAD部件”部分。

2） ?AutoCAD VBA 工程文件（.dvb）与独立的 VB6 工程文件（.bvp）不兼容。 但可以通过在 VBA 集成开发环境 (IDE) 里使用 IMPORT 和 EXPORT 命令在不同的工程文件之间交换窗体、模块和类。

7.1.3 VBA和VB程序代码样例

例7-1: 本例示范应用 VBA 生成一条直线。

Sub Ch7_AddLineVBA()

Dim lineObj As AcadLine

Dim startPoint(0 To 2) As Double

Dim endPoint(0 To 2) As Double

' 定义直线的起点和端点

startPoint(0) = 1

startPoint(1) = 1

startPoint(2) = 0

endPoint(0) = 5

endPoint(1) = 5

endPoint(2) = 0

' 在模型空间中创建直线

Set lineObj = ThisDrawing.ModelSpace.AddLine(startPoint, endPoint)

ZoomAll

End Sub

例7-2: 本例示范使用 VB 生成一条直线。程序中 CreateObject 或 GetObject 函数使用的是依赖版本的 ProgID。

Sub Ch7_AddLineVB()

On Error Resume Next

' 连接至 AutoCAD 应用程序

Dim acadApp As AcadApplication

Set acadApp = GetObject(,"AutoCAD.Application.16")

If Err Then

Err.Clear

Set acadApp = CreateObject("AutoCAD.Application.16")

If Err Then

MsgBox Err.Description

Exit Sub

End If

End If

' 连接至 AutoCAD 图形

Dim acadDoc As AcadDocument

Set acadDoc = acadApp.ActiveDocument

' 创建直线的端点

Dim lineObj As AcadLine

Dim startPoint(0 To 2) As Double

Dim endPoint(0 To 2) As Double

startPoint(0) = 1

startPoint(1) = 1

startPoint(2) = 0

endPoint(0) = 5

endPoint(1) = 5

endPoint(2) = 0

' 在模型空间中创建 Line 对象

Set lineObj = acadDoc.ModelSpace.AddLine _

(startPoint, endPoint)

ZoomAll

acadApp.visible = True

End Sub

7.2 齿轮零件图

本例为参数化绘制圆柱齿轮零件图的应用程序。使用了圆柱齿轮零件图专用的样板图。该样板图是利用在AutoCAD2000中文版中的国标（GB）样板图改动设计的，主要是添加了固定格式的齿轮参数表。

注：对于固定格式的齿轮参数表等设计内容，也可采取表格或图块的方式进行处理。

7.2.1窗体的用户界面和绘图结果

该例的窗体模块如图7-2所示。

图7-2参数化绘制圆柱齿轮零件图的用户界面

使用时，根据用户界面的指示，利用文本框TextBox和组合框ComboBox等输入已知条件及参数。应先单击命令按钮《计算齿轮参数》，这时可看到分组框《计算结果》中的齿轮尺寸参数及与输入的轴孔直径相对应的键槽数据，如图7-2所示。运行参数化自动绘图前，应按用户界面的提示，输入其它的绘图参数，如公差及齿轮精度等级等，并按出图的规格选好样板图（包括路径）。然后单击命令按钮《画齿轮零件图》使程序运行。由AutoCAD自动绘出的齿轮零件图，如图7-3所示。

注：由图7-3所显示的工程图是计算机显示器所显示的图像，当由打印机或绘图仪出图时，将显示轮廓线线宽。

可单击用户界面上的《打印窗体》按钮，打印出该窗体界面上显示的已知条件和计算结果。若单击《退出》按钮，程序结束。

图7-3圆柱齿轮零件图

7.2.2典型的程序段

1．程序框图

图7-4绘制圆柱齿轮事件过程的程序框图

2.典型的程序段

典型的程序段如例7-3和例7-4所述。

例7-3：引用AutoCAD部件对象，程序中 CreateObject 或 GetObject 函数使用的是独立于版本的 ProgID。并利用样板图创建图形文件。

Sub Ch7_ConnectToAcad( )

Option Explicit

Dim acadApp As AcadApplication

Dim acadDocs As AcadDocuments

Dim acadDoc As AcadDocument

On Error Resume Next 

Set acadApp = GetObject(,"AutoCAD.Application")

If Err Then

Err.Clear

Set acadApp = CreateObject("AutoCAD.Application")

If Err Then

MsgBox "Error to connect to AutoCAD"

Exit Sub

End If

End If

acadApp.Visible = True

acadApp.WindowState=acMax

Set acadDocs = acadApp.Documents

Dim dwtName As String '样板图文件名及路径

dwtName = Combo3.Text & Combo4.Text 

If Dir(dwtName) < > " " Then

acadDocs.Add dwtName

Else

MsgBox "文件 " & dwtName & " 不存在!"

acadDocs.Add

End If

Set acadDoc = acadApp.ActiveDocument

acadDoc.WindowState = acMax

End Sub

例7-4：增设图层及设置各层的线宽、线型和颜色。

Sub Ch7_Layer( )

Dim layer0 As AcadLayer

Dim layer1 As AcadLayer

Dim layer2 As AcadLayer

Dim layer3 As AcadLayer

Dim layer4 As AcadLayer

Set layer0 = acadApp.ActiveDocument.Layers.Item(0)

Set layer1 = acadApp.ActiveDocument.Layers.Add("轮廓线层")

Set layer2 = acadApp.ActiveDocument.Layers.Add("中心线层")

Set layer3 = acadApp.ActiveDocument.Layers.Add("剖面线层")

Set layer4 = acadApp.ActiveDocument.Layers.Add("标注及文字层")

layer1.Lineweight = acLnWt080 '轮廓线的线宽

'设置各层颜色

layer1.color = acWhite

layer2.color = acRed

layer3.color = acCyan

layer4.color = acBlue

'加载中心线线型，设置中心线层的线型

On Error Resume Next

acadDoc.Linetypes.Load "centerx2", "acad.lin"

layer2.Linetype = "centerx2"

End Sub

7.3 轴零件图

本例为参数化绘制齿轮减速器输出轴零件图的应用程序。绘制轴零件图前注意选择（或确认）国标通用样板图的图幅。

7.3.1窗体的用户界面和绘图结果

该应用实例的用户界面如图7-5所示。将该轴分为12个断面（站），其中的1、4、8、12为轴的集中载荷的部位。使用者可根据要求的轴结构键入各段轴长和各站轴径。

图7-5参数化绘制轴零件图的用户界面

使用时，首先应确定轴的结构形式。其轴上齿轮轴向定位的缺省形式为《右轴肩》，如图7-5中所示。若要设计为《左轴肩》，应利用组合框进行选择，然后参照用户界面上的轴结构简图，利用文本框输入各站轴径和各段轴长。当单击命令按钮《计算键槽参数》后，可计算出键槽参数，结果如分组框《计算结果》中所示。画图前，还需要输入其它的已知条件和结构参数。如轴端倒角和圆角尺寸、各尺寸公差和形位公差、材料及热处理要求、轴端中心孔规格等。单击《画轴零件图》命令按钮使绘图程序运行。由AutoCAD自动绘出的轴零件图，如图7-6所示。

注：由图7-6所显示的工程图是计算机显示器所显示的图像，当由打印机或绘图仪出图时，将显示轮廓线线宽。

可单击用户界面上的《打印窗体》按钮，打印出原始条件和计算结果。当单击《退出》按钮，该程序结束。

图7-6轴零件图

7.3.2典型的程序段

1.绘制带键槽的轴剖面

（1） 主要的设计思路：

绘制带键槽的轴剖面设计思路框图如图7-7所示。

图7-7绘制带键槽的轴剖面设计思路框图

（2） 源程序

例7-5：创建轴的横剖面

Sub Ch7_AddHatch( )

……

'用AddHatch方法创建轴剖面

acadDoc.ActiveLayer = layer3 '激活剖面线层

Dim patternName As String

Dim PatternType As Long

Dim bAssociativity As Boolean

Dim anObj As AcadHatch

patternName = "ANSI31"

PatternType = acHatchPatternTypePreDefined

Associate = True

Set anObj= acadDoc.ModelSpace.AddHatch _

(PatternType,patternName, Associate)

'确定剖面线外边框

Dim Outer(0 To 2) As AcadEntity

'创建辅助的轻便多段线

acadDoc.ActiveLayer = layer1 '激活轮廓线层

Dim LWP As AcadLWPolyline '按键槽尺寸定义一个辅助的矩形

Vert(0) = Center(0) - b / 2: Vert(1) = Center(1) + L01

Vert(2) = Vert(0): Vert(3) = Center(1) + L04

Vert(4) = Vert(0) + b: Vert(5) = Vert(3)

Vert(6) = Vert(0) + b: Vert(7) = Vert(1)

Set LWP = acadDoc.ModelSpace.AddLightWeightPolyline(Vert)

LWP.Closed = True '令该辅助的矩形封闭

'利用SetBulge方法编辑二维优化多段线的第二段,形成轴剖面的

'外轮廓

Dim h As Double

Dim bbb As Double

h = r + L04

bbb = h / (b / 2)

Call LWP.SetBulge(1, bbb)

LWP.Update

'添加剖面线

Dim OuterLoop(0 To 0) As AcadEntity

Set OuterLoop(0) = LWP

anObj.AppendOuterLoop (OuterLoop)

'计算并显示

anObj.Evaluate

……

End Sub

2．应用替换标注样式

例7-6：生成带上下偏差的直径尺寸标注。

该例只是一段程序，其中的变量已经赋值。

Sub Ch7_DimTextSample( )

……

Dim bz9 As AcadDimAligned

Set bz9 = AcadDoc.ModelSpace.AddDimAligned(N3, n30, n61)

’指定标注尺寸前缀，在直径值前加“Φ”

bz9.TextPrefix = "%%c"

’指定作为十进位标注和公差值中用作小数点分隔符的符号

bz9.DecimalSeparator = "." 

’指定公差是否与标注文字一起显示

bz9.ToleranceDisplay = acTolSymmetrical 

’指定标注中主公差值的精确度

bz9.TolerancePrecision = acDimPrecisionThree 

’指定公差值文字高度相对于标注文字高度的比例

bz9.ToleranceHeightScale = 0.5 

’指定标注尺寸的上偏差

bz9.ToleranceUpperLimit = scd4

’指定标注尺寸的下偏差

bz9.ToleranceLowerLimit = -xcd4

……

End Sub

7.4 V带轮零件图和三维实体

本例为参数化绘制V带轮零件图及其三维实体的应用程序。绘制V带轮二维零件图前，应选择（或确认）专用样板图文件的路径和图幅。生成V带轮三维实体时，采用缺省的通用样板图，与用户界面上标识的样板图无关。

7.4.1窗体的用户界面和绘图结果

窗体的用户界面如图7-8所示。使用时，应按用户界面上的指示从文本框和组合框输入已知条件，然后点击《计算带轮尺寸参数》按钮，在《计算结果》框内显示出计算结果后，即可运行绘图的事件过程。

生成的V带轮零件图和三维实体渲染图如图7-9和图7-10所示。

注：由图7-9所显示的工程图是计算机显示器所显示的图像，当由打印机或绘图仪出图时，将显示轮廓线线宽。

图7-8参数化自动绘制V带轮的用户界面

图7-9 V带轮零件图

图7-10 V带轮三维实体的渲染图

7.4.2典型的程序段

只摘录生成V带轮三维实体的一段源程序。

1．设计思路

生成V带轮三维实体的设计思路如图7-11所示。

图7-11生成V带轮三维实体的设计思路框图

2． 相关的一段源程序

例7-7：按旋转面域的方式创建V带轮的三维实体

Sub Ch7_3DSolidBeltWheel( )

……

'生成轴剖面

acadDoc.ActiveLayer = layer0

'利用二维多段线生成带轮环状坯轴剖面的封闭轮廓线

Dim LWPa As AcadLWPolyline 

Dim points(0 To 13) As Double 

points(0) = 0: points(1) = d / 2

points(2) = 0: points(3) = dw / 2

points(4) = (zz - 1) * e + 2 * f: points(5) = dw / 2

points(6) = (zz - 1) * e + 2 * f: points(7) = dw / 2 - h - 1.2 * kl

points(8) = L: points(9) = points(7)

points(10) = L: points(11) = d / 2

points(12) = 0: points(13) = d / 2

Set LWPa =acadDoc. ModelSpace.AddLightWeightPolyline(points)

'利用二维优化多段线生成V带环形槽轴剖面的封闭轮廓线

Dim points1(0 To 9) As Double

Dim LWP As AcadLWPolyline

points1(0) = f - kl * Tan(17 * 3.14 / 180): points1(1) = dw / 2 - hf

points1(2) = f - (hf + ha + kl) * Tan(17 * 3.14 / 180)

points1(3) = dw / 2 + ha

points1(4) = f + (hf + ha + kl) * Tan(17 * 3.14 / 180)

points1(5) = dw / 2 + ha

points1(6) = f + kl * Tan(17 * 3.14 / 180): points1(7) = dw / 2 - hf

points1(8) = f - kl * Tan(17 * 3.14 / 180): points1(9) = dw / 2 - hf

Set LWP =acadDoc. ModelSpace.AddLightWeightPolyline(points1)

Dim point1(0 To 2) As Double

Dim point2(0 To 2) As Double

point1(0) = f

point1(1) = dw / 2

point1(2) = 0#

point2(0) = point1(0) + e

point2(1) = dw / 2

point2(2) = 0#

Dim i As Integer

Dim curves1(0 To 0) As AcadObject

Dim curves2(0 To 0) As AcadObject

Set curves1(0) = LWPa

Set curves2(0) = LWP

'生成面域

Dim regions1 As Variant

Dim regions2 As Variant

regions1 = acadDoc.ModelSpace.AddRegion(curves1)

regions2 = acadDoc.ModelSpace.AddRegion(curves2)

'复制多个V带槽断面的面域

ReDim anObjm(zz) As AcadObject

For i = 1 To zz

Set anObjm(i) = regions2(0).Copy 

Call regions2(0).Move(point1, point2)

Call regions1(0).Boolean(2, anObjm(i)) 

point1(0) = point1(0) + e

point2(0) = point2(0) + e

Next i

regions2(0).Erase

LWP.Erase

LWPa.Erase

'将面域沿旋转轴转一周生成V带轮三维实体

Dim point10(0 To 2) As Double

point10(0) = 0:point10(1) = 0:point10(2) = 0

Dim point20(0 To 2) As Double

point20(0) = 10:point20(1) = 0:point20(2) = 0

Dim revangle As Double

revangle = 6.2832

Dim anobjj As Acad3Dsolid

Set anobjj = acadDoc.ModelSpace.AddRevolvedSolid _

(regions1(0), point10, point20, revangle)

regions1(0).Erase

'键槽的设计

Dim startpoint4(0 To 2) As Double

Dim endpoint4(0 To 2) As Double

Dim redlobj4 As AcadLine

startpoint4(0) = 0

startpoint4(1) = d / 2 - ttt1

startpoint4(2) = -Bj / 2

endpoint4(0) = 0: endpoint4(1) = d / 2 + ttt1: endpoint4(2) = -Bj / 2

Set redlobj4 = acadDoc.ModelSpace.AddLine(startpoint4, endpoint4)

Dim startpoint5(0 To 2) As Double

Dim endpoint5(0 To 2) As Double

Dim redlobj5 As AcadLine

startpoint5(0) = 0: startpoint5(1) = endpoint4(1)

startpoint5(2) = endpoint4(2)

endpoint5(0) = 0: endpoint5(1) = d / 2 + ttt1: endpoint5(2) = Bj / 2

Set redlobj5 = acadDoc.ModelSpace.AddLine(startpoint5, endpoint5)

Dim startpoint6(0 To 2) As Double

Dim endpoint6(0 To 2) As Double

Dim redlobj6 As AcadLine

startpoint6(0) = 0: startpoint6(1) = endpoint5(1): 

startpoint6(2) = endpoint5(2)

endpoint6(0) = 0: endpoint6(1) = d / 2 - ttt1: endpoint6(2) = Bj / 2

Set redlobj6 = acadDoc.ModelSpace.AddLine(startpoint6, endpoint6)

Dim startpoint7(0 To 2) As Double

Dim endpoint7(0 To 2) As Double

Dim redlobj7 As AcadLine

startpoint7(0) = 0: startpoint7(1) = endpoint6(1)

startpoint7(2) = endpoint6(2)

endpoint7(0) = 0: endpoint7(1) = startpoint4(1)

endpoint7(2) = endpoint4(2)

Set redlobj7 = acadDoc.ModelSpace.AddLine(startpoint7, endpoint7)

ReDim line3obj(0 To 3) As AcadLine

Set line3obj(0) = redlobj4

Set line3obj(1) = redlobj5

Set line3obj(2) = redlobj6

Set line3obj(3) = redlobj7

Dim regions3 As Variant

regions3 = acadDoc.ModelSpace.AddRegion(line3obj)

Dim hh As Double

Dim tt As Double

hh = b + 200

tt = 0

′沿轴线拉伸形成平键（槽）的六方体

Dim anobb As Acad3DSolid

Set anobb = acadDoc.ModelSpace.AddExtrudedSolid(regions3(0), hh, tt)

′差运算，生成要求的V带轮三维实体造型

Call anobjj.Boolean(2, anobb) 

anobjj.Update

acadDoc.SendCommand ("_zoom e ")

anobb.Erase

regions3(0).Erase

line3obj(0).Erase

redlobj4.Erase

redlobj5.Erase

redlobj6.Erase

redlobj7.Erase

Dim dirpoint(0 To 2) As Double

dirpoint(0) = 5

dirpoint(1) = -4

dirpoint(2) = 3

acadApp.ActiveDocument.ActiveViewport.direction = dirpoint

acadApp.ActiveDocument.ActiveViewport= _

acadApp.ActiveDocument.ActiveViewport

acadApp.ZoomExtents

……

End Sub

第8章 《M-CAD 常用机械传动零件计算机辅助设计》软件包简介

《M-CAD常用机械传动零件计算机辅助设计》软件包具有机械设计求解器和参数化自动绘图两种功能。其绘图部分是利用VB编程、应用ActiveX自动化技术对 AutoCAD进行二次开发的综合应用实例。

8.1 M-CAD的功能和用途

M-CAD 包括机械传动装置的总体设计、电动机型号的选择、V带传动设计、套筒滚子链传动设计、圆柱和圆锥齿轮传动设计、普通圆柱蜗杆传动设计、减速器轴的设计以及滚动轴承的选用等多个相互独立的设计模块。本软件包备有设计和使用说明书。

使用者只须输入原始数据和条件，通过各设计模块用户界面上的组合框及其它控件，选用相应的设计参数（条件）和设计准则，即可立即得出设计结果。需要时，可打印出该设计题目的设计条件及设计结果。常用的标准数据已编入本软件包中，使用时由程序自动调用，使用者也可将需要的最新标准数据直接键入，然后进行设计计算。对于滚动轴承选用模块，其标准参数取自软件包中的数据库文件，使用者可随时对该数据库文件进行维护和更新。

本软件的参数化绘图部分应用ActiveX自动化技术，调用AutoCAD通用绘图软件，按设计出的参数并输入必要的结构及绘图参数后，便可自动绘出二维的工程用零件图(如齿轮、V带轮、链轮及减速器的轴等)或生成三维实体(如V带轮)。若自动生成的零件图存在个别不满意之处，还可直接操作已调出的AutoCAD通用软件对已生成的设计图进行修改和完善。

本软件可供高等院校机械类专业从事机械设计和AutoCAD应用开发的师生或企业的技术人员使用。

8.2 M-CAD的工作环境和安装方法

M-CAD适用于Windows 中文版。在操作系统中应装有AutoCAD 中文版，否则将不能利用本软件特有的参数化绘图功能。

注:AutoCAD 2000/2002 属R15版、AutoCAD 2004/2005/2006属R16版，对应有上述两种版本的编译软件。

M-CAD软件包放在一张光盘上，应安装到某盘的一个子目录内。软件包包括主应用程序及附属文件（数据库文件、样板图和使用说明等）。安装时应首先运行其中的文件setup.exe，按照屏幕上的提示就能顺利地完成软件包的安装。

8.3 M-CAD的使用方法

M-CAD是由1个标准模块和17个窗体模块构成的应用程序。为了便于用户操作，只设计两级用户窗体界面。当双击桌面上的图标时，程序开始运行。首先出现第一级用户窗体界面（这里称为“主窗体”），如图8-1所示。用户根据使用需要，单击主窗体模块用户界面上的命令按钮，即可进入第二级用户窗体界面。用毕后再返回主窗体模块用户界面。

图8-1 《 M-CAD常用机械传动零件计算机辅助设计》的主窗体模块用户界面

8.3.1机械传动装置总体设计

机械传动装置总体设计的主要内容包括选择电动机型号、确定总传动比、分配各级传动比以及计算各级传动装置的运动和动力参数。本软件共预置了六种传动装置类型：

（1）电动机—带传动—两级圆柱齿轮减速器—联轴器—工作机。

（2）电动机—带传动—单级圆柱齿轮减速器—联轴器—工作机。

（3）电动机—联轴器—单级圆柱齿轮减速器—开式圆柱齿轮传动—工作机。

（4）电动机—联轴器—普通圆柱蜗杆减速器—联轴器—工作机。

（5）电动机—带传动—两级圆柱齿轮减速器—链传动—工作机。

（6）电动机—带传动—圆锥圆柱齿轮减速器—链传动—工作机。

使用时首先在主用户窗体界面上按照《机械传动装置总体设计》分组框中的组合框选择传动装置的类型，如选定“电动机—带传动—两级圆柱齿轮减速器—联轴器—工作机”。然后单击命令按钮《机械传动装置总体设计》，即可进入下一级的《机械传动装置总体设计》窗体模块用户界面，如图8-2所示。

图8-2《机械传动装置总体设计》窗体模块用户界面

该项设计的原始条件有三种类型：

（1）已知工作阻力、速度和工作机鼓轮直径。

（2）已知工作阻力矩、速度和工作机鼓轮直径。

（3）已知功率和转速。

使用单选按钮进行选择。图8-2中所示的属第一种。

当按图示选择好有关的传动比及各级传动效率后，单击《设计计算》命令按钮，即可求出各级传动的功率、转矩、转速、传动比和需要的电动机型号及其额定功率。若不满意，可单击《清零》命令按钮。改换数据后再重新进行计算。

注：如前面所述，本软件的《机械传动装置总体设计》共有六种传动类型，图8-2所示的窗体模块用户界面属第一种，其它五种用法相类似，不再赘述。

8.3.2 V带传动设计

当单击主窗体模块用户界面上的命令按钮《V带传动设计》后，即出现《V带传动设计》窗体模块用户界面，如图8-3所示。

1.设计的已知条件

（1）传递的功率。

（2）主动轮转速。

（3）传动比。

（4）工作条件及传动位置要求。

（5）工作机的载荷特性。

（6）原动机的工作特性。

（7）每日工作时间。

2.设计的内容

（1）确定V带的截型 （包括普通V带和有效宽度制的窄V带）。

若选用了软件尚未预先编入的V带截型数据（如”Y”型），当单击《设计计算》按钮后，即会弹出对话框，说明”请按对话框的提示直接输入该型带的标准参数及数据”。

（2） 确定V带的计算长度、根数和传动中心距。

（3） 确定V带轮计算直径及其它结构尺寸。

（4） 确定V带的张紧力和压轴力。

图8-3 《V带传动设计》窗体模块用户界面

3.使用方法

设计时首先输入已知条件，再输入或选择窗体模块用户界面上提示的设计条件，然后即可进行设计计算。此时应单击《设计计算》按钮，正常情况下即可得出全部计算结果, 若带根数过多或过少，应更换带截型后再算。若结果不满意或计算失败（有对话框提示），应单击《清零》按钮，重新选择已知条件及设计条件，再进行新一轮的设计计算。

若键入了非标准的小带轮计算直径，程序会自动选取一个最相近的标准值替代。例如: 若键入，则自动生成。同样，计算出的也是标准值。若传动比的误差大于5%，程序会自动通报。

下列选项应用复选框选择：

（1）运行时直接键入V带数据。

（2）运行时直接键入中心距。

（3）运行时直接键入工作情况系数。

（4）当反复起动，工况恶劣时。

当得到满意的设计计算结果后，可单击《打印窗体》按钮，将全部已知条件和设计计算结果打印下来。按照用户界面或输入框的提示输入零件图名和必要的结构及绘图参数后，即可调用AutoCAD通用绘图软件自动绘图（若需要设计出V带轮零件图，应单击《画小带轮零件图》按钮或《画大带轮零件图》按钮；若需要设计出V带轮三维实体，应单击《生成小带轮三维实体》按钮或《生成大带轮三维实体》按钮）。若对该工作图不满意，可退出AutoCAD绘图软件，更改绘图参数后重新生成新的零件图，也可直接利用AutoCAD绘图软件进行修改，直到满意为止。

8.3.3圆柱齿轮传动设计

当单击主窗体模块用户界面上的命令按钮《圆柱齿轮传动设计》后，即出现《圆柱齿轮传动设计》窗体模块用户界面，如图8-4所示。

图8-4《圆柱齿轮传动设计》窗体模块用户界面

1. 圆柱齿轮传动设计的内容和已知条件

本模块适用于设计开式或闭式的外啮合渐开线标准圆柱齿轮传动。基本齿廓符合GB1356-88，模数符合GB/T1357-87标准模数的第一和第二系列。斜齿圆柱齿轮传动的中心距可按结尾为0或5、偶数及整数三种方式圆整，也可直接输入需要的任何数值。

设计的已知条件：

（1）传递的功率。

（2）主动轮转速。

（3）传动比。

（4）传动的用途及工作条件。

（5）工作机的载荷特性。

（6）原动机的工作特性。

（7）预期使用寿命。

2．使用方法

设计时首先输入已知条件，再输入或选择界面上提示的设计模式及条件后，即可进行设计计算，此时应单击《设计计算》按钮，正常情况下即可得出全部计算结果。若结果不满意或计算失败（有对话框提示，如螺旋角超出许用范围）应单击《清零》按钮，重新选择已知条件及设计条件，再进行新一轮的设计计算，直到满意为止。

得到满意的设计计算结果后，可单击《打印窗体》按钮，将全部已知条件和设计计算结果打印下来。若需要设计出齿轮零件图，请单击《画从动齿轮》按钮。按照模块界面或输入框的提示输入零件图名和必要的结构及绘图参数后，即可调用AutoCAD通用绘图软件自动绘出从动齿轮零件图。若对该零件工作图不满意，可退出AutoCAD绘图软件，更改绘图参数后重新生成新的零件图。也可直接利用AutoCAD绘图软件进行修改，直到满意为止。

自动绘出的圆柱齿轮零件图如图8-5所示。

图8-5圆柱齿轮零件图

8.3.4直齿圆锥齿轮传动设计

当单击主窗体模块用户界面上的命令按钮《直齿圆锥齿轮传动设计》后，即出现《直齿圆锥齿轮传动设计》窗体模块用户界面，如图8-6所示。

8-6《直齿圆锥齿轮传动设计》窗体模块用户界面

1. 直齿圆锥齿轮传动的设计内容和已知条件

本模块适用于轴交角为90°的渐开线标准直齿圆锥齿轮传动的设计，其基本齿廓符合GB12369-1990，模数符合GB12368-1990，轮齿几何尺寸按不等顶隙收缩齿计算。

设计的已知条件:

（1）传递的功率。

（2）主动轮转速。

（3）传动比。

（4）传动的用途及工作条件。

（5）工作机的载荷特性。

（6）原动机的工作特性。

（7）预期使用寿命。

2．使用方法

设计时首先输入已知条件，再输入或选择界面上提示的设计模式及条件后，即可进行设计计算，此时应单击《设计计算》按钮，正常情况下即可得出全部计算结果。若结果不满意或计算失败（有对话框提示）应单击《清零》按钮，重新选择已知条件及设计条件，再进行新一轮的设计计算，直到满意为止。

得到满意的设计计算结果后，可单击《打印窗体》按钮，将全部已知条件和设计计算结果打印下来。若需要设计出从动齿轮零件图，请单击《画从动锥齿轮》按钮。按照模块界面或输入框的提示输入零件图名和必要的结构及绘图参数后，即可调用AutoCAD通用绘图软件自动绘出从动锥齿轮零件图。若对该零件工作图不满意，可退出AutoCAD绘图软件，更改绘图参数后重新生成新的零件图。也可直接利用AutoCAD绘图软件进行修改，直到满意为止。

自动绘出的圆锥齿轮零件图如图8-7所示。

图8-7圆锥齿轮零件图

8.3.5普通圆柱蜗杆传动设计

当单击主窗体模块用户界面上的命令按钮《普通圆柱蜗杆传动设计》后，即出现《普通圆柱蜗杆传动设计》窗体模块用户界面，如图8-8(a)所示。

图8-8(a)《普通圆柱蜗杆传动设计》窗体模块用户界面

1.普通圆柱蜗杆传动的设计内容和已知条件

本模块适用于轴交角为90°的普通圆柱蜗杆传动的设计，蜗杆头数为1、2和4，齿形角α=20°、顶隙c=0.2m、齿顶高ha=m。既可设计蜗杆和蜗轮基本参数匹配符合GB/T 10085-1988的标准传动又可设计模数和蜗杆直径系数符合 GB/T 10088-1988的非标准传动（中心距结尾为0或5，或直接输入要求的中心距）。

1．设计的已知条件

（1）传递的功率。

（2）蜗杆转速。

（3）传动比。

（4）传动的用途及工作条件。

（5）工作机的载荷特性。

（6）原动机的工作特性。

（7）预期使用寿命。

（8）工作环境及散热条件。

2．使用方法

设计时首先输入已知条件，再输入或选择介面上提示的设计条件后，即可进行设计计算。此时应单击《设计计算》按钮，正常情况下即可得出全部计算结果。若结果不满意或计算失败（有对话框提示），应单击《清零》按钮，重新选择已知条件及设计条件，再进行新一轮的设计计算。

本程序未编入金属模铸造蜗轮有关的力学性能数据，需要时由使用者直接输入。本程序正常的设计范围为中心距<=500mm。

得到满意的设计计算结果后，可单击《打印窗体》按钮，将已知条件和设计计算结果打印出来。当要求绘制蜗轮零件图，请单击《绘制蜗轮零件图》按钮，结果如图8-8(b)所示。

图8-8(b) 蜗轮零件图

8.3.6套筒滚子链传动设计

当单击主用户窗体界面上的命令按钮《套筒滚子链传动设计》后，即出现《套筒滚子链传动设计》窗体模块用户界面，如图8-9所示。

图8-9《套筒滚子链传动设计》窗体模块用户界面

1．套筒滚子链传动的设计内容和已知条件

1）设计内容:

（1）确定链的节距和排数。

（2）确定传动中心距和链节数。

（3）确定链轮结构尺寸。

（4）确定压轴力。

（5）低速链传动的静强度计算。

（6）自动绘出主动链轮零件图。

2）已知条件:

（1）传递的功率。

（2）主动链轮转速。

（3）传动比。

（4）传动用途、位置要求及润滑条件。

（5）工作机的载荷特性。

（6）原动机的工作特性。

（7）每日工作时间。

2．使用方法

设计时首先输入已知条件，再输入或选择界面上提示的设计条件后，即可进行设计计算，此时应单击《设计计算》按钮，正常情况下即可得出全部计算结果。若结果不满意或计算失败（有对话框提示），应单击《清零》按钮，重新选择已知条件及设计条件，再进行新一轮的设计计算。

本软件已将选自GB/T1243.1-1983的常用滚子链的标准参数及数据编入软件，以备调用。若选用了软件未预先编入的型号（如”08B”等）,当单击《设计计算》按钮后，即会弹出对话框，按对话框的提示直接输入该型滚子链的标准参数及数据即可。

链轮轴向齿廓及尺寸符合GB/T 1244-1985的规定。

得到满意的设计计算结果后，可单击《打印窗体》按钮，将全部已知条件和设计计算结果打印下来。若需要设计出链轮零件图，应首先利用组合框选择链齿位置是《在对称中心线上》或《靠一侧》，并按工作界面上的提示输入必要的结构及尺寸公差等参数后，即可单击《画主动链轮零件图》按钮或《画从动链轮零件图》按钮。再按照输入框的提示输入零件图名和必要的绘图参数后，即可调用AutoCAD通用绘图软件自动绘出链轮零件图。若对该零件工作图不满意，可退出AutoCAD绘图软件，重新计算出参数后再生成新的零件图。也可直接利用AutoCAD通用绘图软件进行修改，直到满意为止。自动绘出的链轮零件图如图8-10所示。

图8-10链轮零件图

8.3.7滚动轴承选用

当单击主用户窗体界面上的命令按钮《滚动轴承选用》后，即出现《滚动轴承选用》窗体模块用户界面，如图8-11所示。

图8-11《滚动轴承选用》窗体模块用户界面

1． 滚动轴承的选用范围和已知条件

本模块适用于深沟球轴承、单列角接触球轴承和单列圆锥滚子轴承这三类轴承的选用计算。两个轴承采用同一型号。

1）已知条件:

（1）轴承1的径向载荷。

（2）轴承2的径向载荷。

（3）作用于轴上的工作轴向力代数和。

（4）轴承转速。

（5）轴承的预期使命寿命。

（6）轴承工作条件,包括载荷性质、工作温度及其它特殊要求。

2）使用方法。

只需按照用户界面上的提示输入已知条件，单击《开始计算》按钮即可。

当得到满意的设计结果后，即可单击《打印窗体》按钮，将已知条件及设计结果都打印下来。

8.3.8齿轮减速器轴的设计

轴设计部分包括轴上轮齿受力分析、按扭转强度条件初估轴最小直径、按弯扭合成强度条件设计或校核以及按安全系数法精确校核等几个独立的计算单元。并可自动生成齿轮减速器轴的零件图。该部分包括输入轴、中间轴及输出轴三个窗体模块，其用户界面如图8-12、图8-13和图8-14所示。

图8-12《输入轴设计》窗体模块用户界面

图8-13《中间轴设计》窗体模块用户界面

图8-14《输出轴设计》窗体模块用户界面

1．使用说明

设计时首先输入已知条件，再输入或选择模块界面上提示的设计条件后，即可进行设计计算。若按弯扭合成强度条件计算，应单击《按弯扭合成条件计算》按钮，若结果不满意，应重新选择已知条件及设计条件，再进行新一轮的设计计算。

对于输入轴或输出轴，若外伸端安装带轮或链轮且压轴力作用方向未确定，此时可按最不利的情况计算，即求出合成弯矩后，再与该力产生的附加弯矩算术相加，应用此算法时，要点击《外伸端压轴力方向未定》复选框（使其值为真），且输入相应的压轴力Fo值后,再开始计算。若外伸端压轴力作用方向已确定，则应计算出该压轴力在垂直和水平方向的分力，按Fro及Fto值输入（Fao=0）。若外伸端安装联轴器，则按Fro=Fto=Fao=0处理，此时应点击《外伸端装联轴器》复选框（使其值为真）。

对于中间轴，注意规定了受力简图中左边的大齿轮为从动齿轮，右边的小齿轮为主动齿轮。

当已知条件未给出作用力，而给出轴传递的扭矩等原始参数时，可利用《轴上轮齿作用力分析》计算单元，对圆柱或圆锥齿轮进行受力分析，计算结果自动显示在原始条件的《轴受力简图》栏内。

当得到满意的设计计算结果后，可单击《打印窗体》按钮，将全部已知条件和设计计算结果打印下来。

2．绘制轴零件图

当按弯扭合成强度条件计算完毕后即可绘制轴零件图。

对于输出轴，应首先根据轴上大齿轮的轴向定位朝向，利用组合框确定固定轴肩为《左轴肩》或《右轴肩》，然后按用户界面上的要求确定各轴段的直径及其它有关的结构参数，最后单击《画输出轴零件图》按钮即可。

对于输入轴，应首先利用组合框确定为《主动齿轮与轴为一体》或《主动齿轮安装在轴上》，然后按用户界面上的要求确定各轴段的直径及其它有关的结构参数，对于绘制齿轮轴，应通过对话框输入齿轮的有关参数。

对于中间轴，一对滚动轴承轴段内有两个齿轮，考虑到主动的小齿轮可能为《主动齿轮与轴为一体》或《主动齿轮安装在轴上》，因此与输入轴一样，应首先利用该组合框确定轴的类别后，再进行绘图操作。

轴设计模块可调用AutoCAD通用绘图软件，由对话框输入零件图名及必要的绘图参数后, 自动绘出轴零件图。若对该零件工作图不满意，可退出AutoCAD通用绘图软件，更改结构及绘图参数后重新生成新的零件图。也可直接利用AutoCAD绘图软件进行修改，直到满意为止。自动绘出的输入轴和中间轴零件图如图8-15、图8-16和图8-17所示。

图8-15输入轴零件图

图8-16中间轴零件图（小齿轮与轴不为一体）

图8-17中间轴零件图（小齿轮与轴为一体）

附录1. 各章编程范例的索引表

序号 通用过程名 功能 页码 3-1 Ch3_CreateSplineUsingTypedArray 使用 CreateTypedArray 方法创建样条曲线 41 3-2 Ch3_CalculateDistance 计算两点之间的距离 42 3-3 Ch3_ConnectToAcad1 利用New创建对AutoCAD部件的引用 43 3-4 Ch3_ConnectToAcad2 利用GetObject和 CreateObject函数创建对AutoCAD部件的引 43 4-1 Ch4_PositionApplicationWindow 改变应用程序窗口的位置和尺寸 44 4-2 Ch4_MaximizeApplicationWindow 最大化应用程序窗口 44 4-3 Ch4_CurrentWindowState 查询窗口当前状态 45 4-4 Ch4_ QNewTemplateFile 用QNewTemplateFile属性设置缺省样板图文件 45 4-5 Ch4_NewDrawing 使用 Add 方法创建新图形 45 4-6 Ch4_OpenDrawing 打开图形文件，先用Dir函数检查是否存在，否则用缺省样板图新建 46 4-7 Ch4_SaveActiveDrawing 首先检查是否存在未保存的更改，并验证用户是否选择“确定”以保存图形。用其当前名称保存当前的图形。然后再用新名保存该图形。 47 4-8 Ch4_CloseDrawing 循序处理图形文件集合，关闭所有打开的图形文件 47 4-9 Ch4_ZoomWindow 将当前图形缩放至两点定义的窗口 48 4-10 Ch4_ZoomScaled 使用指定的缩放比例放大当前图形 48 4-11 Ch4_ZoomCenter 居中对齐对象 49 4-12 Ch4_ZoomAllZoomExtents 将当前图形缩放至图形上所有内容和范围 49 4-13 Ch4_AddView 将命名视图添加到 Views 集合中 50 4-14 Ch4_DeleteView 删除 View 对象 50 4-15 Ch4_DeleteViewFromCollection 从 Views 集合中删除命名视图 50 4-16 Ch4_IteratingViewportWindows 将视口拆分为四个平铺视口 51 4-17 Ch4_UpdateDisplay 使用 Update 方法更新某图形对象，使其在 AutoCAD 中可见 52 4-18 Ch4_Example_Regen 修改字体文件，然后调用Regen方法更新图形中的所有视口 52 4-19 Ch4_ActiveTextStyle 重置当前对象，设置已更新对象的 ActiveTextStyle属性 53 4-20 Ch4_SetVariable 查询和设置不同数据类型的系统变量 54 4-21 Ch4_SendACommandToAutoCAD1 调用AutoCAD命令绘制一个圆（代码形式1） 55 4-22 Ch4_SendACommandToAutoCAD2 调用AutoCAD命令绘制一个圆（代码形式2） 55 4-23 Ch4_TrueColor 使用真彩色 56 4-24 Ch4_ LoadLinetype 将需要的线型加载入图形中 56 4-25 Ch4_ChangeCircleLinetype 改变图形对象的线型 57 4-26 Ch4_FindLayer 查找图层 58 4-27 Ch4_NewLayer 创建图层和设置为当前图层 59 4-28 Ch4_MoveObjectNewLayer 更改对象的图层 59 4-29 Ch4_LayerFreeze 冻结图层 60 4-30 Ch4_LayerLock 锁定图层 60 4-31 Ch4_PrefsSetDrafting 设置草图参数 61 4-32 Ch4_ImportingAndExporting 输出和输入其它格式的图形文件 63 4-33 Ch4_Layer 设置图层（及其颜色、线型、线宽）综合示例 64 5-1 Ch5_CreatePoint 创建点，然后更新 PDMODE 和 PDSIZE 系统变量 67 5-2 Ch5_AddLine 创建一条两点间的直线 68 5-3 Ch5_AddCircle 创建一个圆 68 5-4 Ch5_AddArc 创建一段圆弧 69 5-5 Ch5_AddEllipse 创建一个椭圆 69 5-6 Ch5_AddMLine 创建一条多线 69 5-7 Ch5_AddLightWeightPolyline 创建优化多段线 70 5-8 Ch5_EditPolyline 编辑多段线 71 5-9 Ch5_CreateSpline 创建样条曲线 72 5-10 Ch5_CreateSolid 创建一个四边形实体填充对象 73 5-11 Ch5_AddTrace 使用AddTrace方法绘制一个三角形填充实体对象 73 5-12 Ch5_CreateRegion 创建面域 74 5-13 Ch5_CreateCompositeRegions 进行差运算创建组合面域 75 5-14 Ch5_CreateHatch1 创建颜色深浅渐变的图案填充图 77 5-15 Ch5_CreateHatch2

创建图案填充综合例,包括改变填充图案比例和图案填充样式等 77 5-16 Ch5_CreateTableStyle

Ch5_AddTable 由当前的标准表样式生成新的表样式

创建一个表，然后应用其SetText属性在单元格插入文字 80 5-17 Ch5_CopyMoveObject 用Copy方法进行原位置复制。然后将复制出的圆用Move方法移动并改为红色 81 5-18 Ch5_RotateObject 创建一条闭合的优化多段线，然后将该多段线绕基点旋转 82 5-19 Ch5_OffsetObject 创建一条优化多段线，然后偏移该多段线 83 5-20 Ch5_ArrayingA 先创建一个圆，然后进行环形阵列 83 5-21 Ch5_ArrayRectangular 创建一个圆，然后对该圆执行矩形阵列操作 84 5-22 Ch5_Delete 删除操作 85 5-23 Ch5_LengthenLine 创建一条直线，然后修改其端点拉长这条直线 85 5-24 Ch5_ExplodePolyline 先创建一条二维多段线，然后进行分解操作 86 5-25 Ch5_ScalePolyline 对已有优化多段进行比例缩放 87 5-26 Ch5_MirrorText 镜像一段文字 88 5-27 Ch5_RenamingLayer 重命名对象 89 5-28 Ch5_UpdateTextFont 先利用GetFont方法获取当前文字样式的当前字体值，将字体名改为"仿宋_GB2312"，然后使用 SetFont 方法设置新的字体 90 5-29 Ch5_FontFileBigFontFile 通过更改文字样式对象的FontFile和BigFontFile属性设置工程汉字 91 5-30 Ch5_CreateText 创建单行文字 93 5-31 Ch5_TextAlignment 对齐单行文字 94 5-32 Ch5_FormatMText( ) 在多行文字中使用格式码 96 5-33 Ch5_InsertingABlockwithAttribute 创建带属性定义的粗糙度符号，并插入到模型空间 98 5-34 Ch5_AttachingExternalReference 将图形文件以外部引用方式附着到当前图形文件中 100 5-35 Ch5_AddDimAligned 创建对齐式标注 102 5-36 Ch5_AddDimRotated 创建旋转式标注 102 5-37 Ch5_CreateDiametricDimension 创建直径标注 102 5-38 Ch5_CreateRadialDimension 创建半径标注 103 5-39 Ch5_CreateAngularDimension 创建角度标注 103 5-40 Ch5_CreateOrdinateDimension 创建坐标标注 104 5-41 Ch5_ RotateAndOverrideDimensionText 利用Rotate方法旋转直径标注对象的标注方位；利用多行文字格式码、Unicode字符输入字符串；利用TextOverride属性替换标注文字的办法生成特殊要求的标注文字及尺寸公差 105 5-42 Ch5_ TextPrefixAndTextSuffix 利用TextPrefix 属性指定标注值的前缀；利用TextSuffix属性指定标注值后缀 109 5-43 Ch5_ CreateLeaderAddAnnotation 使用MText 对象作为注释来创建引线标注 111 5-44 Ch5_CreateTolerance 创建带引线的形位公差 113 5-45 Ch5_CreateNewDimStyle 通过改变系统变量的值，将角度单位由度改为度分秒，并将角度标注的文本精度改为3，最后从当前文挡对象复制出专用的标注样式。 115 5-46 Ch5_L2 生成带键槽轴剖面的部分代码 117 6-1 Ch6_Polyline_2D_3D 创建两条多段线，每条都有三个坐标。第一条是二维多段线，第二条是三维多段线 123 6-2 Ch6_NewUCS 创建新的用户坐标系并设置为当前的UCS 124 6-3 Ch6_AddExtrudedSolid 先由圆弧和直线生成面域，再按高度拉伸为三维实体 126 6-4 Ch6_AddExtrudedSolidAlongPath 先生成一个面域，再定义一条样条曲线，最后沿该曲线将面域拉伸成三维实体 128 6-5 Ch6_AddRevolvedSolid 先由圆弧和直线生成面域，再绕指定的轴线旋转形成三维实体 129 6-6 Ch6_RotateSolid 先创建一个三维长方体，然后定义旋转轴，将长方体绕该轴旋转 131 6-7 Ch6_CreateRectangularArray 创建一个球，然后用该球创建M列、N 行、L层的矩形阵列 131 6-8 Ch6_MirrorABox3D 在三维空间中镜像对象 132 6-9 Ch6_Boolean 先生成一个长方体和一个圆柱，然后进行布尔运算 133 6-10 Ch6_CheckInterference 先创建一个长方体和一个圆柱体，然后检查是否发生干涉 134 6-11 Ch6_SliceABox 先创建一个长方体，然后基于由三个点定义的平面剖切该长方体。剖切后的结果作为三维实体返回 136 6-12 Ch6_SectionSolid 先创建一个长方体，然后基于由三个点定义的平面剖切该长方体，所形成的交集以面域返回

137 6-13 Ch6_Lboolean 三维实体综合示例 139 7-1 Ch7_AddLineVBA 在VBA中创建一条直线 143 7-2 Ch7_AddLineVB 在VB中创建一条直线 144 7-3 Ch7_ConnectToAcad 引用AutoCAD应用程序对象及利用样板图创建图形文件 146 7-4 Ch7_Layer 增设图层 148 7-5 Ch7_AddHatch 创建轴的横剖面 150 7-6 Ch7_DimTextSample 生成带上下偏差的直径尺寸标注 151 7-7 Ch7_3DSolidBeltWheel 按旋转面域的方式创建V带轮的三维实体 154 

附录2. 综合应用示例清单（全部源程序均收录在本书附带的光盘上）

1．Acad_firstVBA.dvb（VBA引例）

2. Pjfq.vbp（VB引例:简易加法器）

3. Pjsq.vbp（VB引例:简易计算器）

4. L3-1.vbp（综合示例:创建对AutoCAD部件引用）

内容包括:

1）利用New创建对AutoCAD部件的引用

2）利用GetObject和CreateObject函数创建对AutoCAD部件的引用

3）利用CreateTypedArray方法创建样条曲线

4）求两点距离

5）访问图形实体

5. L4-1.vbp（综合示例:控制AutoCAD环境和组织图形元素）

内容包括:

1） 查询应用程序窗口状态

2） 查询图形窗口状态

3） 设置新图层及其线型、线宽和颜色

4） 查找图层

5） 查找线型

6） 设置为当前图层

7） 查询和设置系统变量

8） 设置草图参数

9） 设置图层及指定给对象

10）设置线型及指定给对象

11）设置真彩色及指定给对象

12) 平铺视口

13）查询直线与X轴的交角

14）求交点

15）输出和输入其它文件格式

6. L5-1.vbp（综合示例:图形对象创建与编辑）

内容包括:

1）调用AutoCAD命令

2）画点

3）画直线和圆

4）画圆弧

5）画椭圆

6）画非有理B样条曲线

7）应用AddSolid方法实体填充

8）应用AddTrace方法实体填充

9）画多线

10）创建及编辑多段线

11）分解多段线

12）编辑优化多段线凸度等

13）创建及编辑剖面线图案填充

14）创建面域

15）创建形文字及其镜像

16）创建TrueType汉字及其镜像

17）多行文字的编辑

18）加载形文件及插入形对象

19）创建及插入图块

20）外部引用

21）线性尺寸标注和公差

22）角度标注

23）标注引线及其注释

24）形位公差

25）设新标注样式

26）利用标注对象的属性添加前缀和后缀

27）利用标注对象的属性改变角度精度

28）改变公差值的高度比例

29）使用光栅图象

7. L5-2.vbp（综合示例:生成带键槽的轴剖面）

8. L6-1.vbp（综合示例:在三维空间工作）

内容包括：

1） 使用UCS

2） 定义和查询二维和三维多段线的坐标

3） 对多段线进行OCS与WCS的坐标变换

4） 例用基本实体形创建三维实体

5） 按高度拉伸生成三维实体

6） 沿路径拉伸生成三维实体

7） 绕轴线旋转生成三维实体

8） 三维实体剖切

9） 生成相交截面图

10）布尔运算综合示例

11）干涉检查

12）三维实体复制和移动

13）三维实体镜像

14）三维实体阵列

15）生成网格曲面

附录3. 参数化绘图应用程序清单（经编译后的可执行文件）

（收录在本书附带的光盘上）

1．Lchi-1.exe（圆柱齿轮零件图，适用于 AutoCAD 2004/05/06）

2．Lchi-1-R15.exe（圆柱齿轮零件图，适用于 AutoCAD 2000/2002）

3．Lzhui-1.exe（圆锥齿轮零件图，适用于 AutoCAD 2004/05/06）

4．Lzhui-1-R15.exe（圆锥齿轮零件图，适用于 AutoCAD 2000/2002）

5．Lzhou-1.exe（轴零件图，适用于 AutoCAD 2004/05/06）

6．Lzhou-1-R15.exe（轴零件图，适用于 AutoCAD 2000/2002）

7．Ldai-1.exe（带轮零件图和三维实体，适用于 AutoCAD 2004/05/06）

8.Ldai-1-R15.exe（带轮零件图和三维实体,适用于AutoCAD 2000/2002）

